



Solution Guide II-B

Matching



HALCON 25.11 *Progress*

How to Use Matching to Find and Localize Objects, Version 25.11.0.0

All rights reserved. No part of this publication may be reproduced, stored in a retrieval system, or transmitted in any form or by any means, electronic, mechanical, photocopying, recording, or otherwise, without prior written permission of the publisher.

Copyright © 2010-2025 by MVTec Software GmbH, Munich, Germany



Protected by the following patents: US 7,751,625, US 7,953,290, US 7,953,291, US 8,260,059, US 8,379,014, US 8,830,229, US 11,328,478. Further patents pending.

Microsoft, Windows, Microsoft .NET, Visual C++ and Visual Basic are either trademarks or registered trademarks of Microsoft Corporation.

All other nationally and internationally recognized trademarks and tradenames are hereby recognized.

More information about HALCON can be found at: <https://www.halcon.com>

About This Manual

In a broad range of applications matching is suitable to find and locate objects in images. This Solution Guide leads you through the variety of approaches that are provided by HALCON.

An introduction to the available matching approaches, including tips for the selection of a specific approach for a specific application, is given in [section 1](#) on page 7.

Some general topics that are valid for multiple approaches are discussed in [section 2](#) on page 17. These comprise, e.g., the selection of a proper template, tips for a speed-up, and the use of the results.

[Section 3](#) on page 47 then provides you with detailed information about the individual matching approaches.

The HDevelop example programs that are presented in this Solution Guide can be found in the specified subdirectories of the directory %HALCONEXAMPLES%.

Symbols

The following symbols are used within the manual:



This symbol indicates a **tip**.



This symbol indicates an information you should **pay attention** to.

Contents

1	Introduction	7
1.1	How to Use This Manual?	7
1.2	What is Matching?	7
1.3	How to Generally Apply a Matching?	8
1.4	Which Approaches are Available?	8
1.5	Which Approach is Suitable in Which Situation?	9
1.5.1	The Matching Approaches: 2D versus 3D	9
1.5.2	Decisions for 3D Objects and 2D Objects in 3D Space	10
1.5.3	First Decisions for Orthogonally Imaged 2D Objects	11
1.5.4	Shape-Based vs. Correlation-Based Matching	12
1.5.5	Quick Guide to the Matching Approaches	12
2	General Topics	17
2.1	Prepare the Template	17
2.1.1	Reduce the Reference Image to a Template Image	17
2.1.2	Influence of the Region of Interest	18
2.1.3	Synthetic Models as Alternatives to Template Images	20
2.2	Reuse the Model	25
2.3	About Subsampling - Image Pyramid	25
2.4	Speed Up the Search	27
2.4.1	Restrict the Search Space	28
2.4.2	Subsampling	28
2.4.3	The Concept of Greediness	29
2.4.4	Limit the Number of Candidates: MinScore and NumMatches	29
2.5	Use the Results of Matching	29
2.5.1	Results of the Individual Matching Approaches	30
2.5.2	About Transformations	30
2.5.3	Display Matches and Perform Transformations Using a 2D Position and Orientation	33
2.5.4	Visualize Matches and Perform Transformations Using a 2D Homography	40
2.5.5	Visualize Matches and Perform Transformations Using a 3D Pose	43
2.5.6	About the Score	45
3	The Individual Approaches	47
3.1	Correlation-Based Matching	47
3.1.1	A First Example	47
3.1.2	Select the Model ROI	48
3.1.3	Create a Suitable NCC Model	48
3.1.4	Optimize the Search Process	50
3.2	Shape-Based Matching	53
3.2.1	General Concept	54
3.2.2	A First Example	55
3.2.3	Select the Model ROI	56
3.2.4	Create a Suitable Shape Model	58
3.2.5	Train the Shape Model	64
3.2.6	Find Object Instances	64
3.2.7	Restrict the Search to a Region of Interest	70
3.2.8	Search for Multiple Models Simultaneously: ModelIDs	72
3.2.9	Optimize the Matching Speed	73

3.2.10	Visualize Score Contribution	75
3.2.11	Use Multiple Shape-Based Matching Results	76
3.2.12	Adapt to a Changed Camera Orientation	77
3.3	Local Deformable Matching	77
3.3.1	A First Example	78
3.3.2	Select the Model ROI	80
3.3.3	Create a Suitable Local Deformable Model	81
3.3.4	Optimize the Search Process	83
3.3.5	Use the Specific Results of Local Deformable Matching	85
3.4	Perspective Deformable Matching	87
3.4.1	A First Example	87
3.4.2	Select the Model ROI	89
3.4.3	Create a Suitable Perspective Deformable Model	89
3.4.4	Optimize the Search Process	92
3.4.5	Use the Specific Results of Perspective Deformable Matching	95
3.5	Descriptor-Based Matching	95
3.5.1	A First Example	96
3.5.2	Select the Model ROI	98
3.5.3	Create a Suitable Descriptor Model	98
3.5.4	Optimize the Search Process	100
3.5.5	Use the Specific Results of Descriptor-Based Matching	103

Chapter 1

Introduction

This section introduces you to HALCON's matching functionality. In particular, it provides you with an overview of

- how to use this manual ([section 1.1](#)),
- the general meaning of matching ([section 1.2](#)),
- how to generally apply a matching ([section 1.3](#)),
- the available approaches ([section 1.4](#)), and
- information about which approaches are suitable in which situation ([section 1.5](#) on page 9).

1.1 How to Use This Manual?

If you have no or only little experience with matching applications using HALCON, the following subsections introduce you to matching with HALCON in general and guide you through the most salient differences between the available matching approaches. Thus, they help you to select the matching approach that is best suited for your specific application. Further, we recommend to read [section 2](#) on page 17 before you step into the subsection of [section 3](#) on page 47 that in detail describes your selected matching approach.

If you are an experienced HALCON user that is familiar with matching and you are looking for further tips on how to optimize your specific application, it may be sufficient to immediately step into the subsection of [section 3](#) on page 47 that describes the matching approach of your choice and possibly have an additional look on selected subsections of [section 2](#) on page 17, which describe some general topics that are needed for multiple matching approaches.

1.2 What is Matching?

With matching, HALCON provides a method for the robust location of objects in images, which can be used for many different applications. To suit the different requirements of the applications, different approaches are available. All approaches consist of a small set of operators for which only a few parameters have to be adjusted. Further, none of the approaches requires an explicit segmentation of the objects in the images. Thus, you can successfully locate your objects even if you have no special knowledge in machine vision.

The main idea of matching is to use a prototype object (template), create a model of it and search the model in other images. For most matching tasks you obtain the model from a **reference image** that shows the object of interest. To suppress other structures or objects that are contained in this image, the image is reduced to a region of interest (ROI) that only contains the object and which may have an arbitrary shape. The reduced image is the **template image** from which the model is created by an approach-specific operator. Another approach-specific operator uses the obtained model to find the object in different search images. That is, it searches for image structures that match the model (within small tolerances).

The different matching approaches provided by HALCON differ amongst others in the image structures that are used to build the model. Some matching approaches use, e.g., the relations of gray values to their surrounding (neighborhood) to build a model. Others use, e.g., the shapes of contours (see [figure 1.1](#)). The result of the matching is in each case information about the position, for most approaches the orientation, and for some approaches also the scale of the found object in the search image.




Figure 1.1: Matching using shapes of contours: a) the image of an attention sign is used to build a b) (shape) model in different resolutions (pyramid levels), which is used to c) locate an instance of the model in an image.

1.3 How to Generally Apply a Matching?

Although different matching approaches use different sets of operators, the main proceeding is similar for all and consists of the following main steps:

1. **Create a model** for the object of interest. A model of an object, i.e., the internal data structure describing a searched object, is created using either a representative reference image of it or a synthetic model. Which way to follow depends on the selected approach and the available data. Certain matching approaches require a subsequent model training.
2. **Find the model** in search images. Instances of the previously created model are searched in images and their position, in most cases their orientation, and for some approaches also their scale are returned.

Additional to these essential steps, most approaches provide means to **modify** a model, **reuse** it (i.e., store it to file and read it from file), and to **query information** from it.

1.4 Which Approaches are Available?

The matching approaches described in this Solution Guide comprise

- an approach that describes a model by the gray-value relations of the contained pixels:
 - the **correlation-based matching** ([section 3.1](#) on page 47), which uses normalized cross correlation (NCC) to match objects or patterns, respectively.
- approaches that describe a model by the shapes of its contours:
 - the **shape-based matching** ([section 3.2](#) on page 53),
 - the **local deformable matching** ([section 3.3](#) on page 77), which can handle and return local deformations of the object and allows to rectify the image part containing the deformed model, and
 - the **perspective deformable matching** ([section 3.4](#) on page 87), which can handle also perspective distortions and provides a calibrated version with which also a 3D pose instead of 2D transformation parameters can be derived.

- an approach that describes a model by a set of significant image points:
 - the **descriptor-based matching** (section 3.5 on page 95) matches a set of so-called interest points. Similar to the perspective deformable matching it can handle perspective distortions and provides a calibrated version with which also a 3D pose instead of 2D transformation parameters can be derived.

Additionally, the so-called point-based matching and the 3D matching are available. For the **point-based matching**, corresponding points are used to combine overlapping images. This method is also called uncalibrated mosaicking. In the broader sense, the search for corresponding points is also a kind of matching, but the focus of this Solution Guide is on matching of “2D objects”. The **3D matching** consists of different methods and is shortly introduced in the following section, but is also not subject to this Solution Guide. Please refer to the Solution Guide I, chapter 11 on page 101 for further details on 3D matching.

1.5 Which Approach is Suitable in Which Situation?

The first step when selecting a matching approach is to decide if two or three spatial dimensions are needed. Afterwards, further criteria like the needed transformation parameters or the object’s appearance within the images can be included into the decision. The following sections provide you with the background for your decisions. For a quick overview, this background is clearly illustrated in the figures of section 1.5.5 on page 12.

1.5.1 The Matching Approaches: 2D versus 3D

The different matching approaches are suitable for different applications. Some are suitable only for 2D objects imaged from an orthogonal view, others can also handle perspective distortions, and some approaches even match 3D shapes in a full 3D space (see figure 1.2). Note that with the term “2D object” a fixed view on a planar object part is meant and not the actual tangible object, which in reality is naturally three-dimensional. In contrast, the “3D object” is an object that is viewed from arbitrary directions.

- 2D objects, imaged from an orthogonal view:

The **correlation-based, shape-based, and local deformable matching** can be used to find 2D objects in images. The objects must be taken from an orthogonal view for the reference image as well as for the search images. Theoretically, you can also use the perspective deformable or the descriptor-based matching to find orthogonally imaged 2D objects, but as these are not as fast as the strict 2D approaches, they are recommended rather for the case that the objects must be imaged from a perspective view.

- 2D objects, imaged from a perspective view:

The **perspective deformable and the descriptor-based matching** can be used to find planar object parts, too. But in addition to the approaches used for the orthogonal view, the objects may be perspective deformed. Additionally, if a camera calibration was applied, not only the 2D position and orientation but the 3D pose of the object can be derived. Although both approaches can be used also for orthogonally imaged 2D objects, they are recommended rather for the perspective case, as they are not as fast as the strict 2D approaches.

- 3D objects:

The **3D matching** searches for “real” 3D objects in 2D images. Here, different approaches are available. For example, for the shape-based 3D matching, no reference image is used to create the model but a synthetic 3D model, in particular a DXF CAD model, must be provided. If a 3D object contains a characteristic planar part which is visible in all search images, alternatively the calibrated perspective 2D approaches, i.e., the **perspective deformable** or the **descriptor-based matching**, can be used, which are more convenient and significantly faster. The 3D matching is needed only if more than one planar part of the object is needed to differentiate the object from other image parts.

Note that the 3D matching is not part of this Solution Guide! For further information on 3D matching, please refer to the Solution Guide I, chapter 11 on page 101.






Figure 1.2: Available 2D and 3D matching approaches.

1.5.2 Decisions for 3D Objects and 2D Objects in 3D Space

If you search for a **complex 3D object** that differs from other objects in all three dimensions (see, e.g., [figure 1.3](#)), the selection of the appropriate matching approach is easy as you have to use the **3D matching**.




Figure 1.3: 3D object: the third dimension is needed to locate the object.

If the **3D object contains a unique but planar part** that significantly differs from the other structures in the expected images or if you search for a planar object that may be oriented arbitrarily in the 3D space you can also use the **perspective deformable** or the **descriptor-based matching**, which can handle perspective distortions of 2D objects. Both approaches are faster and more convenient to use than the 3D matching.

Which of both perspective approaches to select depends on the specific application. If the object of interest is expected to be anisotropically **scaled** in the search images, only the perspective deformable matching is suitable. If it is expected to be only translated and rotated, both approaches are suitable and the appearance of the object in the images must be taken into account. The most important difference between both approaches is the way the object is modeled. The perspective deformable matching describes the model by the shapes of the object's contours. Thus, it is suitable for objects that contain **clearly visible contours** (see, e.g., [figure 1.4](#)).

In contrast, the descriptor-based matching describes the model by interest points. Thus, for objects that are rather **characterized by an arbitrary but fixed texture** (see, e.g., [figure 1.5](#)), it is most probably to be preferred.




Figure 1.4: 3D object: unique parts of the object lie in a plane and can be described by clearly visible contours.



Figure 1.5: 3D object: a plane object part is characterized by a specific texture.

1.5.3 First Decisions for Orthogonally Imaged 2D Objects

If you can image planar objects from an orthogonal view and only the 2D transformation parameters of the found object instances are needed, the perspective deformable or the descriptor-based matching would lead to the desired result as well, but the approaches that are strictly restricted to two dimensions are to be preferred as they are significantly faster. The 2D approaches comprise the correlation-based, shape-based, and local deformable matching.

Three of these approaches are used only in specific situations:

- The **local deformable matching** is applied only if the **object of interest is locally deformed**, e.g., because of a contorted surface (see [section 3.3](#) on page 77 and [figure 1.6](#)). In contrast to the other 2D approaches, it returns only the position but no orientation for the found model instance. In exchange, it allows to rectify the image part containing the deformed object and returns a vector field that describes the actual deformations.

In most orthogonal 2D cases, you have to select one of the two remaining approaches, i.e., either the **shape-based matching** or the **correlation-based matching**.



Figure 1.6: Object with local deformations.

1.5.4 Shape-Based vs. Correlation-Based Matching

To decide if the shape-based or the correlation-based matching is more appropriate for your specific application, you should further investigate the requirements of your application.

For example, you should know which **transformation parameters** are needed to describe the relation of the object in the search image to the object described by the model. Will the object be only translated and rotated or also scaled? If a **scaling** is needed, correlation-based matching can not be used, but one of the shape-based matching approaches must be used. Here, you can further choose between approaches that use uniform scaling or anisotropic scaling, i.e., a scaling with different scaling factors for the x- and y-direction.

Additionally, the **appearance of the object** may change from image to image. Reasons can be, e.g., occlusions, clutter, or illumination changes that may lead to a changing polarity of the object. Furthermore, the images may be defocused or the object is a complex pattern in front of a complex background, i.e., it is textured. To get a robust and fast result, the changes of the object's appearance should be minimized as much as possible already when imaging the object. Nevertheless, sometimes distortions like occlusions, clutter, or defocus can not be avoided. These distortions have to be involved into the decision which approach to use for a specific application.

In particular, shape-based matching should be chosen if occlusions (see, e.g., [figure 1.7](#)) or clutter can not be avoided or if a matching of objects with changing color is applied.



Figure 1.7: Occlusions can be handled by shape-based matching.

In contrast, correlation-based matching is suitable for objects with a random and changing texture or for objects with a slightly changing shape (see [figure 1.8](#)). Additionally, correlation-based matching is to be preferred when handling strongly defocused images (see [figure 1.9](#)).

1.5.5 Quick Guide to the Matching Approaches

[Figure 1.10](#) to [figure 1.15](#) clearly summarize the information needed to decide which matching approach is suitable for a specific application. In particular,




Figure 1.8: Changing shapes can be handled by correlation-based matching.




Figure 1.9: Defocused shapes can be handled by correlation-based matching.

- [figure 1.10](#) summarizes the first coarse decision steps needed to select a matching approach,
- [figure 1.11](#) and [figure 1.12](#) illustrate the transformations that can be handled by the individual matching approaches,
- [figure 1.13](#) lists the transformation parameters that are returned by the individual matching approaches, and
- [figure 1.14](#) and [figure 1.15](#) introduce typical changes of the appearance of objects and show which characteristics of the appearance can be handled by the individual matching approaches.




Figure 1.10: First decision steps when selecting the matching approach.




Figure 1.11: Geometric transformations of an object.

Matching Approach	Translation (2D)	Rotation (2D)	Scaling (uniform)	Scaling (anisotropic)
correlation-based	X	X	-	-
shape-based	X	X	X	X
local deformable	X	X	X	X
perspective deformable	X	X	X	X
descriptor-based	X	X	X	-

Figure 1.12: Transformations that can be handled by the matching approaches.

Matching Approach	Position (2D)	Angle (2D)	Scale (uniform)	Scale (anisotropic)	Projective Transformation Matrix	Pose (3D)
correlation-based	X	X	-	-	-	-
shape-based	X	X	X	X	-	-
local deformable	X	-	-	-	-	-
perspective deformable	-	-	-	-	X	X
descriptor-based	-	-	-	-	X	X

Figure 1.13: Transformation parameters returned by the matching approaches.




Figure 1.14: The object's appearance may change, e.g., because of (from left to right): occlusions, clutter, and illumination changes.

Matching Approach	Occlusion	Clutter	Illumination changes	Texture	Color	Defocus
correlation-based	-	-	X (only linear)	X	-	X
shape-based	X	X	X	-	X	X
local deformable	X	X	X	-	X	-
perspective deformable	X	X	X	-	X	-
descriptor-based	X	X	X	X (needed)	-	-

Figure 1.15: Characteristics of the appearance the matching approaches can cope with.

Chapter 2

General Topics

A **general proceeding** for a matching is to prepare the reference image, e.g., by a preprocessing (if necessary), prepare the template, create the model from it, modify the already existing model (if necessary), store it to file (if it should be reused), query information from it (which is needed, e.g., when reusing a previously stored model), restrict the search space to speed up the following matching, search for the model in (possibly preprocessed) search images, and further process the result of the matching.

In the following, we discuss different **general topics** that are valid for many matching approaches at once. In particular,

- [section 2.1](#) shows how to **prepare a template**,
- [section 2.2](#) on page 25 shows how to **reuse a model**,
- [section 2.4](#) on page 27 shows how to **speed up the search**, and
- [section 2.5](#) on page 29 shows how to **further process the results** of matching.

2.1 Prepare the Template

After an optional preprocessing of the reference image, the first step of the matching is to prepare a template of the object of interest. From this, a model is derived, which in a later step is used to locate instances of the object in search images. In most cases, the model is derived from a reference image, which is reduced to a so-called template image. How to obtain this template image using a region of interest (ROI) is described in [section 2.1.1](#). The influence of the ROI on the further matching process is described in [section 2.1.2](#). For some matching approaches, a synthetic model can be used instead of the template image. This can be either a synthetically created template image or an XLD contour (see [section 2.1.3](#) on page 20).

2.1.1 Reduce the Reference Image to a Template Image

To create a model from a reference image, which is the common way for most matching approaches, the reference image must be reduced to a template image that contains only those structures of the reference image that are needed to derive the model.

For this, you select a region within the reference image that shows the part of the image that should serve as the template or, respectively, from which the model should be derived. After selecting the region, the domain of the reference image is reduced to an ROI using the operator [reduce_domain](#). The resulting image is our template image, which is input to one of the approach-specific operators that are provided for the actual model generation.

Note that a region and therefore also the model can have an arbitrary shape (see the Quick Guide, [section 2.1.1](#) on page 10 for information to regions in HALCON). A region can be created in different ways, as is described in the Solution Guide I, [chapter 3](#) on page 25 and [chapter 4](#) on page 33. Summarized, a **region can be selected**, e.g., by the following means:

- **A region can be specified by explicitly defining its parameters.** That is, HALCON offers multiple operators to create regions, ranging from standard shapes like rectangles ([gen_rectangle2](#)) or ellipses ([gen_ellipse](#)) to free-form shapes (e.g., [gen_region_polygon_filled](#)). These operators can be found in the HDevelop menu Operators ▸ Regions ▸ Creation. To use the operators, you need the parameters of the shape you want to create, e.g., the position, size, and orientation of a rectangle or the position and radius of a circle. If these parameters are not explicitly known, you can get them using draw operators, i.e., operators that let you draw a shape on the displayed image and then return the shape parameters. These operators are available, e.g., in the HDevelop menu Operators ▸ Graphics ▸ Drawing.
- **A region can be specified by image processing** using, e.g., a blob analysis. Then, the image is segmented, e.g., using a [threshold](#) operator, and the obtained region is further processed to select only those parts of it having specific features (commonly applied operators are, e.g., [connection](#), [fill_up](#), and [select_shape](#)). In HDevelop, you can determine suitable features and values using the dialog Visualization ▸ Feature Inspection. More complex regions can be created using set theory, i.e., by adding or subtracting standard regions using the operators [union2](#) and [difference](#). That way, e.g., also ring-shaped ROIs can be created like is shown in [section 2.1.2.2](#).

Before creating the ROI, it is often suitable to optimize the reference image by a preprocessing. Note that when using shape-based matching, you can use an HDevelop Assistant that guides you through the matching process, which includes also the preprocessing of the reference image as well as the creation of the ROI. How to use the Matching Assistant is described in the HDevelop User's Guide, [section 7.3](#) on page 202.

Note that also the gray values outside of the ROI have an influence on the model generation. In particular, the influence is approximately $2^{NumLevels}$ pixels, with n being the number of pyramid levels. Thus, the gray values outside of the ROI selected in the reference image should be similar to those that will occur in the search images.

2.1.2 Influence of the Region of Interest

The ROI used when creating the model determines the quality of the model and thus strongly influences the success of the later search. If the ROI is selected inappropriately, no or the wrong image structures are identified as instances of the model in the search images. So, if a matching is not successful or the result is inaccurate, you should have a very critical look at your ROI and try to enhance it such that the model represents the object and not the clutter that may be contained in the image. For approaches that are based on contours like shape-based matching and local or perspective deformable matching, you can use the operator [inspect_shape_model](#) to visually check a potential model. To create a model of good quality, you must pay attention to the proper selection of the ROI's center, i.e., the "point of reference" (see [section 2.1.2.1](#)) as well as to the proper selection of the ROI's outline (see [section 2.1.2.2](#)).

2.1.2.1 The Point of Reference

The ROI on the one hand influences the quality of the model and thus the general success of the subsequent matching. On the other hand, also the numerical results returned by the matching are influenced. In particular, the center point of the ROI by default acts as the so-called **point of reference** of the model for the estimated position, rotation, and scale. Note that if no ROI is selected, the point of reference is located at the center of the image (see [figure 2.1](#)).

The point of reference also influences the success of the later search, as an object is only found if the point of reference lies within the image, or more exactly, within the domain of the image (see also [section 3.2.7](#) on page 70). That is, if a point of reference is placed away from the actual object (as shown, e.g., in [figure 2.1](#), left), a later matching might fail although the object itself is fully contained in the search image, just because the point of reference is not contained. Thus, it is **very important to select an appropriate ROI** even if the template object is the only object in the image. Note that for some approaches the matches may be refined, which can result in found matches having their point of reference slightly outside the image domain.

For some approaches, after creating a model, you can modify the point of reference. But as a modified point of reference may lead to a decreased accuracy of the estimated position (see [section 3.2.6.6](#) on page 69), **if possible, the point of reference should not be changed!** Additionally, even if you modified the point of reference, the test, if the point of reference lies within the domain of the search image, is always performed for the original point of reference, i.e., the center point of the initially selected ROI. Thus, the selection of an appropriate ROI for the template image is important right from the start.




Figure 2.1: The point of reference depends on the selected ROI, not on the object or its bounding box.

2.1.2.2 The Outline of the ROI

The quality of the model and thus the accuracy of the result of the matching is influenced strongly by the model's outline, in particular, by its shape, size, and orientation.

To allow a good quality of the model, the ROI should be selected so that noise or clutter are minimized. This can be obtained, e.g., by “masking” parts of the object that contain clutter. In [figure 2.2](#), e.g., the model of a capacitor is needed for shape-based matching. To exclude clutter as much as possible, instead of a circular ROI a ring-shaped ROI is created using the difference between two circular regions.

```
draw_circle (WindowHandle, ROI1Row, ROI1Column, ROI1Radius)
gen_circle (ROI1, ROI1Row, ROI1Column, ROI1Radius)
gen_circle (ROI2, ROI1Row, ROI1Column, ROI1Radius - 8)
difference (ROI1, ROI2, ROI)
reduce_domain (ModelImage, ROI, ImageROI)
```




Figure 2.2: Masking the part of a region containing clutter.

Additionally, the accuracy of the location of the object is influenced by the number of contour points contained in the model. That is, a model with many contour points can be found more accurately than a model with few contour points (see [figure 2.3](#)).




Figure 2.3: Location accuracy: the accuracy is better for models with many contour points.

Furthermore, the accuracy of a matching result can vary for different directions, depending on the direction of the contours that are contained in the model. If, e.g., a model consists mainly of horizontal contours as depicted in [figure 2.4a](#), a good vertical accuracy can be obtained but the horizontal accuracy is bad. The model in [figure 2.4b](#) contains contours in vertical and horizontal direction. Thus, a sufficient accuracy in both directions can be obtained. The optimal pattern for a good accuracy in all directions would be a circular structure as shown in [figure 2.4c](#). So please, carefully select the model so that it contains enough contours that are perpendicular to the direction that you want to inspect or measure after the matching.




Figure 2.4: Direction of accuracy depends on the direction of the contours: (a) good vertical but bad horizontal accuracy, (b) sufficient accuracy in both direction, (c) optimal pattern for good accuracy in all directions.

The accuracy of the rotation angle returned by the matching depends on the one hand on the distance of the contour points from the rotation center and on the other hand on the orientation of the contour in relation to the rotation center (see [figure 2.5](#)). In particular, points that are far away from the rotation center can be determined more accurately. That is, assuming the same number of model contours, the orientation of a contour can be determined more accurately if the rotation center lies in the center of the contour and the contour is elongated.




Figure 2.5: Rotational accuracy: the angle for the (left) compact contour is not determined with the same accuracy as the angle for the (right) extended model that is oriented radially to the rotation center.

2.1.3 Synthetic Models as Alternatives to Template Images

For some matching approaches, synthetic models can be used instead of a template image. Three means to work with synthetic models are available:

- create a synthetic template image,
- use XLD contours directly as models (this can be applied only for specific matching approaches as listed in [section 2.1.3.2](#) on page 22), or
- use a DXF model to derive XLD contours, which then can be used directly as model (this can be applied only for specific matching approaches as listed in [section 2.1.3.3](#) on page 24) or which can be used to create a synthetic template image.

A note of caution when using synthetic images:

On synthetic images there is a very sharp gray value transition from the the model to the background. Using such an image for the model definition may lead to automatically estimated parameter values which are not suitable in search images (where the gray values fluctuate and the transition is flattened). An example for such a parameter is Contrast. Furthermore, sharp gray value transitions may lead to ringing artifacts on higher pyramid levels. Hence, when generating the model using synthetic images we recommend to inspect the created model and eventually readjust certain automatically guessed parameters.




Figure 2.6: Locating capacitors using (top) a circular ROI or (bottom) a complex ring-shaped ROI to derive the model.

2.1.3.1 Synthetic Template Image

Synthetic template images are suitable mainly for correlation-based matching and all 2D approaches that are based on contours, i.e., shape-based, local deformable, and perspective deformable matching.

Depending on the application it may be difficult to create a suitable model from a reference image because no image is available that contains a perfect, easy to extract instance of the object. An example of such a case is depicted in [figure 2.6](#) for the location of capacitors. The task seems to be simple at first, as the capacitors are represented by prominent bright circles on a dark background. But the shape model that is derived from a circular ROI is faulty, because inside and outside the circle the image contains clutter, i.e., high-contrast points that are not part of the object. A better result is obtained for a ring-shaped ROI that “masks” the parts of the object containing clutter. Nevertheless, the model is still not perfect, because parts of the circle are missing and the model still contains some clutter points.

In such a case, it is often better to use a synthetic template image. How to create such an image for the capacitors is explained below. To follow the example actively, start the HDevelop program `%HALCONEXAMPLES%\solution_guide\matching\synthetic_circle.hdev`.

Step 1: Create an XLD contour

First, we create a circular region using the operator `gen_ellipse_contour_xld`. You can determine a suitable radius by inspecting the image with the HDevelop dialog `Visualization > Zoom Window` or more conveniently create the region using the ROI tool in HDevelop’s graphics window.




Figure 2.7: Locating capacitors using a synthetic model: a) paint region into synthetic image; b) corresponding model; c) result of the search.

```
RadiusCircle := 43
SizeSynthImage := 2 * RadiusCircle + 10
gen_ellipse_contour_xld (Circle, SizeSynthImage / 2, SizeSynthImage / 2, 0, \
                        RadiusCircle, RadiusCircle, 0, 6.28318, \
                        'positive', 1.5)
```

Note that the synthetic image should be larger than the region because also pixels outside the region are used when creating the image pyramid for the shape-based matching, which in this case is the selected matching approach (for image pyramids, see [section 2.4.2](#) on page 28).

Step 2: Create an image and insert the XLD contour

Then, we create an empty image using the operator `gen_image_const` and insert the XLD contour with the operator `paint_xld`. In [figure 2.7a](#) the resulting image is depicted.

```
gen_image_const (EmptyImage, 'byte', SizeSynthImage, SizeSynthImage)
paint_xld (Circle, EmptyImage, SyntheticModelImage, 128)
```

Step 3: Create the model

The model is created and trained using the synthetic image.

```
create_generic_shape_model (ModelID)
set_generic_shape_model_param (ModelID, 'iso_scale_min', 0.8)
set_generic_shape_model_param (ModelID, 'iso_scale_max', 1.2)
train_generic_shape_model (SyntheticModelImage, ModelID)
```

[Figure 2.7b](#) shows the corresponding model region and [figure 2.7c](#) shows the search results. Note how the image itself, i.e., its domain, acts as the ROI in this example.

2.1.3.2 Models from XLD Contours

For the shape-based matching and the local and perspective deformable matching you do not have to create a synthetic template image to derive a model from an XLD contour, because you can use the XLD contour directly as model. Then, e.g., for shape-based matching, you do not have to provide an image and select an ROI, you can simply call `create_generic_shape_model` (while e.g., for perspective deformable matching you use the xld-version of the create operator: `create_local_deformable_model_xld`). An example for the creation of a shape model from circular XLD contours is given by the HDevelop example program `%HALCONEXAMPLES%\hdevelop\Matching\Shape-Based\create_shape_model_xld.hdev`:

```

gen_circle_contour_xld (ContCircle, 300, 300, MeanRadius, 0, 6.28318, \
                        'positive', 1)
create_generic_shape_model (ModelID)
set_generic_shape_model_param (ModelID, 'metric', 'ignore_local_polarity')
train_generic_shape_model (ContCircle, ModelID)

```

Note that when creating the model from XLD contours, there is no information about the polarity of the model available (see [figure 2.8](#), left). Thus, when creating the model the parameter 'metric' must be set so that the polarity is locally ignored. As this leads to a dramatically slow search, HALCON provides means to determine the polarity for a found model instance. That is, you apply the slow search once only in a search image with a polarity that is representative for your set of search images, project the model contours to the found position within the search image (see [figure 2.8](#), right), and call the operator `set_shape_model_metric` for shape-based matching (or the corresponding one for other matching approaches). Then, polarity information is stored in the model and for the following search passes the parameter 'metric' can be set to a more suitable value, e.g., to 'use_polarity'. This proceeding is strongly recommended for a fast and robust search.

```

find_generic_shape_model (Image, ModelID, MatchResultID, NumMatchResult)
... accessing the indices of the matches
... that represent suitable drill holes
get_generic_shape_model_result (MatchResultID, 0, 'hom_mat_2d', HomMat2D)
set_shape_model_metric (Image, ModelID, HomMat2D, 'use_polarity')
for Index := 2 to 9 by 1
    read_image (Image, 'brake_disk/brake_disk_part_' + Index$'02d')
    find_generic_shape_model (Image, ModelID, MatchResultID, NumMatchResult)
endfor

```

For further information about the polarity of models for shape-based matching, please refer to [section 3.2.4.5](#) on page 62.




Figure 2.8: (left) synthetic XLD model; (right) model projected into a representative search image to determine the polarity of the model.

The Point of Reference

Just like for ROIs, a model created from XLD contours has its point of reference in its center of gravity. Additionally to the aspects mentioned in [section 2.1.2.1](#) on page 18 there is one more point to consider when dealing with models created from XLDs.

Those models have their reference point located at the origin of the image coordinate system (see [figure 2.9](#)). Therefore, part of the model has negative coordinates. This must be kept in mind when, e.g., adding a clutter region for the model, as the spatial relation between model and clutter region must be considered. See [section 3.2.6.5](#) on page 67 to see how to set a clutter region for a model created from an XLD contour.




Figure 2.9: The model (orange) created from XLD contours contains negative coordinates.

2.1.3.3 Models from DXF Files

The models needed for matching can also be derived from DXF files. In particular, you can extract the XLD contours from the DXF file using the operator `read_contour_xld_dxf` and then either use the XLD contours to create a synthetic image or, if the selected matching approach allows it (see [section 2.1.3.2](#) on page 22), use the obtained XLD contours directly as model.

An example for the creation of a synthetic template image for shape-based matching using the XLD contours extracted from an DXF file is given by the HDevelop example program `%HALCONEXAMPLES%\hdevelop\Applications\Object-Recognition-2D\pm_multiple_dxf_models.hdev`. Examples for the creation of perspective deformable models using the XLD contours extracted from DXF files directly as models are given by the HDevelop example programs `create_planar_calib_deformable_model_xld.hdev` and `create_planar_uncalib_deformable_model_xld.hdev` (see [figure 2.10](#)) in the directory `%HALCONEXAMPLES%\hdevelop\Matching\Deformable`.




Figure 2.10: (left) synthetic XLD model accessed from a DXF model; (right) model found in a search image.

2.2 Reuse the Model

If you want to reuse created models or training results in other HALCON applications, all you need to do is to store the relevant information in files and then read them again.

The HDevelop example program `%HALCONEXAMPLES%\solution_guide\matching\reuse_model.hdev` shows exemplarily how to reuse a model for a uniformly scaled shape-based matching. First, the model is created.

```
create_generic_shape_model (ModelID)
set_generic_shape_model_param (ModelID, 'iso_scale_min', 0.6)
set_generic_shape_model_param (ModelID, 'iso_scale_max', 1.4)
```

Then, the model is stored in a file using the operator `write_shape_model`. With the model, HALCON automatically saves the XLD contour, the point of reference, and the parameters that were used in the calls to `set_generic_shape_model_param`.

```
write_shape_model (ModelID, ModelFile)
```

Note that the model region, i.e., the domain of the image, is not saved when storing the model. Thus, if you want to reuse it, too, you can store the template image with `write_image`.

```
ModelRegionFile := 'model_region_nut.png'
write_image (ImageROI, 'png', 0, ModelRegionFile)
```

The model is read using the operator `read_shape_model`.

```
read_shape_model (ModelFile, ReusedModelID)
```

Now, the model can be used as if it was created in the application itself. We set parameters specifying the search, in this example the value for 'min_score', using `set_generic_shape_model_param`.

```
set_generic_shape_model_param (ReusedModelID, 'min_score', 0.8)
```

Model instances are searched and displayed on a new image.

```
read_image (SearchImage, 'rings_and_nuts')
find_generic_shape_model (SearchImage, ReusedModelID, MatchResultID, \
                          NumMatchResult)
get_generic_shape_model_result_object (Objects, MatchResultID, 'all', \
                                      'contours')
dev_display (Objects)
```

Note that the information that is stored in a model and that can be queried after reading a model from file depends on the selected approach.

2.3 About Subsampling - Image Pyramid

For all matching approaches except the descriptor-based matching a so-called image pyramid can be used to speed up the search. In some approaches, the image pyramid is created for the search image as well as for the template image. The pyramid consists of the original, full-sized image and a set of downsampled images. For example, if the original image (first pyramid level) is of the size 600x400, the second level image is of the size 300x200, the third level 150x100, and so on. The object is then searched first on the highest pyramid level, i.e., in the smallest image. The results of this fast search are then used to limit the search in the next pyramid image, whose results are used on the next lower level until the lowest level is reached. Using this iterative method, the search is both fast and accurate. In case of noise it can even improve the robustness of the search, but for artificial images ringing




Figure 2.11: The image and the model region at four pyramid levels (original size and zoomed to equal size).

artifacts may appear. Figure 2.11 depicts four levels of an example image pyramid together with the corresponding model regions.

Note that when images are downsampled, different “events” occur. Primarily, specific structures disappear in higher pyramid levels. As shown in figure 2.11, thin structures disappear sooner than thick ones. Note that “thin structures” includes the different structures of the actual model but also the distances between them. Thus, small structures of the model may disappear, but they can just as well be merged if the distances between them are small. Additionally, the boundaries of structures blur in higher pyramid levels and their contrast decreases. In figure 2.12, this effect is shown for a one pixel wide line.



Figure 2.12: In the pyramid, a (left) one pixel wide high contrast line blurs to a (right) broader line with a lower contrast.

For more complex patterns, this may lead to the disappearance of a pattern or even to the creation of a new pattern. Examples for the disappearance of patterns that are caused by the blurring of structures are given for the small characters in [figure 2.11](#), for a regular grid of one pixel wide high contrast lines in [figure 2.13](#), and for a pattern that consists of high contrast rectangles with a distance of one pixel in [figure 2.14](#). The latter example also shows that new patterns can be created within the pyramid. Before the pattern is blurred to a homogeneous area, a different regular pattern is created. Additionally, it is possible that new structures appear because the region of interest was selected too large and thus, contours that were not part of the model in the first pyramid level become part of the model in a higher level.




Figure 2.13: In the pyramid, a (left) grid with one pixel wide high contrast lines becomes a (right) homogeneous gray-value area.




Figure 2.14: In the pyramid, a (top) regular structure of rectangles with distances of one pixel shows the following behavior: (bottom) level 1 and 2 still represent the correct model contours, level 3 and 4 produce new regular patterns, and in level 5 a homogeneous gray-value area is obtained.

The blurring of structures induces also that besides the size also the contrast determines if a structure is still contained in a higher pyramid level or not. After the downsampling, a high contrast structure of sufficient size may still have enough contrast to be recognized, whereas a structure of the same size but with a low contrast can not be distinguished from the background anymore. For example, in [figure 2.15](#) the large but bright 'V' of the 'MVTec' logo disappears in front of the bright background at a pyramid level in which the large high contrast characters are still contained.

2.4 Speed Up the Search

There are different ways you can speed up the search. Here we present the general concepts of




Figure 2.15: In the pyramid, a part of the model with a low contrast (the 'V' of MVTec) disappears.

- restricting the search space ([section 2.4.1](#))
- using subsampling ([section 2.4.2](#))
- using a “greedy” search to reject candidates early ([section 2.4.3](#))
- limiting the number of candidates directly or by requiring a minimal score ([section 2.4.4](#))

2.4.1 Restrict the Search Space

An important concept in the context of finding objects is that of the so-called search space. Quite literally, this term specifies where to search for the object. Depending on the matching approach, this space encompasses not only the two dimensions of the image, but also other parameters like the possible range of scales and orientations or the question of how much of the object must be visible. The more you can restrict the search space, the faster the search will be.

The most obvious way to restrict the search space, which is suitable for all matching approaches, is to apply the operator that is used to find the model in an image to an ROI instead of the whole image.

Other parameters that can be used to restrict the search space depend on the specific matching approach. Some approaches are already very strict, others allow different changes of the object like rotation or scale and thus a restriction of the search space is strongly recommended. Shape-based matching, e.g., allows arbitrary orientation and scale changes as well as occlusions of the model instances in the search images. Thus, for shape-based matching also the range of orientation and scale as well as the visibility, i.e., the amount of allowed occlusions of the model instance in the image, should be restricted to speed up the search (see [section 3.2](#) on page 53).

2.4.2 Subsampling

The image pyramid (see [section 2.3](#) on page 25) can be used to speed-up the search. In order to benefit from the subsampling, the images should have as much contrast as possible.

The parameter `NumLevels`, which you already specified when creating the model, allows you to use a different (in most cases a more restrictive) value in the search process. By using the value 0 for `NumLevels`, the value specified when creating the model is used.

Optionally, `NumLevels` can contain a second value, so that you can specify not only the highest but also the lowest pyramid level used for the search. If the search is finished on a pyramid level that is higher than the first pyramid level, which corresponds to the original, full-sized image, the search becomes faster. On the other hand, the search is then also less robust and less accurate. If objects should be found also in images of poor quality, e.g., if the object is defocused, deformed, or noisy, you can activate the increased tolerance mode by specifying the second value negatively. Then, the matches on the lowest pyramid level that still provides matches are returned.

In case high noise is affecting the matching results you can also use `adapt_shape_model_high_noise` in order to estimate the lowest pyramid level for the shape model.

2.4.3 The Concept of Greediness

Greedy is a trade between thoroughness and speed. With the parameter `Greedy` you can influence the search algorithm itself and thereby trade thoroughness against speed. If you select the value 0, the search is thorough, i.e., if the object is present (and within the allowed search space and reaching the minimum score), it will be found. In this mode, however, even very unlikely match candidates are also examined thoroughly, thereby slowing down the matching process considerably.

The main idea behind the “greedy” search algorithm is to abort the comparison of a candidate with the model when it seems unlikely that the minimum score will be reached. In other words, the goal is not to waste time on hopeless candidates. This greediness, however, can have unwelcome consequences: In some cases a perfectly visible object is not found because the comparison “starts out on a wrong foot” and is therefore classified as a hopeless candidate and broken off.

You can adjust the `Greedy` of the search, i.e., how early the comparison is broken off, by selecting values between 0 (no break off: thorough but slow) and 1 (earliest break off: fast but unsafe). Note that the parameters `Greedy` and `MinScore` interact, i.e., you may have to specify a lower minimum score in order to use a greedier search. Generally, you can reach a higher speed with a high greediness and a sufficiently lowered minimum score.

2.4.4 Limit the Number of Candidates: `MinScore` and `NumMatches`

The search speed can be increased by limiting the number of candidates tracked through the pyramid levels. Both parameters `MinScore` and `NumMatches` have the intention to limit the number of tracked and returned candidates, but they do in different ways.

With `MinScore` you set a required score every candidate must have on every level of the pyramid. Thus, when a candidate gets a lower score on a higher pyramid level it gets rejected. As a consequence the candidate does not get tracked down to the lowest level, even if he would obtain the best score on the original image (the first pyramid level, see [section 2.3](#) on page 25).

With `NumMatches` you limit the number of instances to be returned to a maximum of the set number. This limit generally applies to every pyramid level (with exceptions as e.g., the top level (where `NumMatches` does not restrict the number of candidates to be considered) or shape-based matching with clutter set which is explained in [set_generic_shape_model_param](#)). The interplay with `MinScore` may lead to unwanted rejections.

Example: Assume our pyramid has three levels and we find 3 suitable candidates on the top pyramid level: A, B, and C. They are tracked to the second pyramid level, where they get the scores $S_A^2: 0.92$, $S_B^2: 0.82$, $S_C^2: 0.81$. On the lowest pyramid level, level 1, they get the score $S_A^1: 0.93$, $S_C^1: 0.79$, $S_C^1: 0.80$. What happens when we would have set for this search `MinScore = 0.8` and `NumMatches = 2`? On the top level we find 3 all candidates. We track them all to the second level, where we keep only the best 2 candidates A and B, while C gets rejected. The search continues on the next lower pyramid level. Tracking the candidates down, they get a new score on this lower level. So we reject B and end up with A, thus only 1 match. C would have got a score high enough but we rejected it already earlier. If we would have kept all 3 candidates from the top level, thus not reject C, we would have found 2 matches. This explains why the number of matches or even the found match can differ when setting `NumMatches`. Also the candidate with the highest score on a higher level is not necessarily the candidate with the highest score on the lowest level. For this reason the returned instance for e.g., `NumMatches` set to 1 can differ from the instance with the highest score returned when more matches are returned by setting `NumMatches` to a value higher 1.

As a consequence we recommend to return several instances and then select the instance with the highest score in case multiple objects with a similar score are expected, but only the one with the highest score is of interest.

2.5 Use the Results of Matching

The main goal of matching is to locate objects in images, i.e., above all get the position and in most cases also the orientation of a model instance in an image. Furthermore, many approaches return additional information like the scale of the object or a score that evaluates the quality of the returned object location. This chapter describes how to use the results returned by matching operators to locate the found instances in the image. The different matching approaches are suitable to different tasks and so they return different results. As a consequence the way how an instance can be located differs. The following sections

- list the individual results of the different matching approaches ([section 2.5.1](#)),

- introduce the different types of transformations often needed to further process the results of matching ([section 2.5.2](#)),
- show how to display matches and perform transformations using a 2D position and orientation ([section 2.5.3](#) on page 33),
- show how to visualize matches and perform transformations using a 2D homography ([section 2.5.4](#) on page 40),
- show how to visualize matches and perform transformations using a 3D pose ([section 2.5.5](#) on page 43), and
- shortly discuss the returned score ([section 2.5.6](#) on page 45).

2.5.1 Results of the Individual Matching Approaches

To locate objects in images, information like the position and orientation of the searched objects must be returned by the matching. This information is available in different representations, depending on the selected matching approach. That is, for a strict 2D matching the position and orientation are returned separately, i.e., the position consists of a row and a column value and the orientation consists of a single value describing the angle. In contrast, the uncalibrated perspective approaches return the position and orientation together in a projective transformation matrix (2D homography) and the calibrated perspective approaches return them together in a 3D pose. Besides the position and orientation, many approaches return further information like the scale of the found object or information about the quality of the match, which is called score.

The results of the individual matching approaches are listed below. How to further process them is described in the following sections.

Matching Approach	Operator(s) and Results
correlation-based matching	find_ncc_model : Position, rotation, and score of the found model.
	find_ncc_models : Positions, rotation angles, and scores for multiple models, and the information to which model each instance belongs.
shape-based matching	Matches found using find_generic_shape_model . • get_generic_shape_model_result: Position, rotation, scaling factors, and score of each found model instance. Additionally the transformation matrix and possible clutter values are returned. • get_generic_shape_model_result_object: Model contours transformed according to the matching result. Additionally possibly regions where no clutter should occur.
local deformable matching	find_local_deformable_model : Position, vector field, rectified image (part), contours of the deformed object, and score of the found model.
perspective deformable matching	find_planar_calib_deformable_model : 3D pose, six mean square deviations or the 36 covariances of the 6 pose parameters, and score.
	find_planar_uncalib_deformable_model : 2D homography and score.
descriptor-based matching	find_calib_descriptor_model : 3D pose and score.
	find_uncalib_descriptor_model : 2D homography and score.

2.5.2 About Transformations

HALCON provides operators for different types of transformations. For the strict 2D matching approaches, 2D affine transformations are available, which allow to, e.g., translate, rotate, or scale 2D objects (see [section 2.5.2.1](#)). For the uncalibrated perspective matching approaches, 2D projective transformations are provided, which are applied in the context of perspective views (see [section 2.5.2.2](#) on page 32). Finally, for the calibrated perspective




Figure 2.16: Typical affine transformations: a) translation along two axes; b) rotation around the IC center; c) scaling around the IC center; d) combining a, b, and c; e) rotation around the upper right corner; f) scaling around the right IC center.

matching approaches, 3D affine transformations are available. You can find more information about 3D transformations in the [Solution Guide III-C - 3D Vision](#).

2.5.2.1 2D Affine Transformations

“Affine transformation” is a technical term in mathematics describing a certain group of transformations. [Figure 2.16](#) shows the types that occur, e.g., in the context of the shape-based matching: An object can be *translated* (moved) along the two axes, *rotated*, and *scaled*. In [figure 2.16d](#), all three transformations were applied in a sequence.

Note that for the rotation and the scaling a fixed point exists, around which the transformation is performed. It corresponds to the point of reference described in [section 2.1.2.1](#) on page 18. In [figure 2.16b](#), e.g., the IC is rotated around its center, in [figure 2.16e](#) around its upper right corner. The point is called fixed point because it remains unchanged by the transformation.

The transformation can be thought of as a mathematical instruction that defines how to calculate the coordinates of object points after the transformation. Fortunately, you need not worry about the mathematical part. HALCON provides a set of operators that let you specify and apply transformations in a simple way.

HALCON allows to transform pixels, regions, images, and XLD contours by providing the operators [affine_trans_pixel](#), [affine_trans_region](#), [affine_trans_image](#), and [affine_trans_contour_xld](#). The transformation in [figure 2.16d](#) corresponds to the line

```
affine_trans_region (IC, TransformedIC, ScalingRotationTranslation, \
                    'nearest_neighbor')
```


The parameter `ScalingRotationTranslation` is a so-called homogeneous transformation matrix that describes the desired transformation. You can create this matrix by adding simple transformations step by step. First, an identity matrix is created with `hom_mat2d_identity`.

```
hom_mat2d_identity (EmptyTransformation)
```

Then, the scaling around the center of the IC is added with `hom_mat2d_scale`.

```
hom_mat2d_scale (EmptyTransformation, 0.5, 0.5, RowCenterIC, ColumnCenterIC, \
                Scaling)
```

Similarly, the rotation and the translation are added with `hom_mat2d_rotate` and `hom_mat2d_translate`.

```
hom_mat2d_rotate (Scaling, rad(90), RowCenterIC, ColumnCenterIC, \
                ScalingRotation)
hom_mat2d_translate (ScalingRotation, 100, 200, ScalingRotationTranslation)
```

Please note that in these operators the coordinate axes are labeled with `x` and `y` instead of `Row` and `Column`! [Figure 2.16a](#) clarifies the relation.

Transformation matrices can also be constructed by a sort of “reverse engineering”. In other words, if the result of the transformation is known for some points of the object, you can determine the corresponding transformation matrix. If, e.g., the position of the IC center and its orientation after the transformation is known, you can get the corresponding matrix via the operator `vector_angle_to_rigid` and then use this matrix to compute the transformed region.

```
vector_angle_to_rigid (RowCenterIC, ColumnCenterIC, 0, \
                    TransformedRowCenterIC, TransformedColumnCenterIC, \
                    rad(90), RotationTranslation)
affine_trans_region (IC, TransformedIC, RotationTranslation, \
                    'nearest_neighbor')
```

If a pixel, image, or contour should be transformed, the proceeding is similar, but instead of `affine_trans_region`, you apply the operator `affine_trans_pixel`, `affine_trans_image`, or `affine_trans_contour_xld`, respectively (see, e.g., [section 2.5.3.1](#) on page 34).

2.5.2.2 2D Projective Transformations

A 2D projective transformation matrix describes a perspective projection as illustrated in [figure 2.17](#). It consists of 3×3 values. Note that if the last row contains the values $[0,0,1]$, it corresponds to a homogeneous transformation matrix, i.e., it describes a 2D affine transformation, which is a special case of the 2D projective transformation.




Figure 2.17: Perspective projection.

HALCON provides several operators to apply projective transformations. Similar to the affine transformation, with a projective transformation you can transform different HALCON structures like pixels (`projective_trans_pixel`), regions (`projective_trans_region`), images (`projective_trans_image`), and XLD




Figure 2.18: The position and orientation of a match: a) The center of the ROI acts as the default point of reference; b) In the template image, the orientation is always 0.

contours (`projective_trans_contour_xld`). But now, also perspective deformations are considered by the transformation. An example for a 2D projective transformation is described in more detail in [section 2.5.4](#) on page 40.

Note that in the context of matching, projective transformation matrices are even easier to handle than affine transformations, as you do not need to create a transformation matrix from corresponding points or by adding several transformation steps. Instead, the projective transformation matrix is directly returned by one of the operators that are used for the uncalibrated matching of perspectively distorted objects, and can be directly used to apply one of the above stated transformations.

2.5.3 Display Matches and Perform Transformations Using a 2D Position and Orientation

The estimated position and orientation (2D pose) for a model instance of a strict 2D matching can be used in different ways. They can be used, e.g.,

- to display the found instance ([section 2.5.3.1](#) for a single match and [section 3.1.4.5](#) on page 51 for multiple matches),
- to align ROIs for other inspection tasks, e.g., measuring ([section 2.5.3.2](#) on page 35), or
- to transform the search image so that the object is positioned as in the template image ([section 2.5.3.3](#) on page 38).

For most matching approaches, the position and orientation returned by the matching is determined relative to the model. That is, no absolute position but the distance to the point of reference is returned. The point of reference is defined by the center of the ROI that was used to create the model (see [figure 2.18a](#)). By default, the point of reference for the model is moved to the coordinates (0,0). But even then, as different tasks in HALCON need different image coordinate systems (the positions for pixel-precise regions differ from those used for subpixel-precise XLD contours by 0.5 pixels, see [page 37](#) and the chapter “[Transformations > 2D Transformations](#)” in the Reference Manual), **the estimated position returned by the matching can not be used directly** but is optimized for the creation of the transformation matrices that are used to apply the applications described above. Additionally, in the template image the object is taken as not rotated, i.e., its angle is 0, even if it seems to be rotated as, e.g., in [figure 2.18b](#).



2.5.3.1 Display the Matches

In many cases, especially during the development of a matching application, it is useful to display the matching results in the search image. This can be realized by different means. If the **model is represented by a contour**, we recommend to overlay the XLD contour on the search image, because XLD contours can be transformed more precisely and quickly than regions. Additionally, by displaying the contour of the model not only the position and orientation of the found model instance but also the deviation from the model's shape can be visualized.

If **no contour is available**, we recommend to overlay the ROI that was used to create the model on the search image. In some cases, the additional visualization of points is suitable. For example, for the descriptor-based matching, the interest points of the model or a found model instance can be queried as described in [section 3.5.3](#) on page 98 and displayed using, e.g., `gen_cross_contour_xld`.

For shape-based matching the found instances and their transformation matrix can directly be retrieved using `get_generic_shape_model_result_object` and `get_generic_shape_model_result`, respectively. For correlation-based matching the procedure `dev_display_ncc_matching_results` can be used to visualize the results. In the following we show how the steps done within this procedure to overlay the XLD contour or the ROI on the search image.

The HDevelop program `%HALCONEXAMPLES%\hdevelop\Matching\Correlation-Based\ncc_matching_workflow.hdev` uses of a correlation-based matching. Once we have a successful match

Step 1: Access the XLD contour containing the model

The XLD version of the model is accessed.

```
get_ncc_model_region (ModelRegion, ModelID[Index])
gen_contour_region_xld (ModelRegion, ModelContours, 'border_holes')
```

Step 2: Determine the affine transformation

A visualization is only reasonable if the matching was successful. Thus, the results of the matching are checked. If the matching failed, the operator `find_ncc_model` returns empty tuples in parameters like `Row`. If the matching was successful, the corresponding affine transformation can be constructed.

```
hom_mat2d_identity (HomMat2DIdentity)
hom_mat2d_rotate (HomMat2DIdentity, Angle[Match], 0, 0, HomMat2DRotate)
hom_mat2d_translate (HomMat2DRotate, Row[Match], Column[Match], HomMat2DTranslate)
```

Note that the transformation matrix can also be constructed as shown in [section 2.5.3.2](#). Doing so, remember that the initial XLD contour of the model is located at the origin of the image and not at the position of the model in the reference image.

Step 3: Transform the XLD

Now, the transformation is applied to the XLD contour of the model using the operator `affine_trans_contour_xld` and the transformed contour is displayed. [Figure 3.1](#) on page 48 in [section 3.1.1](#) shows the search image and the overlaid XLD contour.

```
affine_trans_contour_xld (ModelContours, ContoursAffinTrans, HomMat2DTranslate)
dev_display (ContoursAffinTrans)
```

The general procedure for the projection of a contour or a region is the same.

The **main difference between the projection of a contour and the projection of a region** regards the initial position of the contour or region. The initial position of the contour or region always describes the “relative” position of the point of reference, i.e., its distance to the default point of reference. The default position of the point of reference is (0,0), i.e., it is the origin of the image and not the position of the model in the reference image. But depending on the structure that has to be transformed, the “relative” position can change. Whereas the **XLD contour** of a shape model is derived from the model and thus its point of reference is also the point of reference of the model, a **region** naturally was defined before the model was created. Thus, its point of reference describes the position of the model in the reference image instead of the point of reference of the later created model. That is, instead of (0,0), the values obtained by the operator `area_center` for the respective region have to be passed as the first two parameters to `vector_angle_to_rigid`.




Figure 2.19: Aligning ROIs for inspecting parts of a razor: a) ROIs for the model; b) the model; c) measuring ROIs; d) inspection results with zoomed faults.

Furthermore, if you have changed the reference point of a model (see, e.g., [section 2.1.2](#) on page 18), the values to insert as initial position change in the following way: If you query an XLD contour from a model after changing the point of reference, nothing changes. The initial position still is $(0,0)$, only another part of the model serves as point of reference. If you change the point of reference after querying the XLD contour or if you want to transform a region, which is naturally also accessed before changing the point of reference, you have to add the values of the initial point of reference to the values of the new point of reference.

2.5.3.2 Align ROIs for other Inspection Tasks

The results of matching can be used to align ROIs for other image processing steps. i.e., to position them relative to the image part acting as the model. This method is suitable, e.g., if the object to be inspected is allowed to move or if multiple instances of the object are to be inspected at once.

The HDevelop example program `%HALCONEXAMPLES%\solution_guide\matching\align_measurements.hdev`, e.g., uses shape-based matching to inspect multiple razor blades by measuring the width and the distance of their “teeth”.

First, a shape model is created that will be used later to align the measurement ROIs. [Figure 2.19a](#) shows the model ROI for the shape model, which consists of two united regions, and [figure 2.19b](#) shows the corresponding shape model.

Then, the inspection task is realized with the following steps:




Figure 2.20: Align the measure ROIs relative to the shape model: a) first, move the measure ROIs to lie on the XLD model; b) then, determine the new reference positions for the measure ROIs by their distance to the model.

Step 1: Position the measurement ROIs for the model blade

Two rectangular measurement ROIs are generated with `gen_rectangle2` so that they are placed over the teeth of the razor blade that acts as the model (see figure 2.19c).

```
Rect1Row := 244
Rect1Col := 73
DistColRect1Rect2 := 17
Rect2Row := Rect1Row
Rect2Col := Rect1Col + DistColRect1Rect2
RectPhi := rad(90)
RectLength1 := 122
RectLength2 := 2
gen_rectangle2 (MeasureROI1, Rect1Row, Rect1Col, RectPhi, RectLength1, \
               RectLength2)
gen_rectangle2 (MeasureROI2, Rect2Row, Rect2Col, RectPhi, RectLength1, \
               RectLength2)
```

To be able to transform them later along with the XLD contour of the model, they are moved with `move_region` to lie on the XLD model, whose point of reference is the origin of the image (see figure 2.20a), which is queried with `area_center`. Note that before moving the regions the clipping must be switched off, otherwise the region parts that are moved “outside” the image are clipped. The distances between the center of the model ROI and the centers of the rectangular measure ROIs define the new reference positions for the measure ROIs (see figure 2.20b).

```
area_center (ModelROI, Area, CenterROIRow, CenterROIColumn)
get_system ('clip_region', OriginalClipRegion)
set_system ('clip_region', 'false')
move_region (MeasureROI1, MeasureROI1Ref, -CenterROIRow, -CenterROIColumn)
move_region (MeasureROI2, MeasureROI2Ref, -CenterROIRow, -CenterROIColumn)
set_system ('clip_region', OriginalClipRegion)
DistRect1CenterRow := Rect1Row - CenterROIRow
DistRect1CenterCol := Rect1Col - CenterROIColumn
DistRect2CenterRow := Rect2Row - CenterROIRow
DistRect2CenterCol := Rect2Col - CenterROIColumn
```

Step 2: Find all razor blades

Now, all instances of the shape model are searched for in the search image using `find_generic_shape_model`.

```
find_generic_shape_model (SearchImage, ModelID, MatchResultID, \
                          NumMatchResult)
```

Step 3: Determine the affine transformation

For each instance, i.e., for each found razor blade, the transformation between the model and the found model instance is calculated with `vector_angle_to_rigid` and applied to the XLD model with `affine_trans_contour_xld` so that it lies over the found model instance in the search image.

```
for I := 0 to NumMatchResult - 1 by 1
  get_generic_shape_model_result (MatchResultID, Indices[I], 'row', \
                                RowCheck)
  get_generic_shape_model_result (MatchResultID, Indices[I], 'column', \
                                ColumnCheck)
  get_generic_shape_model_result (MatchResultID, Indices[I], 'angle', \
                                AngleCheck)
  vector_angle_to_rigid (0, 0, 0, RowCheck, ColumnCheck, AngleCheck, \
                        MovementOfObject)
  affine_trans_contour_xld (ModelContours, ModelAtNewPosition, \
                          MovementOfObject)
```

Step 4: Create measurement objects at the corresponding positions

Then, `affine_trans_pixel` is applied to calculate the corresponding positions of the measure ROIs, at which the measure objects are created.

```
affine_trans_pixel (MovementOfObject, DistRect1CenterRow, \
                  DistRect1CenterCol, Rect1RowCheck, Rect1ColCheck)
affine_trans_pixel (MovementOfObject, DistRect2CenterRow, \
                  DistRect2CenterCol, Rect2RowCheck, Rect2ColCheck)
RectPhiCheck := RectPhi + AngleCheck
gen_measure_rectangle2 (Rect1RowCheck, Rect1ColCheck, RectPhiCheck, \
                      RectLength1, RectLength2, Width, Height, \
                      'bilinear', MeasureHandle1)
gen_measure_rectangle2 (Rect2RowCheck, Rect2ColCheck, RectPhiCheck, \
                      RectLength1, RectLength2, Width, Height, \
                      'bilinear', MeasureHandle2)
```

Note that **you must use the operator `affine_trans_pixel` and not `affine_trans_point_2d`**, because the latter uses a different coordinate system. In particular, it uses the standard image coordinate system for which a position corresponds to the center of a pixel (see [figure 2.21](#), right). In contrast, the operators `affine_trans_pixel`, `affine_trans_contour_xld`, `affine_trans_region`, and `affine_trans_image` use the coordinate system depicted in [figure 2.21](#) (left).



In the example application, the individual razor blades are only translated but not rotated relative to the model position. Thus, instead of applying the full affine transformation to the measure ROIs and then creating new measure objects, you can use the operator `translate_measure` to translate the measure objects themselves. The example program contains the corresponding code. You can switch between the two methods by modifying a variable at the top of the program.

Step 5: Measure the width and the distance of the “teeth”

Now, the actual measurements are performed using the operator `measure_pairs`.

```
measure_pairs (SearchImage, MeasureHandle1, 2, 25, 'negative', 'all', \
              RowEdge11, ColEdge11, Amp11, RowEdge21, ColEdge21, \
              Amp21, Width1, Distance1)
measure_pairs (SearchImage, MeasureHandle2, 2, 25, 'negative', 'all', \
              RowEdge12, ColEdge12, Amp12, RowEdge22, ColEdge22, \
              Amp22, Width2, Distance2)
```




Figure 2.21: (left) image coordinate system used for the matching methods and operators like `affine_trans_pixel`; (right) standard image coordinate system.

Step 6: Inspect the measurements

Finally, the measurements are inspected. If a tooth is too short or missing completely, no edges are extracted at this point resulting in an incorrect number of extracted edge pairs. In this case, the faulty position can be determined by checking the distance of the teeth. Figure 2.19d shows the inspection results for the example.

```
NumberTeeth1 := |Width1|
if (NumberTeeth1 < 37)
  for J := 0 to NumberTeeth1 - 2 by 1
    if (Distance1[J] > 4.0)
      RowFault := round(0.5 * (RowEdge11[J + 1] + RowEdge21[J]))
      ColFault := round(0.5 * (ColEdge11[J + 1] + ColEdge21[J]))
      disp_rectangle2 (WindowHandle, RowFault, ColFault, 0, 4, 4)
```

Please note that the example program is not able to display the fault if it occurs at the first or the last tooth.

2.5.3.3 Align the Search Results

The previous sections showed how to use the matching results to determine the so-called forward transformation. This was used to transform objects from the model into the search image or, respectively, to transform ROIs that were specified in the reference image into the search image.

You can also determine the inverse transformation which transforms objects from the search image back into the reference image. With this transformation, you can align the search image (or parts of it), i.e., transform it such that the matched object is positioned as it was in the reference image. This method is useful if the following image processing step is not invariant against rotation, e.g., OCR or the variation model.

Note that by alignment the image is only rotated and translated. To remove perspective or lens distortions, e.g., if the camera observes the scene under an oblique angle, you must rectify the image first (see section 3.2.12 on page 77 for more information).

Inverse Transformation

The inverse transformation can be determined and applied in a few steps. The task of the HDevelop program `%HAL-CONEXAMPLES%\solution_guide\matching\rectify_results.hdev`, e.g., is to extract the serial number on CD covers (see figure 2.22). The matching is realized using a shape-based matching.

Step 1: Calculate the inverse transformation

As a first step the transformation matrix is calculated. This is done as described in section 2.5.2.1 on page 31, with the difference that it is calculated based on the absolute coordinates of the point of reference, because the results have to be transformed such that they appear as in the reference image.


```

get_generic_shape_model_result (MatchResultID, 'best', 'row', \
                                RowMatch)
get_generic_shape_model_result (MatchResultID, 'best', 'column', \
                                ColumnMatch)
get_generic_shape_model_result (MatchResultID, 'best', 'angle', \
                                AngleMatch)
vector_angle_to_rigid (CenterModelROIRow, CenterModelROIColumn, 0, \
                      RowMatch, ColumnMatch, AngleMatch, \
                      MovementOfObject)

```

Note, this transformation matrix can also be created in different ways, e.g., by operating the model transformation matrix on the original center of RoI.

You can invert a transformation easily using the operator `hom_mat2d_invert`.

```
hom_mat2d_invert (MovementOfObject, InverseMovementOfObject)
```

Step 2: Rectify the search image

The inverse transformation is applied to the search image using the operator `affine_trans_image`. Figure 2.22d shows the resulting rectified image of a different CD. Undefined pixels are marked in gray.

```
affine_trans_image (SearchImage, RectifiedSearchImage, \
                  InverseMovementOfObject, 'constant', 'false')
```

Step 3: Extract the numbers

The serial number is positioned correctly within the original ROI and can be extracted by blob analysis without problems. Figure 2.22e shows the result, which could then, e.g., be used as the input for OCR.

```

reduce_domain (RectifiedSearchImage, NumberROI, RectifiedNumberROIImage)
threshold (RectifiedNumberROIImage, Numbers, 0, 128)
connection (Numbers, IndividualNumbers)

```

Unfortunately, the operator `affine_trans_image` transforms the full image even if its domain was restricted with the operator `reduce_domain`. In a time-critical application it may therefore be necessary to crop the search image before transforming it.

Image Cropping

In the following, the steps needed for the image cropping are described. Additionally, they are visualized in figure 2.23.

Step 1: Crop the search image

First, the smallest axis-parallel rectangle surrounding the transformed number ROI is computed using the operator `smallest_rectangle1`. Then, the search image is cropped to this part with `crop_rectangle1`. Figure 2.23b shows the resulting image overlaid on a gray rectangle to facilitate the comparison with the subsequent images.

```

affine_trans_region (NumberROI, NumberROIAtNewPosition, \
                    MovementOfObject, 'nearest_neighbor')
smallest_rectangle1 (NumberROIAtNewPosition, RowRect1, ColumnRect1, \
                    RowRect2, ColumnRect2)
crop_rectangle1 (SearchImage, CroppedSearchImage, RowRect1, ColumnRect1, \
                RowRect2, ColumnRect2)

```

Step 2: Create an extended affine transformation

In fact, the cropping can be interpreted as an additional affine transformation, in particular, as a translation by the negated coordinates of the upper left corner of the cropping rectangle (see figure 2.23a). We therefore add this transformation to the transformation that describes the movement of the object using the operator `hom_mat2d_translate`. Then, we invert this extended transformation with the operator `hom_mat2d_invert`.




Figure 2.22: Rectify the search results: a) ROIs for the model and for the number extraction; b) the model; c) found model and number ROI at matched position; d) rectified search image (only relevant part shown); e) extracted numbers.

```
hom_mat2d_translate (MovementOfObject, -RowRect1, -ColumnRect1, \
                    MoveAndCrop)
hom_mat2d_invert (MoveAndCrop, InverseMoveAndCrop)
```

Step 3: Transform the cropped image

Using the inverted extended transformation, the cropped image can easily be rectified with the operator `affine_trans_image` (figure 2.23c) and then be reduced to the original number ROI (figure 2.23d) in order to extract the numbers.

```
affine_trans_image (CroppedSearchImage, RectifiedROIImage, \
                  InverseMoveAndCrop, 'constant', 'true')
reduce_domain (RectifiedROIImage, NumberROI, RectifiedNumberROIImage)
```

2.5.4 Visualize Matches and Perform Transformations Using a 2D Homography

When applying one of the uncalibrated perspective matching approaches, in contrast to the pure 2D matching approaches no positions, orientations, and scales but projective transformation matrices are returned. These can be used to apply a projective transformation, e.g., to visualize the matching result by overlaying a structure of the




Figure 2.23: Rectifying only a part of the search image: a) smallest image part containing the ROI; b) cropped search image; c) result of the rectification; d) rectified image reduced to the original number ROI.

reference image on the match in the search image. Note that different HALCON structures like pixels, regions, images, or XLD contours can be transformed that way (see also [section 2.5.2.2](#) on page 32).

Depending on the selected matching approach, typically different structures are transformed. For the uncalibrated perspective deformable matching, e.g., the model contour is transformed to visualize the match, whereas for the uncalibrated descriptor-based matching the region that was used to create the model may be transformed and visualized.

An example for a projective transformation in the context of an uncalibrated **perspective deformable matching** is given in the HDevelop example program `%HALCONEXAMPLES%\hdevelop\Applications\Traffic-Monitoring\detect_road_signs.hdev` (see [figure 2.24](#)).

Step 1: Find the perspective deformable model

There, the model of a road sign is searched in an image using `find_planar_uncalib_deformable_model`, which returns the projective transformation matrix `HomMat2D`.

```
find_planar_uncalib_deformable_model (ImageChannel, Models[Index2], 0, \
                                     0, ScaleRMin[Index2], \
                                     ScaleRMax[Index2], \
                                     ScaleCMin[Index2], \
                                     ScaleCMax[Index2], 0.85, 1, 0, 2, \
                                     0.4, [], [], HomMat2D, Score)
```

Step 2: Transform the model contours

After the matching, the model contours are queried from the model with `get_deformable_model_contours`. The returned contours are by default positioned at the origin of the reference image. Thus, to visualize them at the




Figure 2.24: The contours of the model are projected into the search image.

position of the match, the contour must be transformed using the returned projective transformation matrix. This transformation is applied with `projective_trans_contour_xld`.

```
get_deformable_model_contours (ModelContours, Models[Index2], 1)
projective_trans_contour_xld (ModelContours, ContoursProjTrans, \
                             HomMat2D)
dev_display (ContoursProjTrans)
```

An example for a projective transformation in the context of an uncalibrated **descriptor-based matching** is the HDevelop example program `%HALCONEXAMPLES%\hdevelop\Applications\Object-Recognition-2D\detect_brochure_pages.hdev` (see [figure 2.25](#)).




Figure 2.25: (left) Model brochure page with ROI and interest points; (right) search image with projected interest points and ROI.

Step 1: Find the descriptor model

There, the projective transformation matrix `HomMat2D` is returned by `find_uncalib_descriptor_model`.

```
find_uncalib_descriptor_model (ImageGray, ModelIDs[Index2], \
                              'threshold', 600, \
                              ['min_score_descr', \
                               'guided_matching'], [0.003, 'on'], \
                              0.25, 1, 'num_points', HomMat2D, \
                              Score)
```

Step 2: Access and visualize the interest points

The interest points are queried with `get_descriptor_model_points`. In contrast to the contours of a perspective deformable model, which can only be queried for the model, the points of the descriptor model can also be queried for the specific match (`Set` set to 'search'). Thus, they do not have to be transformed anymore.

```
get_descriptor_model_points (ModelIDs[Index2], 'search', 0, Row, \
                             Col)
gen_cross_contour_xld (Cross, Row, Col, 6, 0.785398)
```

Step 3: Transform the model region and its corner points

Projective transformations can be used if, e.g., the transformed region that was used for the creation of the model should be visualized for the match as well or if the transformed corner points of the region are needed to check the angle between edges of the region.

```
projective_trans_region (Rectangle, TransRegion, HomMat2D, \
                        'bilinear')
projective_trans_pixel (HomMat2D, RowRoi, ColRoi, RowTrans, \
                        ColTrans)
angle_ll (RowTrans[2], ColTrans[2], RowTrans[1], ColTrans[1], \
          RowTrans[1], ColTrans[1], RowTrans[0], ColTrans[0], \
          Angle)
Angle := deg(Angle)
if (Angle > 70 and Angle < 110)
    dev_display (TransRegion)
    dev_display (Cross)
endif
```

2.5.5 Visualize Matches and Perform Transformations Using a 3D Pose

When applying one of the calibrated perspective matching approaches, 3D poses are returned that describe the relation between the model and the found model instance in world coordinates. To use such a pose, e.g., to visualize the matching result by overlaying a structure of the reference image on the match in the search image, a 3D affine transformation is needed (for more information to 3D affine transformations see also [Solution Guide III-C - 3D Vision](#)).

Depending on the selected matching approach, different structures from the reference image may be needed in the search image. For the calibrated perspective, deformable matching typically the contour of the model is used to visualize the match. That is, the contour must be transformed from the reference image to the search image. For the calibrated descriptor-based matching, the interest points of the model can be queried directly for the search result, i.e., there is no need for a transformation of the model representation. However, a transformation may be needed if the region that was used to create the model should be transformed as well.

An example for a 3D affine transformation in the context of a calibrated **perspective deformable matching** is given in the HDevelop example program `%HALCONEXAMPLES%\hdevelop\Applications\Position-Recognition-3D\locate_car_door.hdev` (see [figure 2.26](#)). There, a part of a car door is searched in an image using `find_planar_calib_deformable_model`, which returns the 3D pose `Pose`.

```
find_planar_calib_deformable_model (ImageReducedSearch, ModelID, -0.1, \
                                    0.2, 0.6, 1.0, 0.6, 1.0, 0.7, 1, 1, \
                                    0, 0.7, [], [], Pose, CovPose, \
                                    Score)
```





Figure 2.26: The contours of the model are projected into the search image.

For the visualization of the match, the model contour should be overlaid on the match in the search image. This transformation is realized in three steps.

Step 1: Preprocessing for visualization purposes

First, the operator `set_deformable_model_param` is used to set the coordinate system in which the contours are returned to 'world'. When calling the operator `get_deformable_model_contours` the contours are returned in the world coordinate system (WCS). In this example, this step was realized before the actual matching.

```
set_deformable_model_param (ModelID, \
                             'get_deformable_model_contours_coord_system', \
                             'world')
get_deformable_model_contours (ModelContours, ModelID, 1)
```

Step 2: Apply a 3D affine transformation to the points of the contour

Within the WCS, a 3D affine transformation transforms the individual points of the contour according to the 3D pose that is returned by the matching. Before applying this transformation, the 3D pose must be converted with `pose_to_hom_mat3d` into a 3D homogeneous transformation matrix `HomMat3D`.

```
for Index1 := 0 to |Score| - 1 by 1
  tuple_select_range (Pose, Index1 * 7, ((Index1 + 1) * 7) - 1, \
                      PoseSelected)
  pose_to_hom_mat3d (PoseSelected, HomMat3D)
```

Additionally, as 3D affine transformations cannot be applied to 2D contours, the contour, which is now available in the x-y-plane of the WCS, must be split into points. This is realized by the operator `get_contour_xld`, which returns a tuple for the x coordinates and a tuple for the y coordinates. To obtain 3D coordinates, a tuple for the z coordinates with the same number of elements is created, for which all values are 0.

```
gen_empty_obj (FoundContour)
for Index2 := 1 to NumberContour by 1
  select_obj (ModelContours, ObjectSelected, Index2)
  get_contour_xld (ObjectSelected, Y, X)
  Z := gen_tuple_const(|X|, 0.0)
```

The 3D points are now transformed by the 3D affine transformation using `affine_trans_point_3d`.

```
affine_trans_point_3d (HomMat3D, X, Y, Z, Xc, Yc, Zc)
```

Step 3: Project the points into the search image and reconstruct the contour

Then, the transformed 3D points are projected into the search image using `project_3d_point` and the contour is reconstructed with `gen_contour_polygon_xld`.

```
project_3d_point (Xc, Yc, Zc, CamParam, R, C)
gen_contour_polygon_xld (ModelWorld, R, C)
concat_obj (FoundContour, ModelWorld, FoundContour)
endfor
dev_display (FoundContour)
endfor
```

An example for a 3D affine transformation in the context of a calibrated **descriptor-based matching** is given in the HDevelop example program `%HALCONEXAMPLES%\hdevelop\Applications\Object-Recognition-2D\locate_cookie_box.hdev` that is described in more detail in [section 3.5.1](#) on page 96. There, selected image points, in particular the corner points of the rectangular ROI, which was used to build the descriptor model, are transformed into the WCS using `image_points_to_world_plane`.

2.5.6 About the Score

The scores returned by the different matching approaches have different meanings.

For the correlation-based matching, the score is determined by a normalized cross correlation between a pattern and the image.

For the approaches that use contours, in particular for the shape-based matching and the local or perspective deformable matching the score is a number between 0 and 1 that approximately shows how much of the model is visible in the search image. That is, if half of a model is occluded, the score can not be larger than 0.5. Note that besides the pure number of corresponding contour points further influences on the score exist, which comprise, e.g., the orientation of the contour (see also [section 3.2.6.2](#) on page 65).

For the descriptor-based matching different meanings for the score value(s) are available, depending on the selected score type (see [section 3.5.3.2](#) on page 99).

Chapter 3

The Individual Approaches

Now, we go deeper into the individual matching approaches, in particular, for

- the correlation-based matching ([section 3.1](#)),
- the shape-based matching ([section 3.2](#) on page 53),
- the local deformable matching ([section 3.3](#) on page 77),
- the perspective deformable matching ([section 3.4](#) on page 87), and
- the descriptor-based matching ([section 3.5](#) on page 95).

3.1 Correlation-Based Matching

Correlation-based matching is based on gray values. This approach uses a normalized cross correlation to evaluate the correspondence between a model and a search image. It is quite fast and can compensate both additive as well as multiplicative variations in illumination. In contrast to the shape-based matching, also objects with slightly changing shapes, lots of texture, or objects in blurred images (contours vanish in blurred images, e.g., because of defocus) can be found.

The following sections show

- a first example for a correlation-based matching ([section 3.1.1](#)),
- how to select an appropriate ROI to derive the template image from the reference image ([section 3.1.2](#)),
- how to create a suitable model ([section 3.1.3](#)), and
- how to optimize the search ([section 3.1.4](#) on page 50).

3.1.1 A First Example

In this section we give a quick overview of the matching workflow with correlation-based matching. To follow the example actively, start the HDevelop program %HALCONEXAMPLES%\hdevelop\Matching\Correlation-Based\ncc_matching_workflow.hdev.

Step 1: Select the object in the reference image

First, inside the training image a region containing the object is created using [gen_rectangle2](#).

```
read_image (Image, 'face_masks/face_mask_01')
gen_rectangle2 (ROI, 616.5, 708.5, rad(-82.4054), 50, 35)
reduce_domain (Image, ROI, ImageReduced)
```




Figure 3.1: (left) reference image with the ROI that is used to create the model; (right) match of a model instance.

Step 2: Create the model

The reduced image is used to create the NCC model with `create_ncc_model`. As a result, the operator returns a handle for the newly created model (`ModelID`), which can then be used to specify the model, e.g., in calls to the operators `find_ncc_model` or `find_ncc_models`.

```
create_ncc_model (ImageReduced, 'auto', rad(0), rad(360), 'auto', \
                 'use_polarity', ModelID)
```

Step 3: Find the object again

Now, the search images are read in a loop and for each search image the NCC model is searched. The found matches can conveniently be displayed using the procedure `dev_display_ncc_matching_results`. To visualize the match, latter one is overlaid by the region of the model using an affine transformation as described in [section 2.5.2.1](#) on page 31 (note, this is done within the visualization procedure). [Figure 3.1](#) shows the reference image and one of the found model instances.

```
for Index := 1 to NumImages by 1
  read_image (Image, 'face_masks/face_mask_' + Index$'02')
  find_ncc_model (Image, ModelID, rad(0), rad(360), 0.7, 1, 0.5, 'true', \
                 0, Row, Column, Angle, Score)
  dev_display_ncc_matching_results (ModelID, Color, Row, Column, Angle, 0)
endfor
```

3.1.2 Select the Model ROI

As a first step of the correlation-based matching, the region of interest that specifies the template image must be selected as described, e.g., in [section 2.1.1](#) on page 17. This region can have an arbitrary shape, i.e., it can also contain holes or consist of several parts that are not connected. Thus, it is possible to create an ROI that contains only “robust” parts of the object. In most cases, the ROI must be selected so that it contains also some pixels outside of the object of interest, as the immediate surroundings (neighborhood) of the object are needed to obtain the model. But if the object itself contains enough structure to be recognized independently from its outline, it can also be selected smaller than the object. Then, the object can be found also in front of different backgrounds. Note that you can speed up the later search using a subsampling (see [section 3.1.3.1](#)). For that, the ROI should not be too “thin”, because otherwise it vanishes at higher pyramid levels! As a rule of thumb, you are on the safe side if an ROI is $2^{NumLevels-1}$ pixels wide. That is, a width of 8 pixels allows to use 4 pyramid levels. After having selected a suitable ROI, the reduced image is used as template image for the creation of the model.

3.1.3 Create a Suitable NCC Model

Having derived the template image from the reference image, the NCC model can be created with the operator `create_ncc_model`. Here, we will take a closer look at how to adjust the corresponding parameters. In particular,

you can

- use a subsampling to speed up the search by adjusting the parameter `NumLevels` (section 3.1.3.1),
- allow a specific range of orientation by adjusting the parameters `AngleExtent`, `AngleStart`, and `AngleStep` (section 3.1.3.2), and
- specify which pixels are compared with the model in the later search, i.e., specify the polarity of the object by adjusting the parameter `Metric` (section 3.1.3.3).

For the parameters `NumLevels` and `AngleStep`, you can let HALCON suggest values automatically. This can be done either by setting the parameters within `create_ncc_model` to the value 'auto'; then, if you need to know the values, you can query them using `get_ncc_model_params`. Or you apply `determine_ncc_model_params` before you create the model. Then, you get an estimation of the automatically determined values as suggestion so that you can still modify them for the actual creation of the model. Note that both approaches return only approximately the same values and the values returned by `create_ncc_model` are more precise.

Note that after the creation of the model, the model can still be modified. In section 3.1.3.4 the possibilities for the inspection and modification of an already created model are shown.

3.1.3.1 Use Subsampling to Speed Up the Search (`NumLevels`)

To speed up the matching process, subsampling can be used (see also section 2.4.2 on page 28). There, an image pyramid is created, consisting of the original, full-sized image and a set of downsampled images. The model is then created and searched on the different pyramid levels. You can specify how many pyramid levels are used via the parameter `NumLevels`. We recommend to let HALCON select a suitable value itself by specifying the value 'auto'. You can then query the used value via the operator `get_ncc_model_params`.

i

3.1.3.2 Allow a Range of Orientation (`AngleExtent`, `AngleStart`, `AngleStep`)

If the object's rotation may vary in the search images you can specify the allowed range in the parameter `AngleExtent` and the starting angle of this range in the parameter `AngleStart` (unit: radians). We recommend to limit the allowed range of rotation as much as possible in order to speed up the search process. During the matching process, the model is searched for in different angles within the allowed range, at steps specified with the parameter `AngleStep`. If you select the value 'auto', HALCON automatically chooses an optimal step size to obtain the highest possible accuracy by determining the smallest rotation that is still discernible in the image. In section 3.2.6.1 on page 65 tips for the selection of values for all three parameters are given for shape-based matching. These tips are also valid for correlation-based matching.

3.1.3.3 Specify how Gray Values are Compared with the Model (`Metric`)

The parameter `Metric` lets you specify whether the polarity, i.e., the “direction” of the contrast must be observed. If you select the value 'use_polarity' the polarity is observed, i.e., the points in the search image must show the same direction of the contrast as the corresponding points in the model. If, e.g., the model is a bright object on a dark background, the object is found in the search images only if it is also brighter than the background. You can choose to ignore the polarity globally by selecting the value 'ignore_global_polarity'. In this mode, an object is recognized also if the direction of its contrast reverses, e.g., if your object can appear both as a dark shape on a light background and vice versa. This flexibility, however, is obtained at the cost of a slightly lower recognition speed and reduced robustness.

3.1.3.4 Inspect and Modify the NCC Model

To inspect the current parameter values of the model, you query them with `get_ncc_model_params`. This may be necessary if during the creation of the model an automatic parameter selection was used or if the model was created within another program, saved to file with `write_ncc_model`, and read from this file in the current program with `read_ncc_model`. Additionally, you can query the coordinates of the origin of the model using `get_ncc_model_origin`.

After the creation of the model and before you search the object in a search image, you can further modify the model. In particular, you can apply `set_ncc_model_param` to change individual parameters and `set_ncc_model_origin` to change the origin of the model. The latter is not recommended because the accuracy of the matching result may decrease, which is shown in more detail for shape-based matching in [section 3.2.6.6](#) on page 69.

3.1.4 Optimize the Search Process

The actual matching is performed by the operators `find_ncc_model` or `find_ncc_models`. In the following, we show how to select suitable parameters for these operators to adapt and optimize a matching task. In particular, we show how to

- restrict the search space to a region of interest ([section 3.1.4.1](#)),
- restrict the search space by restricting the range of orientation via the parameters `AngleStart` and `AngleExtent` ([section 3.1.4.2](#)),
- restrict the search space to a specific amount of deviations from the model for the object, i.e., specify the similarity of the object via the parameter `MinScore` ([section 3.1.4.3](#)),
- search for multiple instances of the model by adjusting the parameters `NumMatches` and `MaxOverlap` ([section 3.1.4.4](#)),
- search for multiple models with just one operator call ([section 3.1.4.5](#)),
- specify the accuracy that is needed for the results by adjusting the parameter `SubPixel` ([section 3.1.4.6](#) on page 53), and
- restrict the number of pyramid levels (`NumLevels`) for the search process ([section 3.1.4.7](#) on page 53).

3.1.4.1 Restrict the Search to a Region of Interest

The obvious way to restrict the search space and thus speed up the matching is to apply the operator `find_ncc_model` not to the whole image but only to an ROI. The corresponding procedure is explained in more detail for shape-based matching in [section 3.2.7](#) on page 70. For correlation-based matching you simply have to use `find_ncc_model`.

3.1.4.2 Restrict the Range of Orientation (`AngleStart`, `AngleExtent`)

When creating the model you already specified the allowed range of orientation (see [section 3.1.3.2](#) on page 49). When calling the operator `find_ncc_model` you can further **limit** the range with the parameters `AngleStart` and `AngleExtent`. This is useful if you can restrict these ranges by other information, which can, e.g., be obtained by suitable image processing operations. Another reason for using a larger range when creating the model may be that you want to reuse the model for other matching tasks.

3.1.4.3 Specify the Similarity of the Object (`MinScore`)

The parameter `MinScore` specifies the minimum score a potential match must have to be returned as match. The score is a value for the quality of a match, i.e., for the correspondence, or “similarity”, between the model and the search image. For correlation-based matching the score is obtained using the normalized cross correlation between the pattern and the image (for the formula, see the description of `find_ncc_model` in the Reference Manual). To speed up the search, the value of `MinScore` should be chosen as large as possible, but of course still as small as necessary for the success of the search.

3.1.4.4 Search for Multiple Instances of the Object (NumMatches, MaxOverlap)

To find multiple instances of a model in the search image, the maximum number of returned matches is selected via the parameter `NumMatches`. Additionally, the parameter `MaxOverlap` specifies to which degree the instances may overlap. The functional principle of these parameters is similar to the parameters that are used for shape-based matching (see [section 3.2.6.4](#) on page 66). There, tips for the parameter selection are given that are valid also for correlation-based matching. Note that if multiple instances of the object are searched and found, the parameters `Row`, `Column`, `Angle`, and `Score` contain tuples. How to access these results is exemplarily shown for multiple models in [section 3.1.4.5](#).

3.1.4.5 Search for Multiple Models Simultaneously (ModelIDs)

If you are searching for instances of multiple models in a single image, you can of course call the operator `find_ncc_model` multiple times. But a faster alternative is to use the operator `find_ncc_models`. This operator expects similar parameters, with the following differences:

- With the parameter `ModelIDs` you can specify a *tuple* of model IDs instead of a single one. As when searching for multiple instances (see [section 3.1.4.4](#)), the matching result parameters `Row` etc. return tuples of values.
- The output parameter `Model` shows to which model each found instance belongs. Note that the parameter does not return the model IDs themselves but the index of the model ID in the tuple `ModelIDs` (starting with 0).
- The search is always performed in a single image. However, you can restrict the search to a certain region for each model individually by passing an image array.
- You can either use the same search parameters for each model by specifying single values for `AngleStart` etc., or pass a tuple containing individual values for each model.
- You can also search for multiple instances of multiple models. If you search for a certain number of objects independent of their type (model ID), specify this (single) value in the parameter `NumMatches`. By passing a tuple of values, you can specify for each model individually how many instances are to be found. The tuple `[3, 0]`, e.g., specifies to return the best three instances of the first model and all instances of the second model. See also the example used below.

Similarly, if you specify a single value for `MaxOverlap`, the operators check whether a found instance is overlapped by any of the other instances independent of their type. By specifying a tuple of values, each instance is only checked against all other instances of the same type.

We illustrate this using an example in which we search for two different text parts punched on a fabric. They are illustrated in [figure 3.2](#).




Figure 3.2: Different model regions to be found using ncc matching.

Step 1: Create the models

The models are created the usual way.

```

create_ncc_model (ImageReducedCE, 'auto', rad(0), rad(360), 'auto', \
                  'use_polarity', ModelIDCE)
create_ncc_model (ImageReducedBrand, 'auto', rad(0), rad(360), 'auto', \
                  'use_polarity', ModelIDBrand)
ModelIDs := [ModelIDCE, ModelIDBrand]

```

When searching for multiple models simultaneously, it may be useful to store the information about the models, i.e., the XLD models, in tuples.

Step 2: Access the model XLD

The XLD contours corresponding to the two models are accessed with the operator `get_ncc_model_region`.

```

get_ncc_model_region (ModelRegionCE, ModelIDCE)
gen_contour_region_xld (ModelRegionCE, ModelContoursCE, 'border_holes')
get_ncc_model_region (ModelRegionBrand, ModelIDBrand)
gen_contour_region_xld (ModelRegionBrand, ModelContoursBrand, \
                        'border_holes')

```

Step 3: Save the information about the models in tuples

To facilitate the access to the shape models later, the XLD contours are saved in tuples in analogy to the model IDs. However, when concatenating XLD contours with the operator `concat_obj`, one must keep in mind that XLD objects are already tuples as they may consist of multiple contours! To access the contours belonging to a certain model, you therefore need the number of contours of a model and the starting index in the concatenated tuple. The former is determined using the operator `count_obj`. The contours of the ring start with the index 1, the contours of the nut with the index 1 plus the number of contours of the ring.

```

count_obj (ModelContoursCE, NumContoursCE)
count_obj (ModelContoursBrand, NumContoursBrand)
concat_obj (ModelContoursCE, ModelContoursBrand, ShapeModels)
StartContoursInTuple := [1, NumContoursCE + 1]
NumContoursInTuple := [NumContoursCE, NumContoursBrand]

```

Step 4: Search the model instances

In every image there is only a single face mask and hence for each model only a single instance is to be detected. As a consequence we set `NumMatches` to `[1,1]`, as explained above.

```

find_ncc_models (Image, ModelIDs, rad(0), rad(360), [0.5, 0.43], [1,1], \
                 0.5, 'true', 0, Row, Column, Angle, Score, Model)

```

Step 5: Access the found instances

As described in [section 3.1.4.4](#) on page 51, in case of multiple matches the output parameters `Row` etc. contain tuples of values, which are typically accessed in a loop, using the loop variable as tuple index. When searching for multiple models, a second index is involved: The output parameter `Model` indicates to which model a match belongs. This is done by storing the index of the corresponding model ID within the tuple of IDs specified in the parameter `ModelIDs`. This may sound confusing, but can be realized in an elegant way in the code: For each found instance, the model ID index is used to select the corresponding information from the tuples created above. As already noted, the XLD representing the model can consist of multiple contours. Therefore, you cannot access them directly using the operator `select_obj`. Instead, the contours belonging to the model are selected via the operator `copy_obj`, specifying the start index of the model in the concatenated tuple and the number of contours as parameters. Note that `copy_obj` does not copy the contours, but only the corresponding HALCON objects, which can be thought of as references to the contours.


```

NumMatches := |Score|
for Match := 0 to NumMatches - 1 by 1
  ModelIndex := Model[Match]
  hom_mat2d_identity (HomMat2DIdentity)
  hom_mat2d_rotate (HomMat2DIdentity, Angle[Match], 0, 0, \
    HomMat2DRotate)
  hom_mat2d_translate (HomMat2DRotate, Row[Match], Column[Match], \
    MoveAndScalingOfObject)
  *
  copy_obj (ShapeModels, ShapeModel, StartContoursInTuple[ModelIndex], \
    \
    NumContoursInTuple[ModelIndex])
  affine_trans_contour_xld (ShapeModel, ModelAtNewPosition, \
    MoveAndScalingOfObject)
  dev_display (ModelAtNewPosition)
  affine_trans_pixel (MoveAndScalingOfObject, 0, 0, RowTrans, \
    ColTrans)
  gen_cross_contour_xld (Cross, RowTrans, ColTrans, 6, Angle[Match])
  dev_display (Cross)
endfor

```

Note that you can alternatively also visualize the ncc matches conveniently using the visualization procedure.

```

dev_display_ncc_matching_results ([ModelIDCE, ModelIDBrand], \
  VisualizationColor, Row, Column, \
  Angle, ModelIndex)

```

3.1.4.6 Specify the Needed Accuracy (SubPixel)

The parameter `SubPixel` specifies if the position and orientation of a found model instance is returned with pixel accuracy (`SubPixel` set to `'false'`) or with subpixel accuracy (`SubPixel` set to `'true'`). As the subpixel accuracy is almost as fast as the pixel accuracy, we recommend to set `SubPixel` to `'true'`.

3.1.4.7 Restrict the Number of Pyramid Levels (NumLevels)

The parameter `NumLevels`, which you already specified when creating the model, allows you to use a different (in most cases a more restrictive) value in the search process. By using the value 0 for `NumLevels`, the value specified when creating the model is used.

Optionally, `NumLevels` can contain a second value, so that you can specify not only the highest but also the lowest pyramid level used for the search. If the search is finished on a pyramid level that is higher than the first pyramid level, which corresponds to the original, full-sized image, the search becomes faster. On the other hand, the search is then also less robust and less accurate.

3.1.4.8 Set the NCC Model Parameter 'timeout' via `set_ncc_model_param`

If your application demands that the search must be carried out within a specific time, you can set the parameter `'timeout'` with the operator `set_ncc_model_param` to specify the maximum period of time after which the search is guaranteed to terminate, i.e., you can make the search interruptible. But note that for an interrupted search no result is returned. Additionally, when setting the timeout mechanism, the runtime of the search may be increased by up to 10%.

3.2 Shape-Based Matching


The shape-based matching does not use the gray values of pixels and their neighborhood as template but describes the model by the shapes of contours. The model can also be extended by further constraints regarding neighboring edges. The following sections

- introduce the general concept of a shape-based matching application ([section 3.2.1](#)),
- show a first example for a shape-based matching ([section 3.2.2](#)),
- explain how to select an appropriate ROI to derive the template image from the reference image ([section 3.2.3](#)),
- explain how to create a suitable model ([section 3.2.4](#)),
- explain how to train the model ([section 3.2.5](#))
- explain how to optimize the search of object instances ([section 3.2.6](#)),
- explain how to restrict the search to a specific region of interest ([section 3.2.7](#)),
- explain how to search for multiple models simultaneously ([section 3.2.8](#)),
- give ideas how to optimize the matching speed ([section 3.2.9](#)),
- explain how to use multiple shape-based matching results, ([section 3.2.11](#)), and
- explain how to adapt to a changed camera orientation ([section 3.2.12](#)).

You can also learn about shape based matching and many other topics in interactive online courses at our [MVTec Academy](#).

3.2.1 General Concept

Shape-based matching using the generic interface consists of these main steps:



Step 1: Select the object in the reference image ([section 3.2.3](#) on page 56)

After grabbing the reference image the first task is to create a region containing the object. This object will serve as training template in *Step 3: Train the model*.

Step 2: Create a suitable shape model ([section 3.2.4](#) on page 58)

The model is created using the operator `create_generic_shape_model`. As a result, the operator returns a handle for the newly created model (`ModelID`).

Parameters modifying the model `ModelID` itself have to be set in this step using `set_generic_shape_model_param`. Changing a corresponding value later makes it necessary to retrain the adjusted model before the matching. This is also indicated by the parameter 'needs_training', which you can retrieve using `get_generic_shape_model_param`.

Note that if you use HALCON's .NET, or C++ interface and call the operator via the class `HShapeModel`, no handle is returned because the instance of the class itself acts as your handle.

Step 3: Train the model ([section 3.2.5](#) on page 64)

Train the created model using `train_generic_shape_model` in order to find the training pattern given in `Template`.

Step 4: Find object instances (section 3.2.6 on page 64)

Find the model object instances using `find_generic_shape_model`. Parameters optimizing the search of a model instance can be set using `set_generic_shape_model_param` as their modification does not necessitate a model training. Note, that this is not the case if they are set from a non-estimated value to a value leading to an automatic estimation during `train_generic_shape_model`.

3.2.2 A First Example

In this section we give a first example showing the general steps of a matching process mentioned in section 3.2.1 on page 54. To follow the example actively, start the HDevelop program `%HALCONEXAMPLES%\solution_guide\matching\first_example_shape_matching.hdev`, which locates the print on an IC.

Step 1: Select the object in the reference image

After grabbing the reference image the first task is to create a region containing the object. In the example program, a rectangular region is created using the operator `gen_rectangle1`. Alternatively, you can draw the region interactively using, e.g., `draw_rectangle1` or use a region that results from a previous segmentation process. Then, an image containing just the selected region, i.e., the template image, is created using the operator `reduce_domain`. The result is shown in figure 3.3.

```
gen_rectangle1 (ROI, 450, 470, 580, 755)
reduce_domain (Image, ROI, ModelImage)
```

Step 2: Create a suitable shape model

With the operator `create_generic_shape_model`, the model is created.

```
create_generic_shape_model (ModelID)
```

For this example the we set the parameter 'num_levels' in order to illustrate a possible way how to determine and set the model levels.

Step 3: Train the model

The created model needs to be trained to find the wanted object. This is done calling `train_generic_shape_model` with the model to be trained, `ModelID`, and the training pattern given in `Template`.

```
train_generic_shape_model (ModelImage, ModelID)
```

Step 4: Find object instances

To find the object in a search image, all you need to do is call the operator `find_generic_shape_model`. Figure 3.4 shows the result for one of the example images.




Figure 3.3: (left) reference image with the ROI that specifies the object; (right) the internal model (4 pyramid levels).




Figure 3.4: Finding the object in other images.

```

for i := 2 to 9 by 1
    read_image (SearchImage, 'board/board_' + i$'02')
    find_generic_shape_model (SearchImage, ModelID, MatchResultID, \
                             NumMatchResult)

    if (NumMatchResult > 0)
        get_generic_shape_model_result_object (Objects, MatchResultID, \
                                              'all', 'contours')
    endif
endfor

```

The operator `find_generic_shape_model` returns the number of found instances `NumMatchResults` and a handle `MatchResultID`. Using the handle `inspect` you can directly inspect the found instances (see the HDevelopUser's Guide, [section 6.22.3](#) on page 170 for more information to the handle `inspect`). But the primary function of the handle is to store the data of the found instances. With this handle you can call the operator `get_generic_shape_model_result` to retrieve results like position or score of a found instance or call `get_generic_shape_model_result_object` to retrieve iconic results. Such an iconic result is, e.g., the model contour transformed according to the matching result, which can directly be displayed.

The following sections go deeper into the details of the individual steps of a shape-based matching and the parameters that have to be adjusted.

3.2.3 Select the Model ROI

As a first step of the shape-based matching, the region of interest that specifies the template image must be selected. This region can have an arbitrary shape, i.e., it can also contain holes or consist of several parts that are not connected. Thus, it is possible to create an ROI that contains only “robust” parts of the object. The ROI must be selected so that it contains all prominent structures of the object and also some pixels outside of them, as their immediate surroundings (neighborhood) are needed to derive the model. Furthermore, you can speed up the later search using a subsampling (see [section 3.2.4.2](#) on page 59). For that, the ROI should not be too “thin”, because otherwise it vanishes at higher pyramid levels! As a rule of thumb, you are on the safe side if an ROI is $2^{NumLevels-1}$ pixels wide. That is, a width of 8 pixels allows to use 4 pyramid levels. After having selected a suitable ROI, the reduced image is used as template image for the creation of the model.

When using shape-based matching in the presence of clutter in the reference image, you can also use the operator `inspect_shape_model` to improve an interactively selected ROI by additional image processing. This is shown in the HDevelop example program `%HALCONEXAMPLES%\solution_guide\matching\process_shape_model.hdev`, which locates the arrows that are shown in [Figure 3.5](#).

Step 1: Select the arrow

There, an initial ROI is created around the arrow, without trying to exclude clutter (see [figure 3.5a](#)).

```

gen_rectangle1 (ROI, 361, 131, 406, 171)
reduce_domain (ModelImage, ROI, ImageROI)

```




Figure 3.5: Processing the result of `inspect_shape_model`: a) interactive ROI; b) models for different values of `Contrast`; c) processed model region and corresponding ROI and model; d) result of the search.

Step 2: Create a first model region

Then, this ROI is inspected via `inspect_shape_model`.

```
inspect_shape_model (ImageROI, ShapeModelImage, ShapeModelRegion, 1, 30)
```

Figure 3.5b shows the shape model regions that would be created for different values of the parameter `Contrast`. As you can see, you cannot remove the clutter without losing characteristic points of the arrow itself.

Step 3: Process the model region

This problem can be solved by exploiting the fact that the operator `inspect_shape_model` returns the shape model region. Thus, you can process it like any other region. The main idea to get rid of the clutter is to use the morphological operator `opening_circle`, which eliminates small regions. Before this, the operator `fill_up` must be called to fill the inner part of the arrow, because only the boundary points are part of the (original) model region.

```
fill_up (ShapeModelRegion, FilledModelRegion)
opening_circle (FilledModelRegion, ROI, 3.5)
```

Step 4: Create the final model

The obtained region is then used to create the model for a matching that locates all arrows successfully. Figure 3.5c shows the processed region, the corresponding region of interest, and the final model region.

```
reduce_domain (ModelImage, ROI, ImageROI)
create_generic_shape_model (ModelID)
train_generic_shape_model (ImageROI, ModelID)
```

3.2.4 Create a Suitable Shape Model

Having derived the template image from the reference image, the shape model can be trained. Before the training, an empty shape model needs to be created using `create_generic_shape_model` and possibly configured. As mentioned in [section 3.2.1](#) on page 54, parameters modifying the model itself need to be set before the training. In the following we will explain some of the concepts and the parameters to adjust them:

- Specify which pixels are part of the model: 'contrast_low', 'contrast_high', and 'min_size' ([section 3.2.4.1](#)),
- Speed up the search by using a subsampling, i.e., by adjusting the parameter 'num_levels', and by reducing the number of model points, i.e., by adjusting the parameter 'optimization' ([section 3.2.4.2](#)),
- Allow a specific range of scale ([section 3.2.4.3](#)),
- Modify the angle sampling ([section 3.2.4.4](#)),
- Specify how pixels are compared with the model in the later search: 'metric' ([section 3.2.4.5](#)),

Note that when adjusting the parameters you can also let HALCON assist you:

- Use automatic parameter suggestion: For many parameters you can let HALCON suggest suitable parameters. Then, you can either leave 'auto' as the corresponding parameter value (default value for the parameters where HALCON can automatically determine value suggestions, see [set_generic_shape_model_param](#)), or you apply [determine_shape_model_params](#) to automatically determine parameters for a shape model from a template image and then decide individually if you set the suggested values for the shape model. Note that both approaches return only approximately the same values and [train_generic_shape_model](#) and [find_generic_shape_model](#), respectively, return the more precise values.
- Inspect the model: HALCON offers different options to inspect the model and check automatically set parameters, see [section 3.2.5.1](#).

3.2.4.1 Specify Pixels that are Part of the Model: 'contrast_low', 'contrast_high', and 'min_size'

The shape of the object needs to be extracted without possible clutter. This is done by:

- Selecting all those points whose neighboring contrast exceeds a certain threshold.
- Rejecting components smaller than the minimum number of points.

Alternative methods to remove clutter are to modify the ROI as described in [section 3.2.3](#) on page 56 or create a synthetic model.

[Figure 3.6](#) shows an example. The task is to create a model for the outline of the pad. If the complete outline is selected, the model also contains clutter ([figure 3.6a](#)). If the clutter is removed, parts of the outline are missing ([figure 3.6b](#)).



You can let HALCON select suitable values itself by specifying the value 'auto' for the respective parameters. But it is recommended to verify them using [inspect_shape_model](#) before training the model. In the following we will have a look at these parameters.

Setting the required model contrast-parameters: For the model those pixels are selected, whose contrast, i.e., gray value difference to neighboring pixels, exceeds a threshold. In order to obtain a suitable model the contrast should be chosen in such a way that the *significant* pixels of the object are included, i.e., those pixels that characterize it and allow to discriminate it clearly from other objects or from the background. Obviously, the model should not contain clutter, i.e., structures that do not belong to the object.

The threshold can be specified as a single value or as a hysteresis threshold, depending on the values for 'contrast_low' and 'contrast_high', which can be set using [set_generic_shape_model_param](#).




Figure 3.6: Selecting significant pixels by adapting the contrast using `set_generic_shape_model_param`:

- a) complete object containing clutter: `'contrast_low' = 'contrast_high' = 40`;
- b) little clutter but incomplete object: `'contrast_low' = 'contrast_high' = 70`;
- c) hysteresis threshold: `'contrast_low' = 40, 'contrast_high' = 70`;
- d) minimum contour size: `'contrast_low' = 'contrast_high' = 40` and `'min_size' = 30`.

A hysteresis threshold is useful in cases where it is impossible to find a single threshold value that removes all the clutter but not also parts of the object. It uses two thresholds, a lower and an upper threshold (see also the operator `hysteresis_threshold`). For the model, first pixels that have a contrast higher than the upper threshold are selected. Then, pixels that have a contrast higher than the lower threshold and that are connected to a high-contrast pixel, either directly or via another pixel with contrast above the lower threshold, are added. This method enables you to select contour parts whose contrast is locally low.

Returning to the example with the pad: As you can see in [figure 3.6c](#), with a hysteresis threshold you can create a model for the complete outline of the pad without clutter.

The following line of code shows how to inspect the two thresholds (note, operator `inspect_shape_model` uses a different syntax):

```
inspect_shape_model (ImageReduced, ModelImages, ModelRegions, 2, [40,70])
```

Note also, the parameter `'contrast_low'` and `'contrast_high'` are for the model. The minimal contrast a point must at least have in a search image in order to be compared with the model is determined by `'min_contrast'`.

Setting a minimum model size using `'min_size'`: An other possibility to remove clutter is to specify a minimum size, i.e., number of pixels, for the contour components. [Figure 3.6d](#) shows the result for the example task. The minimum size is set through `'min_size'` using `set_generic_shape_model_param`.

The following line of code shows how to inspect the minimum size (note, operator `inspect_shape_model` uses a different syntax):

```
inspect_shape_model (ImageReduced, ModelImages, ModelRegions, 2, [40,40,30])
```

3.2.4.2 Speed Up the Search using Subsampling and Point Reduction: `'num_levels'` and `'optimization'`

The search can be sped up by reducing the number of points to examine. This can be achieved by either

- reducing the number of image points via subsampling,

- reducing the number of model points via model optimization.

In the following, we will have a look at these possibilities.

Subsampling: Subsampling can be used to speed up the search (see also [section 2.3](#) on page 25 and [section 2.4.2](#) on page 28). There, an image pyramid is created, consisting of the original, full-sized image and a set of downsampled images. The model is then trained to search on the different pyramid levels.

You can specify how many pyramid levels are used via the parameter 'num_levels' using [set_generic_shape_model_param](#). We recommend to choose the highest pyramid level at which the model contains at least 10-15 pixels and in which the shape of the model still resembles the shape of the object.



A much easier method is to **let HALCON select a suitable value itself** by specifying the value 'auto' for 'num_levels'. You can then query the used value via the operator [get_generic_shape_model_param](#).

Note that if you want to set the number of pyramid levels manually, you can inspect the template image pyramid using the operator [inspect_shape_model](#), e.g., as shown in the HDevelop program %HALCONEX-AMPLES%\solution_guide\matching\first_example_shape_matching.hdev. After the call to [inspect_shape_model](#), the model regions on the selected pyramid levels are displayed in HDevelop's Graphics Window. You can have a closer look at them using the online zooming (menu entry Visualization ▸ Zoom Window). The code lines following the operator call loop through the pyramid and determine the highest level on which the model contains at least 15 points. This value is then used in the call to the operator [train_generic_shape_model](#).

```
inspect_shape_model (ModelImage, ShapeModelImages, ShapeModelRegions, 8, 30)
area_center (ShapeModelRegions, AreaModelRegions, RowModelRegions, \
             ColumnModelRegions)
* Possible way to determine the model levels.
count_obj (ShapeModelRegions, HeightPyramid)
for i := 1 to HeightPyramid by 1
    if (AreaModelRegions[i - 1] >= 15)
        NumLevels := i
    endif
endfor
create_generic_shape_model (ModelID)
set_generic_shape_model_param (ModelID, 'num_levels', NumLevels)
train_generic_shape_model (ModelImage, ModelID)
```

Note that the operator [inspect_shape_model](#) returns the pyramid images in form of an image tuple (array). The individual images can be accessed like the model regions with the operator [select_obj](#). Please note that **object tuples start with the index 1, whereas control parameter tuples start with the index 0!**



Model optimization: A further reduction of model points can be enforced via the parameter 'optimization' using [set_generic_shape_model_param](#). This may be useful to speed up the matching in the case of particularly large models. Again, we recommend to **specify the value 'auto' to let HALCON select a suitable value itself**. Please note that regardless of your selection all points passing the contrast criterion are displayed, i.e., you cannot check which points are part of the model.



3.2.4.3 Allow a Range of Scale

You can specify an allowed range of scale. To do so, two forms are on hand:

Isotropic scaling: Identical scaling in row and column direction (uniform scaling). To do so, you specify the smallest and largest allowed scaling value as well as the step length between two sampled points in order to determine the granularity of the scaling. See also [set_generic_shape_model_param](#).

Anisotropic scaling: Different scaling in row and column direction. To do so, you also specify the smallest and largest allowed scaling value as well as the step length between two sampled points in order to determine the granularity of the scaling. But in case of anisotropic scaling you specify these values for the row and column direction independently. See also [set_generic_shape_model_param](#).

Note that you can further limit the allowed range during the search. Thus, if you want to reuse a model for different tasks requiring a different range of scale, you can use a large range when creating the model and a smaller range for the search.

Note that if you are searching for the object on a large range of scales you should **create the model based on a large scale** because HALCON cannot “guess” model points when precomputing model instances at scales larger than the original one. On the other hand, the image pyramid should be created such that the highest level contains enough model points also for the smallest scale. See [section 2.3](#) on page 25 and [section 3.2.4.2](#) on page 59 for further information about the image pyramid and its setting.

If you select the value 'auto' for the step parameters 'iso_scale_step' and its anisotropic equivalents, HALCON automatically chooses a suitable step size to obtain the highest possible accuracy by determining the smallest scale change that is still discernible in the image. A scaled object is clearly discernible from the original if the point that lies farthest from the center of the object is moved by at least 2 pixels. Therefore, the corresponding scale change Δs_{opt} is calculated as follows:

$$\Delta s = \frac{d}{l} \Rightarrow \Delta s_{opt} = \frac{2}{l}$$

with l being the maximum distance between the center and the object boundary and $d = 2$ pixels. For some models, the such estimated scale step size is still too large. In these cases, it is divided by 2 automatically.

The automatically determined scale step size is suitable for most applications. Therefore, **we recommend to select the value 'auto'**. By selecting a higher value you can speed up the search process, however, at the cost of a decreased accuracy of the estimated scale. Note that for very high values the matching may fail altogether!

The value chosen for the step parameters should not deviate too much from the optimal value ($\frac{1}{3}\Delta s_{opt} \leq \Delta s \leq 3\Delta s_{opt}$). Note that choosing a very small step size does not result in an increased scale accuracy!

3.2.4.4 Modify the Angle Sampling: 'angle_step'

During the matching process, the model is searched for in different angles within the allowed range, at steps specified with the parameter 'angle_step'. Thus, for orientation angles

$$'angle_start' + n \times 'angle_step'$$

with n being an integer. In this section we will have a look at possible values for 'angle_step'.

If you select the value 'auto', HALCON automatically chooses an optimal step size ϕ_{opt} to obtain the highest possible accuracy by determining the smallest rotation that is still discernible in the image. The underlying algorithm is explained in [figure 3.7](#): The rotated version of the cross-shaped object is clearly discernible from the original if the point that lies farthest from the center of the object is moved by at least 2 pixels. Therefore, the corresponding angle ϕ_{opt} is calculated as follows:

$$d^2 = l^2 + l^2 - 2 \cdot l \cdot l \cdot \cos \phi \Rightarrow \phi_{opt} = \arccos \left(1 - \frac{d^2}{2 \cdot l^2} \right) = \arccos \left(1 - \frac{2}{l^2} \right)$$

with l being the maximum distance between the center and the object boundary and $d = 2$ pixels. For some models, the angle step size estimated this way is too large. In these cases, it is divided by 2 automatically.




Figure 3.7: Determining the minimum angle step size from the extent of the model.

The automatically determined angle step size ϕ_{opt} is suitable for most applications. Therefore, **we recommend to select the value 'auto'**. You can query the used value after the creation via the operator




Figure 3.8: The effect of a large 'angle_step' on the matching.

`get_generic_shape_model_param`. By selecting a higher value you can speed up the search process, however, at the cost of a decreased accuracy of the estimated orientation. Note that for very high values the matching may fail altogether!

The value chosen for 'angle_step' should not deviate too much from the optimal value ($\frac{1}{3}\phi_{opt} \leq \phi \leq 3\phi_{opt}$). Note that choosing a very small step size does not result in an increased angle accuracy!

What happens in case of not suitable step size?

If the value for the parameter 'angle_step' is significantly larger than the automatically selected minimum value, the effect depicted in figure 3.8 can occur: If the object lies between two precomputed angles, points lying far from the center are not matched to a model point, and therefore the score decreases.

Of course, the same line of reasoning applies to the parameter 'scale_step' and its variants for anisotropic scaling (see section 3.2.4.3 on page 60).

3.2.4.5 Specify how Pixels are Compared with the Model: 'metric'

Polarity, i.e., the direction of the contrast, can vary between different images. You can specify whether and how the polarity must be observed (see figure 3.9) using the parameter 'metric'.




Figure 3.9: The parameter 'metric' specifies how to consider the polarity of the model.

Different ways to handle the polarity and thus different values for 'metric':

'use_polarity': The polarity is observed, i.e., the points in the search image must show the same direction of the contrast as the corresponding points in the model. If, e.g., the model is a bright object on a dark background, the object is found in the search images only if it is also brighter than the background.

'ignore_global_polarity': The polarity is globally ignored. In this mode, an object is recognized also if the direction of its contrast reverses, e.g., if your object can appear both as a dark shape on a light background and vice versa. This flexibility, however, is obtained at the cost of a slightly lower recognition speed.

'ignore_local_polarity': The object is found even if the contrast changes locally. This mode can be useful, e.g., if the object consists of a part with a medium gray value, within which either darker or brighter sub-objects lie. Please note however, that the recognition speed and the robustness may decrease dramatically in this mode, especially if you allowed a large range of rotation (see [section 3.2.6.1](#) on page 65).

'ignore_color_polarity': You can perform the matching in color images (or, more generally, in multi-channel images). An example is %HALCONEXAMPLES%\hdevelop\Applications\Position-Recognition-2D\matching_multi_channel_yogurt.hdev.

If you will train your **model with XLD contours**, there is no information about the polarity of the model available. Thus, for such a model the value of 'metric' must be set to 'ignore_local_polarity'. You can change this setting after a first successful matching. With [get_generic_shape_model_result](#) you can retrieve the transformation matrix to project the contour onto the search image. Then, with the operator [set_shape_model_metric](#) you can determine the polarity of the first search image, which is used as training image for the polarity, and set the match metric to 'use_polarity' or 'ignore_global_polarity'. Using the value 'use_polarity', i.e., if the following search images have the same polarity as the training image, the search becomes faster and more robust.

An example is %HALCONEXAMPLES%\hdevelop\Matching\Shape-Based\create_shape_model_xld.hdev.

Note that [set_shape_model_metric](#) is only available for models that are created from XLD contours!



3.2.4.6 Use Extended Parameter Estimation

Fine-tuning of the matching parameters can be challenging, as many factors need to be taken into account simultaneously. Extended parameter estimation can be used to automatically estimate parameter values based on sample images. The parameter that controls extended parameter estimation is 'extended_parameter_estimation'.

The samples need to be provided as a tuple of dictionaries, each containing both an image and one or multiple ROIs defining the respective ground-truth of instances of the object to be searched for. They are then added to the shape model using [set_generic_shape_model_param](#) with the parameter 'training_samples'.

```
TrainingSamples[0] := dict{image: SampleImage1, region: SampleImage1Labels}
TrainingSamples[1] := dict{image: SampleImage2, region: SampleImage2Labels}
set_generic_shape_model_param (ModelID, 'training_samples', TrainingSamples)
```

After the samples are set, (re-)training the shape model with [train_generic_shape_model](#) adapts the parameter value according to the selected estimation algorithm.

```
set_generic_shape_model_param (ModelID, 'extended_parameter_estimation', \
                               'linear')
```

If further search parameters are changed after training, it is recommended to retrain the model, as the change might impact the estimation result. For each training that is supposed to take samples into account the samples need to be set explicitly. Intermediate results of the extended parameter estimation process can not be accessed.

The estimation highly depends on the selection of sample images and the respective sample regions, therefore the following criteria should be considered when choosing a set of samples:

Representation

The samples need to be representative of all instances that can occur in the search process. For example, if objects can be occluded in search images, this should be reflected in the sample images as well.

Number of samples

The ideal number of samples depends on the variety of object instances that should be detected. Most cases are sufficiently covered by providing about 10 samples.

Sample region

The sample region must contain the complete model instance, meaning all contours of the model instance have to be inside the sample region.

Model reference point

The model's reference point needs to be within the sample region.

3.2.5 Train the Shape Model

Once the model is created and all parameters modifying the model are set, it is time to train the model to find the object. This is done using `train_generic_shape_model`, where the shape is presented via the `Template`. The model parameter `'needs_training'` is changed to `'false'` (see `get_generic_shape_model_param`).

Note that if you derive your model from an XLD contour, after a first match, it is strongly recommended to determine the polarity information for the model with `set_shape_model_metric` (see [section 3.2.4.5](#) on page 62 and [section 2.1.3.2](#) on page 22 for details).

3.2.5.1 Inspect the Shape Model

You can apply the operator `inspect_shape_model` to the template image to try different values for the parameters `'num_levels'` and contrast parameters. The operator returns the resulting representation of a shape model with multiple pyramid levels and thus allows you to visually check if the object of interest is represented adequately by the model. If several combinations of parameter values do not lead to a satisfying representation of the object, maybe the template image, i.e., the model's ROI was not selected properly. How to select a suitable ROI is described in [section 2.1](#) on page 17.

If you want to visually inspect an already created shape model, you can use `get_generic_shape_model_object` to get the XLD contours that represent the model in a specific pyramid level. Note that the contours are stored within the model handle and you can inspect them also using the handle inspect (see the HDevelopUser's Guide, [section 6.22.3](#) on page 170 for more information to the handle inspect).

To inspect the current parameter values of a model, you query them with `get_generic_shape_model_param` or inspect them using the handle inspect. This may be necessary if during the creation of the model an automatic parameter selection was used or if the model was created within another program, saved to file with `write_shape_model`, and read from this file in the current program with `read_shape_model`. Additionally, you can query the coordinates of the origin of the model using `get_shape_model_origin`.

You still can change model parameters using `set_generic_shape_model_param`, even for a trained model, but `'needs_training'` will be changed to `'true'` and necessitates to retrain the model.

3.2.6 Find Object Instances

This section addresses the search of model instances. Shape-based matching uses the operator `find_generic_shape_model` to find object instances and the results can be retrieved using `get_generic_shape_model_result` and `get_generic_shape_model_result_object` for iconic results.

We discuss different concepts with the corresponding search parameters to optimize your specific search problem, namely:

Modify allowed instances

- Restrict the Model Orientation ([section 3.2.6.1](#)),

Sort out found matches

- Specify the Visibility of the Object: `'min_score'` ([section 3.2.6.2](#)),
- Search for Multiple Instances of the Object: `'num_matches'` ([section 3.2.6.3](#) on page 66),
- Sort out Matches According to their Overlap: `'max_overlap'` ([section 3.2.6.4](#) on page 66),
- Restrict the Search by Specifying Clutter ([section 3.2.6.5](#)).

Specify the Needed Accuracy

- Specify the Needed Accuracy: `'subpixel'`, `'max_deformation'` ([section 3.2.6.6](#) on page 69)

As prerequisite you need a trained model with suitable settings of parameters modifying the model itself (for the model inspection see [section 3.2.5.1](#)).

3.2.6.1 Restrict the Model Orientation

Shape models created with `create_generic_shape_model` have a rotation range from 0 to $6.28 (= 2\pi)$ (unit: radians). Thereby the range of rotation is defined relative to the reference image, i.e., a starting angle of 0 corresponds to the orientation the object has in the reference image.

This range of orientation can be restricted, meaning `find_generic_shape_model` searches model instances of different angles within the range defined by 'angle_start' and 'angle_end'. In case the range is not a multiple of the step size the start and end value are automatically adapted (see also [section 3.2.4.4](#) on page 61 and the part Automatic modifications in `set_generic_shape_model_param`). In certain cases it is advantageous to restrict the model orientation:

- If you are only interested in matches within a certain orientation.
- If the object is (almost) symmetric you should limit the allowed range during the search by `find_generic_shape_model`. Otherwise, the search process will find multiple, almost equally good matches on the same object at different angles. Which match (at which angle) is returned as the best can therefore change seemingly random from image to image. The suitable range of rotation depends on the symmetry: For a cross-shaped or square object the allowed extent must be less than 90° , for a rectangular object less than 180° , and for a circular object 0° (see [figure 3.10](#)). To limit the range you can set 'angle_start' or 'angle_end' using `set_generic_shape_model_param`.




Figure 3.10: Suitable angle ranges for rotation symmetric objects.

3.2.6.2 Specify the Visibility of the Object: 'min_score'

With the parameter 'min_score' you can specify how much of the model must be visible. A typical use of this mechanism is to allow a certain degree of occlusion as demonstrated in [figure 3.11](#): The security ring is found if 'min_score' is set to 0.7.

Let's take a closer look at the term "visibility": When comparing a part of a search image with the model, the matching process calculates the so-called score, which is primarily a measure of how many model points could be matched to points in the search image (ranging from 0 to 1). A model point may be "invisible" and thus not matched because of multiple reasons:

- Parts of the object's contour are occluded, e.g., as depicted in [figure 3.11](#).

Please note that by default **it depends on your system settings whether objects which are clipped at the image border are found**. This behavior can be changed with `set_generic_shape_model_param`




Figure 3.11: Searching for partly occluded objects: a) model of the security ring; b) search result for 'MinScore' = 0.8; c) search result for 'MinScore' = 0.7.

(ModelID, 'border_shape_models', 'true'). An example is %HALCONEXAMPLES%\hdevelop\Applications\Position-Recognition-2D\matching_image_border.hdev.

- Parts of the contour have a contrast lower than specified in the parameter 'min_contrast'.
- The polarity of the contrast changes globally or locally (see [section 3.2.4.5](#) on page 62).
- If the object is deformed, which includes also the case that the camera observes the scene under an oblique angle, parts of the contour may be visible but appear at an incorrect position and therefore do not fit the model anymore. In such cases the following options may help:
 - Deformed objects might be found if you set the parameter 'max_deformation' (see [section 3.2.6.6](#) on page 69).
 - Deformed or defocused objects might be found if the increased tolerance mode is activated, see 'pyramid_level_robust_tracking' in [set_generic_shape_model_param](#).
 - Perspective deformations that occur because of an oblique camera view can be handled as described in [section 3.2.12](#) on page 77.

Besides these obvious reasons, which have their root in the search image, there are some not so obvious reasons caused by the matching process itself:

- HALCON precomputes the model for intermediate angles within the allowed range of orientation. During the search, a candidate match is then compared to all precomputed model instances. A not-suitable value for the parameter 'angle_step' may lead to a significant score drop, see [section 3.2.4.4](#) on page 61. Note, the same line of reasoning occurs also for model scaling.
- Another stumbling block lies in the use of an image pyramid which was introduced in [section 2.3](#) on page 25: When comparing a candidate match with the model, the specified minimum score must be reached on each pyramid level. However, on different levels the score may vary, with only the score on the lowest level being returned in the parameter 'score'. This sometimes leads to the apparently paradox situation that 'min_score' must be set significantly lower than the resulting 'score'. Note that if the matches are not tracked to the lowest pyramid level it might happen that instances with a score slightly below 'min_score' are found.



Recommendation: The higher 'min_score', the faster the search!

To automatically determine a suitable value based on provided sample images you can use extended parameter estimation (see [section 3.2.4.6](#) on page 63).

3.2.6.3 Search for Multiple Instances of the Object: 'num_matches'

The maximal number of returned instances can be determined using 'num_matches'. Its exact behavior depends on whether clutter is set, see also [set_generic_shape_model_param](#). Note, in case of no clutter set, promising matches are limited on every pyramid level to the number 'num_matches'. This may result in rejecting instances on a certain pyramid level even though they would achieved a higher score on a lower pyramid level. As a consequence it is possible that you obtain a different instance (with a lower score) by setting 'num_matches' to 1 than the instance found with highest score by setting 'num_matches' to a higher number or even 'all'.

3.2.6.4 Sort out Matches According to their Overlap: 'max_overlap'

The parameter, 'max_overlap', lets you specify how much two matches may overlap (as a fraction). In [figure 3.12b](#), e.g., the two security rings overlap by a factor of approximately 0.2. In order to speed up the matching as much as possible, however, the overlap is calculated neither for the models themselves nor for the models ROI but for their smallest surrounding rectangle. This must be kept in mind when specifying the maximum overlap. In most cases, therefore a larger value is needed (e.g., compare [figure 3.12b](#) and [figure 3.12d](#)).

[Figure 3.12e](#) shows a “pathological” case: Even though the rings themselves do not overlap, their surrounding rectangles do to a large degree. Unfortunately, this effect cannot be prevented.




Figure 3.12: A closer look at overlapping matches: a) model of the security ring; b) model overlap; c) smallest rectangle surrounding the model; d) rectangle overlap; e) pathological case.

3.2.6.5 Restrict the Search by Specifying Clutter

In some cases, although multiple matches with high 'score' values are found within a search image, not all of them are wanted as results. For example, in [Figure 3.13](#), although only the structures consisting of eight dots are searched for, more instances are found. In order to get only the wanted matches, the instances have to be distinguished. This can be done considering the absence of edges in the neighborhood of the object as a distinctive feature.

To restrict the search to candidates that have no (or too faint) edges within a region specified relative to the model contours, you can use `set_generic_shape_model_object` to set a clutter region via 'clutter_region'.

The HDevelop example `%HALCONEXAMPLES%\solution_guide\matching\pick_and_place_2d.hdev` uses clutter to identify the objects with enough free space so a robot can pick them, see [Figure 3.14](#)

Set the clutter region

How to set a clutter region differs according to whether it is set before or after the training. In both cases, make sure that the clutter region does not contain negative coordinates and that it does not intersect the model contours.

Before the training the clutter region can be specified with regard to the template used for training.

```
gen_rectangle1 (ROI, 1685, 2710, 2215, 3120)
reduce_domain (Image, ROI, ImageReduced)
gen_rectangle2 (ROI_0, 1761, 2781, rad(-110), 55, 40)
gen_rectangle2 (ROI_1, 1856, 3083, rad(75), 55, 40)
union2 (ROI_0, ROI_1, ClutterRegion)
```

The transformation is estimated automatically in `train_generic_shape_model` when setting the selected region as the clutter region.

```
create_generic_shape_model (ModelID)
set_generic_shape_model_object (ClutterRegion, ModelID, 'clutter_region')
train_generic_shape_model (ImageReduced, ModelID)
```

A clutter region can also be specified after training, in relation to an instance found on a search image. Thus, also the transformation matrix describing the respective instance position needs to be set.




Figure 3.13: Extending a shape model by a clutter region to retrieve only the wanted results: a) select shape model (orange) from image; b) extracted shape model; c) matching results (green) without the use of clutter parameters (note, for visualization purpose not all rotated matches are shown); d) additionally define clutter region (blue); e) shape model (orange), extended by a clutter region (blue); f) matching result (green) with the use of clutter parameters.




Figure 3.14: Clutter is used to find all screws (green) with enough empty space at a defined place around (blue) so the robot can pick them.

Left: Regions to define the model (orange) and the clutter region (blue).

Right: Search image with found instances (green) and their corresponding clutter regions (blue).

Initially, we generate a model (without clutter) and train it.

```
create_generic_shape_model (ModelXLDID)
train_generic_shape_model (ModelXLD, ModelXLDID)
```

We open the image we want to use to define the clutter region. In this image we search for a model instance which we could use to define the clutter region. In the following code snippet it is Match 1.




Figure 3.15: Extending a shape model by a clutter region. The original shape model (orange) has its origin on the top left image corner (0,0). To define the clutter region (blue) relative to the model instance (green), the transformation from the shape model to the model instance needs to be given..

```
read_image (Image, 'screws/screws_backlight/screws_backlight_03')
find_generic_shape_model (Image, ModelXLDID, MatchResultDefID, \
                          NumMatchResult)

Match := 1
get_generic_shape_model_result_object (Objects, MatchResultDefID, Match, \
                                      'contours')
get_generic_shape_model_result (MatchResultDefID, Match, 'hom_mat_2d', \
                              HomMat2D)
```

For the found model instance we define the respective clutter region.

```
gen_rectangle2 (Rectangle1, 237, 2546, rad(-16), 55, 25)
gen_rectangle2 (Rectangle2, 557, 2435, rad(-16), 55, 25)
union2 (Rectangle1, Rectangle2, ClutterRegion)
```

Now we can set the clutter region, 'clutter_region', and use the transformation matrix of the found model instance, 'HomMat2D', to define the clutter transformation 'clutter_hom_mat_2d'.

```
set_generic_shape_model_object (ClutterRegion, ModelXLDID, 'clutter_region')
set_generic_shape_model_param (ModelXLDID, 'clutter_hom_mat_2d', HomMat2D)
```

Clutter and speed

Setting clutter is suitable to select matches but contrary to other options (see [section 3.2.9](#) on page 73) it does not reduce the runtime.

Clutter regions have their impact only at the end of the matching pipeline. This means, possible candidates are tracked through the image pyramid and only at the end the clutter region is checked based on the set clutter parameters. Thus, no candidate is rejected when tracking the candidates through the image pyramid. As a consequence, the runtime for a search using clutter is always at least as high as when returning all matches.

3.2.6.6 Specify the Needed Accuracy: 'subpixel', 'max_deformation'

Different aspects can affect the accuracy of a match, e.g., how precisely it is located (refinement: 'subpixel'), if it is slightly deformed ('max_deformation'), or its point of reference may be inapt. In this section we will have a look at these influences.

The matching process locates found instances in two steps:

1. Find coarse matches
2. Refine the matches

In the first step matches are located with an accuracy of 1 pixel for the position ('row', 'column') while the accuracy of the orientation and scale is equal to the values selected for the parameters 'angle_step' and 'scale_step' (or the corresponding parameters for anisotropic scaling), respectively (see [section 3.2.6.1](#) on page 65 and [section 3.2.4.3](#) on page 60).

The following refinement is optional and can be determined using the parameter 'subpixel':

'none': No refinement is performed.

'interpolation': HALCON examines the matching scores at the neighboring positions, angles, and scales around the best match and determines the maximum by interpolation. Using this method, the position is therefore estimated with subpixel accuracy ($\approx \frac{1}{20}$ pixel can be achieved). The accuracy of the estimated orientation and scale depends on the size of the object, like the optimal values for the parameters 'angle_step' and 'scale_step' or the corresponding parameters for anisotropic scaling (see [section 3.2.6.1](#) on page 65 and [section 3.2.4.3](#) on page 60): The larger the size, the more accurately the orientation and scale can be determined.

'least_squares', 'least_squares_high', or 'least_squares_very_high': A least-squares adjustment is used instead of an interpolation, resulting in a higher accuracy. However, this method requires additional computation time.

A note regarding the interplay with the score: While the refinement using the parameter 'subpixel' improves the position, orientation, and scale of a match, it does not update its score.

As mentioned above, in the first step the matches are located with an accuracy of 1 pixel. As a consequence, models that solely differ in their reference point (see [section 2.1.2.1](#) on page 18) may find the corresponding match with different scores, if the reference points do not differ by an integer in both directions.

Sometimes objects are not found or found only with a low accuracy because they are slightly deformed compared to the model. If your object is most probably deformed, you can allow a maximum deformation of a few pixels for the model by setting 'max_deformation' using [set_generic_shape_model_param](#). For example, if you set 'max_deformation' to 2, the contour of the searched object may differ by up to two pixels from the shape of the model. To get a meaningful score value and to avoid erroneous matches, we recommend to always combine the allowance of a deformation with a least-squares adjustment. Note that high values for the maximal allowed deformation increase the runtime and the risk of finding wrong model instances, especially for small models. Thus, the value should be chosen as small as possible but as high as necessary.



Please note that the **accuracy of the estimated position may decrease if you modify the point of reference** (e.g., using [set_shape_model_origin](#) or [set_generic_shape_model_param](#)). This effect is visualized in [figure 3.16](#): As you can see in the right-most column, an inaccuracy in the estimated orientation “moves” the modified point of reference, while the original point of reference is not affected. The resulting positional error depends on multiple factors, e.g., the offset of the point of reference and the orientation of the found object. The main point to keep in mind is that the error increases linearly with the *distance* of the modified point of reference from the original one (compare the two rows in [figure 3.16](#)).

An inaccuracy in the estimated scale also results in an error in the estimated position, which again increases linearly with the distance between the modified and the original point of reference.

For maximum accuracy in case the point of reference is moved, the position should be determined using the least-squares adjustment. Note that the accuracy of the estimated orientation and scale is not influenced by modifying the point of reference.

3.2.7 Restrict the Search to a Region of Interest

The obvious way to restrict the search space is to apply the operator [find_generic_shape_model](#) not to the whole image but only to an ROI. [Figure 3.17](#) shows such an example. The reduction of the search space can be realized in a few lines of code.

Step 1: Create a region of interest

First, you create a region, e.g., with the operator [gen_rectangle1](#) (see [section 2.1.1](#) on page 17 for more ways to create regions).




Figure 3.16: Two examples for the effect of the inaccuracy of the estimated orientation on a moved point of reference.

```
Row1 := 141
Column1 := 159
Row2 := 360
Column2 := 477
gen_rectangle1 (SearchROI, Row1, Column1, Row2, Column2)
```

Step 2: Restrict the search to the region of interest

Then, each search image is reduced to this ROI using the operator `reduce_domain`. In this example, the search speed is almost doubled using this method.

```
for i := 1 to 20 by 1
  read_image (SearchImage, 'board/board-' + i$'02')
  reduce_domain (SearchImage, SearchROI, SearchImageROI)
  find_generic_shape_model (SearchImageROI, ModelID, MatchResultID, \
    NumMatchResult)
endfor
```




Figure 3.17: Searching in a region of interest.




Figure 3.18: Searching for multiple models : a) models of ring and nut; b) search ROIs for the two models.

Note that by restricting the search to an ROI you actually restrict the position of the point of reference of the model, i.e., the center of gravity of the model ROI (see [section 2.1.2](#) on page 18). This means that the size of the search ROI corresponds to the extent of the allowed movement. For example, if your object can move ± 10 pixels vertically and ± 15 pixels horizontally you can restrict the search to an ROI of the size 20×30 . In order to assure a correct boundary treatment on higher pyramid levels, we recommend to enlarge the ROI by $2^{NumLevels-1}$ pixels in each direction. Thus, if you specified `NumLevels = 4`, you can restrict the search to an ROI of the size 36×46 .

Please note that even if you modified the point of reference (e.g., using `set_shape_model_origin` or `set_generic_shape_model_param` although this is not recommended), the original one, i.e., the center point of the model ROI, is used during the search. Thus, **you must always specify the search ROI relative to the original point of reference.**

3.2.8 Search for Multiple Models Simultaneously: ModelIDs

If you are searching for instances of multiple models in a single image, you can of course call the operator `find_generic_shape_model` multiple times. But a much faster alternative is to call the operator once and passing the different models simultaneously as a tuple to `ModelID`. Doing so we would mention the following points:

- In case of a single model, the resulting match instances contain an entry referring to the identifier of the respective model.
- To retrieve the result of a specific shape model using `get_generic_shape_model_result` and `get_generic_shape_model_result_object`, respectively, you can pass its identifier or handle to `MatchSelector`. See also `get_generic_shape_model_result` for more information about the `MatchSelector`.
- The search is always performed in a single image. However, you can restrict the search to a certain region for each model individually by passing an image array (see below for an example).
- The search parameters are the ones set for every model.

The example HDevelop program `%HALCONEXAMPLES%\solution_guide\matching\multiple_models.hdev` uses the operator `find_generic_shape_model` to search simultaneously for the rings and nuts depicted in [figure 3.18](#).

Step 1: Create the models

First, two models are created and trained, one for the rings and one for the nuts. The two model IDs are then concatenated into a tuple using the operator `assign`.


```

create_generic_shape_model (ModelIDRing)
set_generic_shape_model_param (ModelIDRing, 'iso_scale_min', 0.8)
set_generic_shape_model_param (ModelIDRing, 'iso_scale_max', 1.2)
train_generic_shape_model (ImageROIring, ModelIDRing)
create_generic_shape_model (ModelIDNut)
set_generic_shape_model_param (ModelIDNut, 'iso_scale_min', 0.6)
set_generic_shape_model_param (ModelIDNut, 'iso_scale_max', 1.4)
train_generic_shape_model (ImageROINut, ModelIDNut)
ModelIDs := [ModelIDRing, ModelIDNut]

```

Note how we set different model parameters restricting possible scaling.

Step 2: Specify individual search ROIs

In the example, the rings and nuts appear in non-overlapping parts of the search image. Therefore, it is possible to restrict the search space for each model individually. As explained in [section 3.2.7](#) on page 70, a search ROI corresponds to the extent of the allowed movement. Thus, narrow horizontal ROIs can be used in the example (see [figure 3.18b](#)).

The two ROIs are concatenated into a region array (tuple) using the operator `concat_obj` and then “added” to the search image using the operator `add_channels`, i.e., in each region the corresponding gray values of the search image are “painted”. The result of this operator is an array of two images, both having the same image matrix. The domain of the first image is restricted to the first ROI and the domain of the second image is restricted to the second ROI.

```

gen_rectangle1 (SearchROIring, 110, 10, 130, Width - 10)
gen_rectangle1 (SearchROINut, 315, 10, 335, Width - 10)
concat_obj (SearchROIring, SearchROINut, SearchROIs)
add_channels (SearchROIs, SearchImage, SearchImageReduced)

```

Step 3: Find all instances of the two models

Now, the operator `find_generic_shape_model` is applied to the created image array. Note that we can set the search parameters for each model individually. In this example we access the found instances and their model identifier one by one for illustration purpose, but we could also have accessed them all at once using a different match selector.

```

find_generic_shape_model (SearchImageReduced, ModelIDs, MatchResultID, \
                          NumMatchResult)
for i := 0 to NumMatchResult - 1 by 1
    get_generic_shape_model_result (MatchResultID, i, 'model_identifier', \
                                    Model)
    get_generic_shape_model_result_object (Objects, MatchResultID, i, \
                                           'contours')
    dev_display (Objects)
endfor

```

3.2.9 Optimize the Matching Speed

In the following, we show how to optimize the matching process in two steps. Please note that in order to optimize the matching it is very important to have a **set of representative test images from your application** in which the object appears in all allowed variations regarding its position, orientation, occlusion, and illumination.

Step 1: Assure that all objects are found

Before tuning the parameters for speed, we recommend to find settings such that the matching succeeds in all test images, i.e., that all object instances are found. If this is not the case when using the default values, check whether one of the following situations applies:

? Is the search algorithm “too greedy”?

As described in [section 2.4.3](#) on page 29, in some cases a perfectly visible object is not found if the ‘greediness’ is too high. Select the value 0 to force a thorough search.



? Is the object partly occluded?

If the object should be recognized in this state nevertheless, reduce the parameter `'min_score'`. For further information see [section 3.2.6.2](#) on page 65.

? Does the matching fail on the highest pyramid level?

As described in [section 3.2.6.2](#) on page 65, in some cases the minimum score is not reached on the highest pyramid level even though the score on the lowest level is much higher. Test this by reducing `'num_levels'`. Alternatively, set `'pyramid_level_highest'` or reduce `'min_score'`.

? Does the object have a low contrast?

If the object should be recognized in this state nevertheless, reduce the parameter `'min_contrast'` (see [set_generic_shape_model_param](#) and [section 3.2.4.1](#) on page 58).

? Is the polarity of the contrast inverted globally or locally?

If the object should be recognized in this state nevertheless, use the appropriate value for the parameter `'metric'` when creating the model (see [section 3.2.4.5](#) on page 62). If only a small part of the object is affected, it may be better to reduce `'min_score'` instead.

? Does the object overlap another instance of the object?

If the object should be recognized in this state nevertheless, increase the parameter `'max_overlap'` (see [section 3.2.6.4](#) on page 66).

? Are multiple matches found on the same object?

If the object is almost symmetric, restrict the allowed range of rotation as described in [section 3.2.6.1](#) on page 65 or decrease the parameter `'max_overlap'` (see [section 3.2.6.4](#) on page 66).

Step 2: Tune the Parameters Regarding Speed

If all objects are found but the application needs to be sped up, there are some ways to tune it, depending on both the model and the search parameters. Having a look at the runtime of the main parts of the matching pipeline can be a good indicator which steps of the search have significant/relevant speedup potential. Note that the time measurements will vary between calls, as they are strongly influenced by external factors such as, e.g., operating system interrupts or other processes like, e.g., antivirus software. The time measurements can be activated using [set_generic_shape_model_param](#) with `'pipeline'`. The result handles of the matching operation can then be queried using [get_generic_shape_model_result](#) and show the runtime along the matching pipeline. If significant time is spent in the top level stage, you could proceed like this:

- Increase `'min_score'` as far as possible, i.e., as long as the matching succeeds.
- Now, increase `'greediness'` until the matching fails. Try reducing `'min_score'`. If this does not help restore the previous values.
- Restrict the allowed range of rotation and scale as far as possible as described in [section 3.2.6.1](#) on page 65 and similar for the scaling. Alternatively, adjust the corresponding parameters already at model level as described in [section 3.2.4.3](#) on page 60.
- Restrict the search to a region of interest as described in [section 3.2.7](#) on page 70.

If significant time is spent in the tracking part and one is willing to trade off accuracy and robustness against speed, an approach would be this:

- Terminate the search on a higher pyramid level as described in [section 3.2.4.2](#) on page 59.
- Restrict the maximally allowed model deformation (see also [section 3.2.6.6](#) on page 69)

If the subpixel refinement takes a significant time in the matching pipeline, you could proceed like this:

- Switch subpixel refinement to `'interpolation'` or even switch it off with `'none'` to trade off accuracy against speed.

When acquiring your images and creating your model you can already influence the speed of the search significantly by preferring models with dominant structures to models with weak structures (see also [section 3.2.4.2](#) on page 59).

These additional methods are more “risky”, i.e., the matching may fail if you choose unsuitable parameter values:

- Increase 'min_contrast' as long as the matching succeeds.
- If you are searching for a particularly large object, it typically helps to select a higher point reduction with the parameter 'optimization' (see [section 3.2.4.2](#) on page 59).
- Increase 'angle_step' (and 'iso_scale_step' or the corresponding parameters for anisotropic scaling) as long as the matching succeeds.

3.2.10 Visualize Score Contribution

By retrieving score visualization contours, you can investigate how different parts of the model are contributing to the matching score.

Score visualization contours can, for example, help you to:

- check which model parts are matched well in images, and which model parts do not represent the actual object instances accurately,
- investigate why certain matches might have received higher or lower scores than expected.

For every found match, a score contribution value is calculated for each model point. While the 'score' of a match represents how well the model matches the object instance as a whole, contribution values are an indicator of how well the individual model points represent their respective object part. See the description of `score_visualization_contours` in `get_generic_shape_model_result_object` for details on what is considered in the calculation of these values.

You can use `get_generic_shape_model_result_object` to retrieve model contour parts for different score contribution ranges. The HDevelop example `%HALCONEXAMPLES%\hdevelop\Matching\Shape-Based\get_generic_shape_model_result_score_visualization_contours.hdev` shows how to extract and visually present the contours for further investigation.




Figure 3.19: Visualization of contours with different score contribution ranges in different colors. The left image shows that some parts of the model contours have significantly lower score contribution values (red) than the rest (green), which suggests that the red parts are not reliable representations and should be removed from the model. The right image shows why some instances result in lower matching scores, as partial occlusion decreases the score contribution of some model parts (red, orange, and yellow parts).

Choosing reasonable intervals for the selection of score visualization contours depends on your task. For instance, using 'min_score' as a threshold shows which points of the model helped the score of each match to exceed 'min_score' (green and yellow contours in [figure 3.19](#)) and which model points are not resembling the object enough to reach 'min_score' (orange and red contours).

Score contribution values can also be negative when the shape model's 'metric' is set to 'use_polarity' or 'ignore_global_polarity'. However, in all cases, a higher score contribution value indicates a better match between the model point and the object instance.




Figure 3.20: Displaying multiple matches; the used model is depicted in [figure 3.11a](#) on page 65.

3.2.11 Use Multiple Shape-Based Matching Results

Shape-based matching can return multiple instances of a model and additionally can deal with multiple models simultaneously. How to use these specific results is described in [section 3.2.11.1](#) and [section 3.2.11.2](#).

3.2.11.1 Deal with Multiple Matches

If multiple instances of the object are searched and found, the handle `MatchResultID` contains multiple entries. They are accessible by `get_generic_shape_model_result` and `get_generic_shape_model_result_object`, respectively.

Note, in this example we access the results one by one in order to illustrate how you can retrieve further characteristics of every found instance as e.g., the transformation matrix and the position of the instance. But we could also have accessed them all at once using a different match selector, see `get_generic_shape_model_result`.

```
find_generic_shape_model (SearchImage, ModelID, MatchResultID, \
                          NumMatchResult)
for j := 0 to NumMatchResult - 1 by 1
    get_generic_shape_model_result_object (Objects, MatchResultID, j, \
                                          'contours')
    get_generic_shape_model_result (MatchResultID, j, 'hom_mat_2d', \
                                   HomMat2D)
    get_generic_shape_model_result (MatchResultID, j, 'row', Row)
    get_generic_shape_model_result (MatchResultID, j, 'column', Column)
    affine_trans_pixel (HomMat2D, -120, 0, RowArrowHead, ColumnArrowHead)
    disp_arrow (WindowHandle, Row, Column, RowArrowHead, ColumnArrowHead, 2)
endfor
```

In this example, the transformation is also used to display an arrow that visualizes the orientation (see [figure 3.20](#)). For this, the position of the arrow head is transformed using `affine_trans_pixel` with the transformation matrix resulting from the instance. If you want to transform the matches yourself, **you must use `affine_trans_pixel`** and not `affine_trans_point_2d`.

3.2.11.2 Deal with Multiple Models

When searching for multiple models simultaneously as described in [section 3.2.8](#) on page 72, it is useful to store the information about the models, i.e., the XLD models, in tuples. The following example code stems from the already partly described HDevelop program `%HALCONEXAMPLES%\solution_guide\matching\multiple_models.hdev`, which uses the operator `find_generic_shape_model` to search simultaneously for the rings and nuts depicted in [figure 3.18](#) on page 72.

As already shown in [section 3.2.11](#), you can access the found instances of the different models directly. In this example we access them one by one to retrieve also the respective model identifier.

```
find_generic_shape_model (SearchImageReduced, ModelIDs, MatchResultID, \
                          NumMatchResult)
for i := 0 to NumMatchResult - 1 by 1
    get_generic_shape_model_result (MatchResultID, i, 'model_identifier', \
                                   Model)
    get_generic_shape_model_result_object (Objects, MatchResultID, i, \
                                          'contours')
    dev_display (Objects)
endfor
```

3.2.12 Adapt to a Changed Camera Orientation

As shown in the sections above, HALCON's shape-based matching allows to localize objects even if their position and orientation in the image or their scale changes. However, the shape-based matching fails if the camera observes the scene under an oblique angle, i.e., if it is not pointed perpendicularly at the plane in which the objects move, because an object then appears distorted due to perspective projection. Even worse, the distortion changes with the position and orientation of the object.

In such a case you can on the one hand rectify images *before* applying the matching. This is a three-step process: First, you must calibrate the camera, i.e., determine its position and orientation and other parameters, using the operator [calibrate_cameras](#) (see Solution Guide III-C, [section 3.2](#) on page 61). Secondly, the calibration data is used to create a mapping function via the operator [gen_image_to_world_plane_map](#), which is then applied to images with the operator [map_image](#) (see Solution Guide III-C, [section 3.4](#) on page 80).

On the other hand, you can also use the uncalibrated perspective deformable matching. It works similar to the shape-based matching but already considers perspective deformations of the model and returns a projective transformation matrix (2D homography) instead of a 2D pose consisting of a position and an orientation. Additionally, a calibrated perspective deformable matching is provided for which a camera calibration has to be applied and which results in the 3D pose of the object. For further information on perspective deformable matching please refer to [section 3.4](#) on page 87 or to the Solution Guide I, [chapter 10](#) on page 89 for the uncalibrated case and Solution Guide III-C, [section 4.6](#) on page 114 for the calibrated case.

Note that if the same perspective view is used for all images, rectifying the images before applying the matching is faster and more accurate than using the perspective deformable matching. But if different perspective views are needed, you must use the perspective matching.

3.3 Local Deformable Matching

Like shape-based matching, local deformable matching extracts contours and matches their shapes against the shapes of previously created models. But in contrast to shape-based matching, slightly deformed contours are not only found but the deformations are returned as result. Note that the actual location of the object is restricted to determining its position, whereas the orientation and scale are interpreted as part of the deformation.

The following sections show

- a first example for a local deformable matching ([section 3.3.1](#)),
- how to select an appropriate ROI to derive the template image from the reference image ([section 3.3.2](#) on page 80),
- how to create a suitable model ([section 3.3.3](#) on page 81),
- how to optimize the search ([section 3.3.4](#) on page 83), and
- how to deal with the results that are specific for the local deformable matching ([section 3.3.5](#) on page 85).



Figure 3.21: (left) “MVTec” logo used for the model creation; (right) deformed logo instance overlaid by the model contours.

3.3.1 A First Example

In this section we give a quick overview of the matching process with local deformable matching. To follow the example actively, start the HDevelop example program `%HALCONEXAMPLES%\hdevelop\Matching\Deformable\find_local_deformable_model.hdev`, which locates differently deformed “MVTec” logos (see [figure 3.21](#)).

Step 1: Prepare the template

First, the template is prepared. In particular, the procedure `create_mvtec_logo_broadened` extracts the “MVTec” logo from an image and creates a synthetic image with a slightly broader logo. The obtained colored image ([figure 3.21](#), left) is transformed into a gray value image from which an ROI, i.e., the template image ([figure 3.22](#), left) is derived.

```
create_mvtec_logo_broadened (LogoImage, 0, 200, Width, Height)
rgb1_to_gray (LogoImage, GrayImage)
gen_rectangle1 (Rectangle, 82, 17, 177, 235)
reduce_domain (GrayImage, Rectangle, ImageReduced)
```

Step 2: Create the model

The template image is then used to create a model of the logo using `create_local_deformable_model`. The contours of the model can be queried with `get_deformable_model_contours`, e.g., to overlay a later match by it to visually inspect the deformations (see [figure 3.21](#), right).

```
create_local_deformable_model (ImageReduced, 'auto', [], [], 'auto', 1, [], \
                              'auto', 1, [], 'auto', 'none', \
                              'use_polarity', 'auto', 'auto', [], [], \
                              ModelID)
get_deformable_model_contours (ModelContours, ModelID, 1)
```

Step 3: Find the object again

The created model is used to find instances of the logo in search images. For demonstration purposes, the example uses synthetic search images that show the logo with various random deformations. After transforming the colored images again into the corresponding gray value images (see, e.g., [figure 3.22](#), middle), the matching is applied with `find_local_deformable_model`.



Figure 3.22: From left to right: template image, corresponding part of the search image, rectified image.

```
rgb1_to_gray (SearchImage, GrayImage)
find_local_deformable_model (GrayImage, ImageRectified, VectorField, \
    DeformedContours, ModelID, 0, 0, 1, 1, 1, 1, \
    0.5, 1, 1, 4, 0.9, ['image_rectified', \
    'vector_field', 'deformed_contours'], \
    ['deformation_smoothness', 'expand_border', \
    'subpixel'], [Smoothness, 0, 1], Score, Row, \
    Column)
```

By default, the operator returns the position of the object in the image. With the parameter `ResultType`, you can additionally specify some iconic objects that are returned. In the example program, all available iconic objects, i.e., a rectified version of the part of the search image that corresponds to the bounding box of the ROI that was used to create the model (see figure 3.22, right), the vector field that describes the deformations of the matched model instance, and the contours of the deformed model instance are queried.

Step 4: Visualize the deformations

To visualize the deformations of a found model instance, the procedure `gen_warped_mesh` creates a regular grid and deforms it with the information that is contained in the returned vector field. For that, the vector field is first converted into the two real-valued images `DRow` and `DCol` using `vector_field_to_real`. For each point of the model, or more precisely of the bounding box that surrounds the template image, `DRow` contains the corresponding row coordinates and `DCol` the corresponding column coordinates of the search image. Then, a regular grid with the size of the model is created. The horizontal and vertical grid lines are created in separate loops using the operators `tuple_gen_sequence` and `tuple_gen_const`. Using the images `DRow` and `DCol`, for each point of a grid line `get_grayval_interpolated` queries the corresponding “deformed” point, i.e., the corresponding coordinates of the point in the search image. These coordinates are used to create polygons that represent the deformed horizontal and vertical grid lines.

```
procedure gen_warped_mesh (VectorField, WarpedMesh, Step)
gen_empty_obj (WarpedMesh)
count_obj (VectorField, Number)
for Index := 1 to Number by 1
    select_obj (VectorField, ObjectSelected, Index)
    vector_field_to_real (ObjectSelected, DRow, DCol)
    get_image_size (VectorField, Width, Height)
    for ContrR := 0.5 to Height[0] - 1 by Step
        Col1 := [0.5:Width[0] - 1]
        tuple_gen_const (Width[0] - 1, ContrR, Row1)
        get_grayval_interpolated (DRow, Row1, Col1, 'bilinear', GrayRow)
        get_grayval_interpolated (DCol, Row1, Col1, 'bilinear', GrayCol)
        gen_contour_polygon_xld (Contour, GrayRow, GrayCol)
        concat_obj (WarpedMesh, Contour, WarpedMesh)
    endfor
    for ContC := 0.5 to Width[0] - 1 by Step
        Row1 := [0.5:Height[0] - 1]
        tuple_gen_const (Height[0] - 1, ContC, Col1)
        get_grayval_interpolated (DRow, Row1, Col1, 'bilinear', GrayRow)
        get_grayval_interpolated (DCol, Row1, Col1, 'bilinear', GrayCol)
        gen_contour_polygon_xld (Contour, GrayRow, GrayCol)
        concat_obj (WarpedMesh, Contour, WarpedMesh)
    endfor
endfor
return ()
```


The grid is displayed together with the returned contours of the deformed logo as shown in [figure 3.23](#).

```
dev_display (SearchImage)
dev_display (WarpedMesh)
dev_display (DeformedContours)
```




Figure 3.23: Search image (before transforming it into a gray value image) with the grid that illustrates the deformations of the matched “MVTec” logo.

The difference between the returned rectified image and the template image is visualized as shown in [figure 3.24](#).

```
crop_domain (ImageReduced, ImagePart)
abs_diff_image (ImagePart, ImageRectified, ImageAbsDiff1, 1)
dev_display (ImageAbsDiff1)
```



Figure 3.24: Difference between the rectified image and the template image.

The following sections go deeper into the details of the individual steps of a local deformable matching and the parameters that have to be adjusted.

3.3.2 Select the Model ROI

As a first step of the local deformable matching, the region of interest that specifies the template image must be selected as described, e.g., in [section 2.1.1](#) on page 17. This region can have an arbitrary shape, i.e., it can also contain holes or consist of several parts that are not connected. Thus, it is possible to create an ROI that contains only “robust” parts of the object. The ROI must be selected so that it contains all prominent structures of the object and also some pixels outside of them, as their immediate surroundings (neighborhood) are needed to obtain the

model. Furthermore, you can speed up the later search using a subsampling (see [section 3.3.3.2](#) on page 82). For that, the ROI should not be too “thin”, because otherwise it vanishes at higher pyramid levels! As a rule of thumb, you are on the safe side if an ROI is $2^{NumLevels-1}$ pixels wide. That is, a width of 8 pixels allows to use 4 pyramid levels. After having selected a suitable ROI, the reduced image is used as template image for the creation of the model.

3.3.3 Create a Suitable Local Deformable Model

Having derived the template image from the reference image, the local deformable model can be created. Note that the local deformable matching consists of different methods to find the trained objects in images. Depending on the selected method, one of the following operators is used to create the model:

- `create_local_deformable_model` creates a model for a local deformable matching that uses a template image to derive the model.
- `create_local_deformable_model_xld` creates a model for a local deformable matching that uses XLD contours to derive the model. Note that after a first match, it is strongly recommended to determine the polarity information for the model with `set_local_deformable_model_metric` (see [section 2.1.3.2](#) on page 22 for details).

Here, we will take a closer look at how to adjust the corresponding parameters. In particular, you can

- specify which pixels are part of the model by adjusting the parameter `Contrast` ([section 3.3.3.1](#)),
- speed up the search by using a subsampling, i.e., by adjusting the parameter `NumLevels`, and by reducing the number of model points, i.e., by adjusting the parameter `Optimization` ([section 3.3.3.2](#)),
- specify which pixels are compared with the model in the later search by adjusting the parameters `MinContrast` and `Metric` ([section 3.3.3.4](#)), and
- adjust some additional (generic) parameters within `GenParamName` and `GenParamValue`, which is needed only in very rare cases ([section 3.3.3.5](#)).

Note that when adjusting the parameters you can also let HALCON assist you:

- Use automatic parameter suggestion:
You can let HALCON suggest suitable values for many of these parameters by either setting the corresponding parameters to the value 'auto' within the operator that is used to create the model or by applying `determine_deformable_model_params` to automatically determine values for a local deformable model from a template image and then decide individually whether you use the suggested values for the creation of the model. Note that both approaches return only approximately the same values and the values that are returned by the operators that are used to create the model are more precise.
- Apply `inspect_shape_model`:
As the local deformable matching uses XLD contours to build the model, which is similar to shape-based matching, you can use `inspect_shape_model` to try different values for the parameters `NumLevels` and `Contrast`. The operator returns the resulting representation of a shape model with multiple pyramid levels and thus allows you to visually check if the object of interest is represented adequately by the model. If several combinations of parameter values do not lead to a satisfying representation of the object, maybe the template image, i.e., the model's ROI was not selected properly. How to select a suitable ROI is described in [section 2.1](#) on page 17.

Note that after the creation of the model, the model can still be modified. In [section 3.3.3.6](#) on page 83 the possibilities for the inspection and modification of an already created model are shown.

3.3.3.1 Specify Pixels that are Part of the Model (Contrast)

For the model those pixels are selected whose contrast, i.e., gray value difference to neighboring pixels, exceeds a threshold specified by the parameter `Contrast` when calling `create_local_deformable_model`. The parameter `Contrast` is similar to the corresponding parameter of shape-based matching that is described in more detail in [section 3.2.4.1](#) on page 58. The only difference is that here small structures are not suppressed within `Contrast` but with the separate generic parameter `'min_size'` that is set via `GenParamName` and `GenParamValue` as described in [section 3.3.3.5](#).

3.3.3.2 Speed Up the Search using Subsampling and Point Reduction (NumLevels, Optimization)

To speed up the matching process, subsampling can be used (see also [section 2.4.2](#) on page 28). There, an image pyramid is created, consisting of the original, full-sized image and a set of downsampled images. The model is then created and searched on the different pyramid levels.

You can specify how many pyramid levels are used via the parameter `NumLevels`. A further reduction of model points can be enforced via the parameter `Optimization`. This may be useful to speed up the matching in the case of particularly large models. The functional principle of both parameters is similar to the parameters that are used for shape-based matching (see [section 3.2.4.2](#) on page 59).

3.3.3.3 Allow a Range of Orientation (AngleExtent, AngleStart, AngleStep) and Scale (Scale*Min, Scale*Max, Scale*Step)

Note that `AngleStart`, `AngleExtent`, `ScaleRMax`, and `ScaleCMax` are not used by this operator. Instead, they have to be specified within the operator `find_local_deformable_model`.

`ScaleRMin` and `ScaleCMin` can be relevant if you expect matches that are smaller than in the original image `Template`. Then, if `NumLevels` is set to `'auto'`, the number of used pyramid levels may decrease.

The parameters `AngleStep`, `ScaleRStep`, and `ScaleCStep` can be determined automatically by setting their values to `'auto'`. Generally, this should lead to good results.

For more information about how these parameters work, please refer to the operator `find_local_deformable_model`.

3.3.3.4 Specify which Pixels are Compared with the Model (MinContrast, Metric)

The parameter `MinContrast` lets you specify which contrast a point in a search image must at least have in order to be compared with the model, whereas `Metric` lets you specify whether and how the polarity, i.e., the direction of the contrast must be observed.

The parameters are similar to the corresponding parameters of shape-based matching that are described in [section 3.2.4.5](#) on page 62. The only difference is that for local deformable matching a further value for `Metric` is available that allows to observe the polarity in sub-parts of the model. A sub-part is a set of model points that are adjacent. For different calculations of the local deformable matching, the model is divided into a set of sub-parts that have approximately the same size (see also [section 3.4.3.5](#) on page 91). When setting `Metric` to `'ignore_part_polarity'`, the different sub-parts may have different polarities as long as the polarities of the points within the individual sub-parts are the same. `'ignore_part_polarity'` is a trade-off between `'ignore_global_polarity'`, which does not work with locally changing polarities, and `'ignore_local_polarity'`, which requires a very fine structured inspection and thus slows down the search significantly.

3.3.3.5 Adjust Generic Parameters (GenParamName, GenParamValue)

Typically, there is no need to adjust the generic parameters that are set via `GenParamName` and `GenParamValue`. But in rare cases, it might be helpful to adjust the splitting of the model into sub-parts, which are needed for different calculations of the local deformable matching, or to suppress small connected components of the model contours.

The size of the sub-parts is adjusted setting `GenParamName` to `'part_size'` and `GenParamValue` to `'small'`, `'medium'`, or `'big'`. Small connected components of the model contours can be suppressed by setting `GenParamName` to `'min_size'`. The corresponding numeric value that is specified in `GenParamValue` describes the number of points a connected part of the object must at least contain to be considered for the following calculations. The effect corresponds to the suppression of small structures that can be applied for shape-based matching within the parameter `Contrast`, which is described in [section 3.2.4.1](#) on page 58.

3.3.3.6 Inspect and Modify the Local Deformable Model

If you want to visually inspect an already created deformable model, you can use `get_deformable_model_contours` to get the XLD contours that represent the model in a specific pyramid level. Note that the XLD contour of the model is located at the origin of the image and thus a transformation may be needed for a proper visualization (see [section 2.5.2](#) on page 30).

To inspect the current parameter values of a model, you query them with `get_deformable_model_params`. This may be necessary if during the creation of the model an automatic parameter selection was used or if the model was created within another program, saved to file with `write_deformable_model`, and read from this file in the current program with `read_deformable_model`. Additionally, you can query the coordinates of the origin of the model using `get_deformable_model_origin`.

After the creation of the model and before you search the object in a search image, you can further modify the model. In particular, you can apply `set_deformable_model_origin` to change its point of reference. But similar to shape-based matching, when modifying the point of reference, the accuracy of the estimated position may decrease (see [section 3.2.6.6](#) on page 69). Therefore, **if possible, the point of reference should not be changed**. Instead, a suitable ROI for the model creation should be selected right from the start (see [section 2.1.2](#) on page 18). 

3.3.4 Optimize the Search Process

The actual matching is applied with `find_local_deformable_model`. In the following, we show how to select suitable parameters for it to adapt and optimize a matching task. In particular, we show how to

- restrict the search space to a region of interest ([section 3.3.4.1](#)),
- restrict the search space by restricting the range of orientation and scale via the parameters `AngleStart`, `AngleExtent`, `ScaleMin`, and `ScaleMax` ([section 3.3.4.2](#)),
- restrict the search space to a specific amount of allowed occlusions for the object, i.e., specify the visibility of the object via the parameter `MinScore` ([section 3.3.4.3](#)),
- specify the used search heuristics, i.e., trade thoroughness versus speed by adjusting the parameter `Greediness` ([section 3.3.4.4](#)),
- search for multiple instances of the model by adjusting the parameters `NumMatches` and `MaxOverlap` ([section 3.3.4.5](#)),
- restrict the number of pyramid levels (`NumLevels`) for the search process ([section 3.3.4.6](#)),
- specify the types of iconic results the matching should return (`ResultType`, [section 3.3.4.6](#)), and
- adjust some additional (generic) parameters within `GenParamName` and `GenParamValue` ([section 3.3.4.8](#)).

3.3.4.1 Restrict the Search to a Region of Interest

The obvious way to restrict the search space and thus speed up the matching is to apply the operator `find_local_deformable_model` not to the whole image but only to an ROI. The corresponding procedure is explained in more detail for shape-based matching in [section 3.2.7](#) on page 70. For local deformable matching you simply have to use `find_local_deformable_model`.

3.3.4.2 Restrict the Range of Orientation and Scale (AngleStart, AngleExtent, Scale*Min, Scale*Max)

When calling `find_local_deformable_model` you can limit the orientation and scale ranges with the parameters `AngleStart`, `AngleExtent`, `ScaleMin`, and `ScaleMax`. This is useful if you can restrict these ranges by other information, which can, e.g., be obtained by suitable image processing operations.

3.3.4.3 Specify the Visibility of the Object (MinScore)

With the parameter `MinScore` you can specify how much of the model must be visible. A typical use of this mechanism is to allow a certain degree of occlusion. The parameter is similar to the corresponding parameter for shape-based matching that is described in more detail in [section 3.2.6.2](#) on page 65.

3.3.4.4 Trade Thoroughness vs. Speed (Greediness)

With the parameter `Greediness` you can influence the search algorithm itself and thereby trade thoroughness against speed, see [section 2.4.3](#) on page 29.

3.3.4.5 Search for Multiple Instances of the Object (NumMatches, MaxOverlap)

All you have to do to search for more than one instance of the object is to set the parameter `NumMatches` to the maximum number of instances you want to find. If you select the value 0, all matches are returned.

Locating more than one instance, the operator `find_local_deformable_model` returns the results for the individual model instances concatenated in the output tuples (see also [section 3.3.5.4](#) on page 87). Note that a search for multiple objects is only slightly slower than a search for a single object.

A second parameter, `MaxOverlap`, lets you specify how much two matches may overlap (as a fraction). This parameter is similar to the corresponding parameter of shape-based matching that is described in [section 3.2.6.4](#) on page 66.

3.3.4.6 Restrict the Number of Pyramid Levels (NumLevels)

The parameter `NumLevels`, which you already specified when creating the model, allows you to use a different (in most cases a more restrictive) value in the search process. By using the value 0 for `NumLevels`, the value specified when creating the model is used.

3.3.4.7 Specify the Iconic Objects Returned by the Matching (ResultType)

By default, a position and a score are returned for each match. If you additionally want to obtain some iconic results, you can set the parameter `ResultType` to `'image_rectified'`, `'vector_field'`, and `'deformed_contours'`. With `'image_rectified'` you get a rectified version of the part of the search image that corresponds to the bounding box of the ROI that was used to create the model ([RectifiedImage](#)), with `'vector_field'` you get the vector field that describes the deformations of the matched model instance ([VectorField](#)), and with `'deformed_contours'` you get the contours of the deformed model instance ([DeformedContours](#)). The individual result types are introduced in more detail in [section 3.3.5](#).

3.3.4.8 Adjust Generic Parameters (GenParamName, GenParamValue)

Besides the parameters that correspond more or less to those used for shape-based matching, local deformable matching allows to adjust some generic parameters that are set via `GenParamName` and `GenParamValue`. Typically, there is no need to adjust them. But if, e.g., a low accuracy of the matching is sufficient, you can speed up the matching by a reduction or deactivation of the subpixel precision or by changing the discretization steps for the orientation or scale that were set during the creation of the model.

Additionally, you can adjust the expected “smoothness” of the deformation. In particular, if the deformations vary locally, i.e., if the deformations are expected to be rather different within a small neighborhood, the value of

'deformation_smoothness' may be set to a smaller value, whereas for an object with an expected “smooth” transition between the deformations the parameter may be set to a higher value.

Furthermore, you can expand the border of the optionally returned rectified image (see [section 3.3.4.7](#)). By default, the section of the search image that is rectified corresponds to the bounding box of the ROI that was used to create the model. If a larger section is needed, e.g., if neighboring parts that are visible in the search image are needed as well for a further processing of the image, you can expand the border of the rectified image with the parameters 'expand_border', 'expand_border_top', 'expand_border_bottom', 'expand_border_left', or 'expand_border_right', respectively. If, e.g., in the example program only the letters “MVT” would have been contained in the model, but the letters “ec” would have been needed as well, the parameter 'expand_border_right' could have been set to 50 to expand the image section by 50 pixels. Note that the rectification is reliable only for the image section that corresponds to the model ROI, as for the neighboring parts of the search image an extrapolation is applied. Thus, the larger the expanded border is, the less accurate is the result towards the image border.

For further information, please refer to the description of [find_local_deformable_model](#) in the Reference Manual.

3.3.5 Use the Specific Results of Local Deformable Matching

The results of local deformable matching differ from those of the other matching approaches. Though it returns also a position and a score, it does not return an orientation or even a scale. Instead, a set of iconic objects can be returned that helps to inspect the deformations of the found object instance. Depending on the selected values of [ResultType](#), the following iconic objects are returned:

- [RectifiedImage](#): the rectified part of the search image that corresponds to the bounding box of the ROI that was used to create the model ([section 3.3.5.1](#)).
- [VectorField](#): the vector field describing the deformations of the found model instance ([section 3.3.5.2](#)).
- [DeformedContours](#): the contours of the deformed model instance ([section 3.3.5.3](#)).

Similar to shape-based matching, local deformable matching can return multiple instances of a model. How to use these multiple instances is shown in [section 3.4.5.1](#) on page 95.

3.3.5.1 The Rectified Image

The first of the iconic objects that can be returned by the local deformable matching is the rectified part of the search image that corresponds to the bounding box of the ROI that was used to create the model (see [figure 3.25](#)).

It can be used, e.g., as was introduced in the first example to show the difference between the rectified image and the corresponding part of the reference image (see [section 3.3.1](#) on page 78). Another use may be that the rectified image is needed for a further image processing task. Note that if a larger section of the search image is needed for the further processing, it is possible to expand the borders of [ImageRectified](#) as is described in [section 3.3.4.8](#) on page 84.

3.3.5.2 The Vector Field

Another iconic object that is returned by the matching is a vector field that can be used, e.g., to visualize the deformations. For that, [VectorField](#) can be transformed into a grid as was shown in the first example (see [section 3.3.1](#) on page 78). Note that to directly display the vector field is not reasonable, as in contrast to a vector field that is returned, e.g., by optical flow, no relative but absolute coordinates are returned. In particular, the vector field that is returned, e.g., by [optical_flow_mg](#) describes the relative movement of the points from one image to another image of the same size. The components of the vector field that can be queried with [vector_field_to_real](#) contain for each point the relative movement in row and column direction. In contrast, the vector field that is returned by local deformable matching describes the movement of points from the model, or more precisely from the bounding box that surrounds the template image, to a search image that typically is larger than the model. Thus, instead of the relative movement of each point, the vector field contains for each model point the corresponding absolute coordinates of the search image.



Figure 3.25: (left) search image; (right) rectified image part.



Figure 3.26: (left) search image; (right) search image overlaid by deformed contours.

Besides visualization purposes, the vector field can be used to get a rectification map that can be used for other applications as well. If, e.g., the deformations of a model instance do not stem from a deformed object but from significant camera distortions, you may use the returned vector field to compensate for the camera distortions in other images. For that, the model either must be a synthetic model (see [section 2.1.3](#) on page 20) or it must be created from an image that was made with a camera without significant distortions. The vector field that is returned by the matching in a search image that is made by the distorted camera can then be transformed into a map with `convert_map_type`. This map can then be used to rectify other images that were made with the same camera.

3.3.5.3 The Deformed Contours

At last, the deformed contours can be returned. Similar to the model contours the `DeformedContours` can be used for visualization purposes (see [figure 3.26](#)) or for a visual inspection of the difference between the model and the found model instance. For the latter, you can either overlay the matching result with the model contours or overlay the model with the deformed contours. Whereas the model contours by default are located at the origin of the image and thus must be transformed for a proper visualization as described in [section 2.5.3.1](#) on page 34, the deformed contours are located already at the position of the match.

3.3.5.4 Deal with Multiple Matches

If multiple instances of the object are searched and found, the results are concatenated in tuples. For example, `Score`, which is a single value for a single model instance, is now a tuple for which the number of elements corresponds to the number of found model instances. The access of the value for a specific instance is similar to the proceeding described in [section 3.1.4.5](#) on page 51 for correlation-based matching.

The iconic output object `DeformedContours` is already returned as a tuple for a single model instance. For multiple instances, the values are simply concatenated. To access them for a specific match, you must know the number of elements returned for a single model instance. This can be obtained by calling `count_obj` for the model contours that can be queried with `get_deformable_model_contours` after the model creation. If, e.g., the model consists of five contours, in the tuple returned for the deformed contours, the first five values belong to the first instance, the next five values belong to the second instance, etc. An example for the access of the values for each instance is described in more detail for calibrated perspective deformable matching in [section 3.4.4.5](#) on page 94.

3.4 Perspective Deformable Matching

Like shape-based matching, perspective deformable matching extracts contours and matches their shapes against the shapes of previously created models. But in contrast to shape-based matching, also perspectively deformed contours can be found. For the perspective deformable matching, an uncalibrated as well as a calibrated version is provided. The following sections show

- a first example for a perspective deformable matching ([section 3.4.1](#)),
- how to select an appropriate ROI to derive the template image from the reference image ([section 3.4.2](#) on page 89),
- how to create a suitable model ([section 3.4.3](#) on page 89),
- how to optimize the search ([section 3.4.4](#) on page 92), and
- how to deal with the results that are specific for the perspective deformable matching ([section 3.4.5](#) on page 95).

3.4.1 A First Example

In this section we give a quick overview of the matching process with perspective deformable matching. To follow the example actively, start the HDevelop example program `%HALCONEXAMPLES%\hdevelop\Applications\Traffic-Monitoring\detect_road_signs.hdev`, which locates road signs independently from their direction. Actually, the program searches an attention sign and a dead end road sign. In the following, we show the proceeding exemplarily for the attention sign.

Step 1: Select the object in the reference image

First, the colored reference image is read and a channel that clearly shows the attention sign (see [figure 3.27](#), left) is accessed with `access_channel`. As the attention sign in the search images is expected to be much smaller than in the reference image, the reference image is scaled with `zoom_image_factor` to better fit the expected size (see [figure 3.27](#), middle). The creation of perspective deformable models is based on XLD contours, which is similar to shape-based matching. Thus, `inspect_shape_model` (see [section 3.2.5.1](#) on page 64) can be used to get suitable values for the parameters `NumLevels` and `Contrast`, which are needed for the creation of shape models as well as for the creation of perspective deformable models. The returned representation of a shape model with multiple pyramid levels can be used to visually check a potential model (see [figure 3.27](#), right).

```
read_image (ImageAttentionSign, 'road_signs/attention_road_sign')
access_channel (ImageAttentionSign, Image, Channel[0])
zoom_image_factor (Image, ImageZoomed, 0.1, 0.1, 'weighted')
inspect_shape_model (ImageZoomed, ModelImages, ModelRegions, 3, 20)
```





Figure 3.27: From left to right: specific channel of colored reference image, scaled image, inspection of the shape model.

Step 2: Create the model

How to obtain a proper template image from a reference image is described in [section 2.1](#) on page 17. In this case, a further reduction to a more restricted ROI is not needed. The model is obtained from the template image with `create_planar_uncalib_deformable_model`.

```
create_planar_uncalib_deformable_model (ImageZoomed, 3, [], [], 0.1, \
                                         ScaleRMin[0], [], 0.05, \
                                         ScaleCMin[0], [], 0.5, 'none', \
                                         'use_polarity', 'auto', 'auto', [], \
                                         [], ModelID)
```

Step 3: Find the object again

To speed up the matching, the search space is restricted first to a rectangular region and then, within this region, to an ROI consisting of a set of blobs (see [figure 3.28](#), left). This set of blobs is obtained by the procedure `determine_area_of_interest`, which exploits the available color information and applies a blob analysis.

```
gen_rectangle1 (Rectangle1, 28, 71, 69, 97)
for Index := 1 to 16 by 1
    read_image (Image, 'road_signs/street_' + Index$.02')
    determine_area_of_interest (Image, Rectangle, AreaOfInterest)
    reduce_domain (Image, AreaOfInterest, ImageReduced)
```



Figure 3.28: (left) ROI derived from the color image; (right) found match for the model of the attention sign.

Within the resulting ROI, the actual matching is applied using the operator `find_planar_uncalib_deformable_model`. Note that the search is applied in the same image channel that was already used for the model creation.

```

for Index2 := 0 to |Models| - 1 by 1
  access_channel (ImageReduced, ImageChannel, Channel[Index2])
  find_planar_uncalib_deformable_model (ImageChannel, Models[Index2], \
                                         0, 0, ScaleRMin[Index2], \
                                         ScaleRMax[Index2], \
                                         ScaleCMin[Index2], \
                                         ScaleCMax[Index2], 0.85, 1, \
                                         0, 2, 0.4, [], [], HomMat2D, \
                                         Score)

```

The result of the matching for each found model instance is a 2D projective transformation matrix (homography) and a score that represents the quality of the matching. Using the homographies, the result is visualized for the successful matches. In particular, the contours of the model are queried and projected into the search image as described in more detail in [section 2.5.4](#) on page 40 (see [figure 3.28](#) on page 88, right).

```

if (|HomMat2D|)
  get_deformable_model_contours (ModelContours, Models[Index2], 1)
  projective_trans_contour_xld (ModelContours, ContoursProjTrans, \
                                HomMat2D)
endif
endfor
endfor

```

Perspective Deform.

The following sections go deeper into the details of the individual steps of a perspective deformable matching and the parameters that have to be adjusted.

3.4.2 Select the Model ROI

As a first step of the perspective deformable matching, the region of interest that specifies the template image must be selected as described, e.g., in [section 2.1.1](#) on page 17. This region can have an arbitrary shape, i.e., it can also contain holes or consist of several parts that are not connected. Thus, it is possible to create an ROI that contains only “robust” parts of the object. The ROI must be selected so that it contains all prominent structures of the object and also some pixels outside of them, as their immediate surroundings (neighborhood) are needed to obtain the model. Furthermore, you can speed up the later search using a subsampling (see [section 3.4.3.2](#)). For that, the ROI should not be too “thin”, because otherwise it vanishes at higher pyramid levels! As a rule of thumb, you are on the safe side if an ROI is $2^{\text{NumLevels}-1}$ pixels wide. That is, a width of 8 pixels allows to use 4 pyramid levels. After having selected a suitable ROI, the reduced image is used as template image for the creation of the model.

3.4.3 Create a Suitable Perspective Deformable Model

Having derived the template image from the reference image, the perspective deformable model can be created. Note that the perspective deformable matching consists of different methods to find the trained objects in images. Depending on the selected method, one of the following operators is used to create the model:

- `create_planar_uncalib_deformable_model` creates a model for an uncalibrated perspective deformable matching that uses a template image to derive the model.
- `create_planar_uncalib_deformable_model_xld` creates a model for an uncalibrated perspective deformable matching that uses XLD contours to derive the model. Note that after a first match, it is strongly recommended to determine the polarity information for the model with `set_planar_uncalib_deformable_model_metric` (see [section 2.1.3.2](#) on page 22 for details).
- `create_planar_calib_deformable_model` creates a model for a calibrated perspective deformable matching that uses a template image to derive the model.
- `create_planar_calib_deformable_model_xld` creates a model for a calibrated perspective deformable matching that uses XLD contours to derive the model. Note that after a first match, it is strongly recommended to determine the polarity information for the model with `set_planar_calib_deformable_model_metric` (see [section 2.1.3.2](#) on page 22 for details).

Here, we will take a closer look at how to adjust the corresponding parameters. In particular, you can

- specify which pixels are part of the model by adjusting the parameter [Contrast](#) ([section 3.4.3.1](#)),
- speed up the search by using a subsampling, i.e., by adjusting the parameter [NumLevels](#), and by reducing the number of model points, i.e., by adjusting the parameter [Optimization](#) ([section 3.4.3.2](#)),
- specify which pixels are compared with the model in the later search by adjusting the parameters [MinContrast](#) and [Metric](#) ([section 3.4.3.4](#)),
- adjust some additional (generic) parameters within [GenParamName](#) and [GenParamValue](#), which is needed only in very rare cases ([section 3.4.3.5](#)), and
- specify the camera parameters and the reference pose that are needed only for the calibrated matching ([CamParam](#) and [ReferencePose](#), see [section 3.5.3.3](#) on page 100).

Note that when adjusting the parameters you can also let HALCON assist you:

- Use automatic parameter suggestion:

You can let HALCON suggest suitable values for many of these parameters by either setting the corresponding parameters to the value 'auto' within the operator that is used to create the model or by applying [determine_deformable_model_params](#) to automatically determine values for a perspective deformable model from a template image and then decide individually if you use the suggested values for the creation of the model. Note that both approaches return only approximately the same values and the values that are returned by the operators that are used to create the model are more precise.

- Apply [inspect_shape_model](#):

As the perspective deformable matching uses XLD contours to build the model, which is similar to shape-based matching, you can use [inspect_shape_model](#) to try different values for the parameters [NumLevels](#) and [Contrast](#). The operator returns the resulting representation of a shape model with multiple pyramid levels and thus allows you to visually check if the object of interest is represented adequately by the model. If several combinations of parameter values do not lead to a satisfying representation of the object, maybe the template image, i.e., the model's ROI was not selected properly. How to select a suitable ROI is described in [section 2.1](#) on page 17.

Note that after the creation of the model, the model can still be modified. In [section 3.4.3.7](#) on page 92 the possibilities for the inspection and modification of an already created model are shown.

3.4.3.1 Specify Pixels that are Part of the Model ([Contrast](#))

For the model those pixels are selected whose contrast, i.e., gray value difference to neighboring pixels, exceeds a threshold specified by the parameter [Contrast](#) when calling [create_planar_uncalib_deformable_model](#) or [create_planar_calib_deformable_model](#). The parameter [Contrast](#) is similar to the corresponding parameter of shape-based matching that is described in more detail in [section 3.2.4.1](#) on page 58. The only difference is that here small structures are not suppressed within [Contrast](#) but with the separate generic parameter 'min_size' that is set via [GenParamName](#) and [GenParamValue](#) as described in [section 3.4.3.5](#).

3.4.3.2 Speed Up the Search using Subsampling and Point Reduction ([NumLevels](#), [Optimization](#))

To speed up the matching process, subsampling can be used (see also [section 2.4.2](#) on page 28). There, an image pyramid is created, consisting of the original, full-sized image and a set of downsampled images. The model is then created and searched on the different pyramid levels.

You can specify how many pyramid levels are used via the parameter [NumLevels](#). A further reduction of model points can be enforced via the parameter [Optimization](#). This may be useful to speed up the matching in the case of particularly large models. The functional principle of both parameters is similar to the parameters that are used for shape-based matching (see [section 3.2.4.2](#) on page 59).

3.4.3.3 Allow a Range of Orientation (AngleExtent, AngleStart, AngleStep) and Scale (Scale*Min, Scale*Max, Scale*Step)

Note that [AngleStart](#), [AngleExtent](#), [ScaleRMax](#), and [ScaleCMax](#) are not used by this operator. Instead, they have to be specified within the operator [find_planar_uncalib_deformable_model](#).

[ScaleRMin](#) and [ScaleCMin](#) can be relevant if you expect matches that are smaller than in the original image [Template](#). Then, if [NumLevels](#) is set to 'auto', the number of used pyramid levels may decrease.

The parameters [AngleStep](#), [ScaleRStep](#), and [ScaleCStep](#) can be determined automatically by setting their values to 'auto'. Generally, this should lead to good results.

For more information about how these parameters work, please refer to the operator [find_planar_uncalib_deformable_model](#).

3.4.3.4 Specify which Pixels are Compared with the Model (MinContrast, Metric)

The parameter [MinContrast](#) lets you specify which contrast a point in a search image must at least have in order to be compared with the model, whereas [Metric](#) lets you specify whether and how the polarity, i.e., the direction of the contrast must be observed.

The parameters are similar to the corresponding parameters of shape-based matching that are described in [section 3.2.4.5](#) on page 62. The only difference is that for perspective deformable matching a further value for [Metric](#) is available that allows to observe the polarity in sub-parts of the model. A sub-part is a set of model points that are adjacent. For different calculations of the perspective deformable matching, the model is divided into a set of sub-parts that have approximately the same size (see also [section 3.4.3.5](#)). When setting [Metric](#) to 'ignore_part_polarity', the different sub-parts may have different polarities as long as the polarities of the points within the individual sub-parts are the same. This is helpful, e.g., if due to the movement in the 3D space different reflections on the object are expected. As these reflections typically are not too small, 'ignore_part_polarity' is a trade-off between 'ignore_global_polarity', which does not work with locally changing polarities, and 'ignore_local_polarity', which requires a very fine structured inspection and thus slows down the search significantly.

3.4.3.5 Adjust Generic Parameters (GenParamName, GenParamValue)

Typically, there is no need to adjust the generic parameters that are set via [GenParamName](#) and [GenParamValue](#). But in rare cases, it might be helpful to adjust the splitting of the model into sub-parts, which are needed for different calculations of the perspective deformable matching, or to suppress small connected components of the model contours.

The size of the sub-parts is adjusted setting [GenParamName](#) to 'part_size' and [GenParamValue](#) to 'small', 'medium', or 'big'.

Small connected components of the model contours can be suppressed by setting [GenParamName](#) to 'min_size'. The corresponding numeric value that is specified in [GenParamValue](#) describes the number of points a connected part of the object must at least contain to be considered for the following calculations. The effect corresponds to the suppression of small structures that can be applied for shape-based matching within the parameter [Contrast](#), which is described in [section 3.2.4.1](#) on page 58.

3.4.3.6 Specify the Camera Parameters and the Reference Pose (CamParam, ReferencePose)

For the calibrated matching, the internal camera parameters ([CamParam](#)) and a reference pose ([ReferencePose](#)) have to be specified. We recommend to obtain both using a camera calibration as described in the Solution Guide III-C, [section 3.2](#) on page 61. Other approaches for determining the pose of the object plane comprise a manual measurement of the extents of the model, which is rather intricate and often inaccurate, stereo vision (see Solution Guide III-C, [chapter 5](#) on page 117), or 3D laser triangulation, e.g., using sheet of light (see Solution Guide III-C, [chapter 6](#) on page 147).

3.4.3.7 Inspect and Modify the Perspective Deformable Model

If you want to visually inspect an already created deformable model, you can use `get_deformable_model_contours` to get the XLD contours that represent the model in a specific pyramid level. In case that the model was generated by `create_planar_calib_deformable_model_xld`, the contours by default are returned in the world coordinate system in metric units. Here, the contours must be transformed by the returned pose for visualizing a match. In all other cases, the contours of the model by default are returned in the image coordinate system in pixel units. For the calibrated matching you can specify the coordinate system in which the contours are returned when calling `get_deformable_model_contours` with the operator `set_deformable_model_param`.

To inspect the current parameter values of a model, you query them with `get_deformable_model_params`. This may be necessary if during the creation of the model an automatic parameter selection was used or if the model was created within another program, saved to file with `write_deformable_model`, and read from this file in the current program with `read_deformable_model`. Additionally, you can query the coordinates of the origin of the model using `get_deformable_model_origin`.

After the creation of the model and before you search the object in a search image, you can further modify the model. In particular, you can apply `set_deformable_model_origin` to change its point of reference and thus, in case of a calibrated matching, also its reference pose. But similar to shape-based matching, when modifying the point of reference, the accuracy of the estimated position may decrease (see [section 3.2.6.6](#) on page 69). Therefore, **if possible, the point of reference should not be changed**. Instead, a suitable ROI for the model creation should be selected right from the start (see [section 2.1.2](#) on page 18).

For the case that you nevertheless need to modify the point of reference, e.g., if the origin should be placed on a specific part of the object, e.g., a corner of the object or a drill hole, and you want to apply a calibrated matching, we here shortly describe the relation between the reference pose, the model pose, and an offset that is manually applied to the point of reference. In particular, when creating the template model for a calibrated matching, the reference pose, which was obtained, e.g., by a camera calibration, is automatically modified by an offset so that its origin corresponds to the projection of the center of gravity of a rectified version of the template image (i.e., the ROI) onto the reference plane and the axes of the obtained model coordinate system are parallel to those of the initial reference pose (see [figure 3.29](#)). If you use `set_deformable_model_origin` to additionally apply an offset to the origin of the thus obtained model pose, the offset is set in image coordinates for the internally rectified template image. To determine the necessary number of pixels for the row and column direction it is convenient to query the rectified contour of the object from the template model with `get_deformable_model_contours`. After setting a manual offset, the obtained image point is automatically projected to the object plane as origin of the modified pose and the pose is adapted accordingly for all following operations.

3.4.4 Optimize the Search Process

The actual matching is applied by one of the following operators:

- `find_planar_uncalib_deformable_model` searches for the best matches of an uncalibrated perspective deformable model. It returns a 2D projective transformation matrix (homography) and the score describing the quality of the match.
- `find_planar_calib_deformable_model` searches for the best matches of a calibrated perspective deformable model. It returns the 3D pose of the object, the six mean square deviations, respectively the 36 covariances of the pose parameters, and the score describing the quality of the match.

In the following, we show how to select suitable parameters for these operators to adapt and optimize a matching task. In particular, we show how to

- restrict the search space to a region of interest ([section 3.4.4.1](#)),
- restrict the search space by restricting the range of orientation and scale via the parameters `AngleStart`, `AngleExtent`, `ScaleMin`, and `ScaleMax` ([section 3.4.4.2](#)),
- restrict the search space to a specific amount of allowed occlusions for the object, i.e., specify the visibility of the object via the parameter `MinScore` ([section 3.4.4.3](#) on page 94),




Figure 3.29: Internal and manual offset for the calibrated perspective matching: the offset from the reference pose to the model pose is calculated automatically from the reference pose, the internal camera parameters, and the template image. An additional offset can be applied manually and is defined in image coordinates.

- specify the used search heuristics, i.e., trade thoroughness versus speed by adjusting the parameter [Greediness](#) (section 3.4.4.4 on page 94),
- search for multiple instances of the model by adjusting the parameters [NumMatches](#) and [MaxOverlap](#) (section 3.4.4.5 on page 94),
- restrict the number of pyramid levels ([NumLevels](#)) for the search process (section 3.4.4.6), and
- adjust some additional (generic) parameters within [GenParamName](#) and [GenParamValue](#) (section 3.4.4.7).

3.4.4.1 Restrict the Search to a Region of Interest

The obvious way to restrict the search space and thus speed up the matching is to apply the operator [find_planar_uncalib_deformable_model](#) or [find_planar_calib_deformable_model](#) not to the whole image but only to an ROI. The corresponding procedure is explained in more detail for shape-based matching in section 3.2.7 on page 70. For perspective deformable matching you simply have to use [find_planar_uncalib_deformable_model](#) or [find_planar_calib_deformable_model](#), respectively.

3.4.4.2 Restrict the Range of Orientation and Scale (AngleStart, AngleExtent, Scale*Min, Scale*Max)

When calling [find_planar_uncalib_deformable_model](#) or [find_planar_calib_deformable_model](#) you can limit the orientation and scale ranges with the parameters [AngleStart](#), [AngleExtent](#), [ScaleMin](#), and [ScaleMax](#). This is useful if you can restrict these ranges by other information, which can, e.g., be obtained by suitable image processing operations.

3.4.4.3 Specify the Visibility of the Object (MinScore)

With the parameter [MinScore](#) you can specify how much of the model must be visible. A typical use of this mechanism is to allow a certain degree of occlusion. The parameter is similar to the corresponding parameter for shape-based matching that is described in more detail in [section 3.2.6.2](#) on page 65.

3.4.4.4 Trade Thoroughness vs. Speed (Greediness)

With the parameter [Greediness](#) you can influence the search algorithm itself and thereby trade thoroughness against speed [section 2.4.3](#) on page 29.

3.4.4.5 Search for Multiple Instances of the Object (NumMatches, MaxOverlap)

All you have to do to search for more than one instance of the object is to set the parameter [NumMatches](#) to the maximum number of instances you want to find. If you select the value 0, all matches are returned.

Locating more than one instance, the operators [find_planar_uncalib_deformable_model](#) and [find_planar_calib_deformable_model](#) return the results for the individual model instances concatenated in the output tuples. That is, the Score, which is a single value for a single model instance, is now returned as a tuple for which the number of elements corresponds to the number of found model instances. For the results that are already returned as tuples for a single model instance, the number of elements multiplies by the number of found instances. These results comprise the projective transformation matrix (HomMat2D) for the uncalibrated matching or the 3D pose (Pose) and the standard deviations or covariances (CovPose) for the calibrated matching. How to access the values for a specific match is shown in [section 3.4.5.1](#). Note that a search for multiple objects is only slightly slower than a search for a single object.

A second parameter, [MaxOverlap](#), lets you specify how much two matches may overlap (as a fraction). This parameter is similar to the corresponding parameter of shape-based matching that is described in [section 3.2.6.4](#) on page 66.

3.4.4.6 Restrict the Number of Pyramid Levels (NumLevels)

The parameter [NumLevels](#), which you already specified when creating the model, allows you to use a different (in most cases a more restrictive) value in the search process. By using the value 0 for [NumLevels](#), the value specified when creating the model is used.

Optionally, [NumLevels](#) can contain a second value, so that you can specify not only the highest but also the lowest pyramid level used for the search. If the search is aborted on a pyramid level that is higher than the first pyramid level, which corresponds to the original, full-sized image, the search becomes faster. On the other hand, the search is then also less robust and less accurate. If objects should be found also in images of poor quality, e.g., if the object is defocused or noisy, you can activate the increased tolerance mode by specifying the second value negatively. Then, the matches on the lowest pyramid level that still provides matches are returned.

3.4.4.7 Adjust Generic Parameters (GenParamName, GenParamValue)

Besides the parameters that correspond more or less to those used for shape-based matching, perspective deformable matching allows to adjust some generic parameters that are set via [GenParamName](#) and [GenParamValue](#). Typically, there is not need to adjust them. But if, e.g., a low accuracy of the matching is sufficient, you can speed up the matching by a reduction or deactivation of the subpixel precision or by changing the discretization steps for the orientation or scale that were set during the creation of the model. To avoid false positive matches, you can also restrict the distortions of the angles and scales. For further information, please refer to the description of [find_planar_uncalib_deformable_model](#) in the Reference Manual.

3.4.5 Use the Specific Results of Perspective Deformable Matching

The perspective deformable matching returns for the uncalibrated case a 2D projective transformation matrix ([HomMat2D](#)), for the calibrated case a 3D pose ([Pose](#)) and either the standard deviations or the covariances ([CovPose](#)), and for both cases a score ([Score](#)) that evaluates the quality of the returned object location. The 2D projective transformation matrix and the 3D pose can be used, e.g., to transform a structure of the reference image into the search image as described in [section 2.5.4](#) on page 40 for the 2D projective transformation matrix and in [section 2.5.5](#) on page 43 for the 3D pose. Similar to the shape-based matching, perspective deformable matching can return multiple instances of a model. How to use these multiple instances is shown in [section 3.4.5.1](#).

3.4.5.1 Deal with Multiple Matches

If multiple instances of the object are searched and found, the results are concatenated in tuples. In particular, [Score](#), which is a single value for a single model instance, is now a tuple for which the number of elements corresponds to the number of found model instances. The access of the value for a specific instance is similar to the proceeding described in [section 3.1.4.5](#) on page 51 for correlation-based matching.

The parameters [HomMat2D](#), [Pose](#), and [CovPose](#) are already returned as tuples for a single model instance. For multiple instances, the values are simply concatenated. To access them for a specific match, you must know the number of elements returned for a single model instance, which is nine for a single projective transformation matrix, seven for a single 3D pose, and either six ('default') or 36 elements (generic parameter 'cov_pose_mode' set to 'covariances') for the single instance's standard deviations or covariances, respectively. Then, e.g., in the tuple containing the 3D poses that are returned by a calibrated matching, the first seven values belong to the first instance, the next seven values belong to the second instance, etc. An example for the access of the values for each instance is the HDevelop example program `%HALCONEXAMPLES%\hdevelop\Applications\Position-Recognition-3D\locate_engine_parts.hdev`, which locates the engine parts shown in [figure 3.30](#). In the example, the 3D poses of the individual model instances are accessed by selecting specific subsets of the tuples using [tuple_select_range](#).

```
find_planar_calib_deformable_model (Image, ModelID, rad(0), rad(360), 1, \
                                   1, 1, 1, 0.65, 0, 0, 3, 0.75, [], \
                                   [], Pose, CovPose, Score)

for Index1 := 0 to |Score| - 1 by 1
    tuple_select_range (Pose, Index1 * 7, ((Index1 + 1) * 7) - 1, \
                       PoseSelected)
    pose_to_hom_mat3d (PoseSelected, HomMat3D)
endfor
```

3.5 Descriptor-Based Matching

Similar to the perspective deformable matching, the descriptor-based matching is able to find objects even if they are perspectively deformed. Again, the matching can be applied either for a calibrated camera or for an uncalibrated camera. In contrast to the perspective deformable matching, the template is not built by the shapes of contours but by a set of so-called interest points. These points are first extracted by a detector and then are described, i.e., classified according to their location and their local gray value neighborhood, by a descriptor.

The following sections show

- a first example for a descriptor-based matching ([section 3.5.1](#)),
- how to select an appropriate ROI to derive the template image from the reference image ([section 3.5.2](#) on page 98),
- how to create a suitable model ([section 3.5.3](#) on page 98),
- how to optimize the search ([section 3.5.4](#) on page 100), and
- how to deal with the results that are specific for descriptor-based matching ([section 3.5.5](#) on page 103).




Figure 3.30: 3D poses of engine parts obtained by a calibrated perspective deformable matching.

3.5.1 A First Example

In this section we give a quick overview of the matching process with (calibrated) descriptor-based matching. To follow the example actively, start the HDevelop program `%HALCONEXAMPLES%\hdevelop\Applications\Object-Recognition-2D\locate_cookie_box.hdev`, which locates a cookie box label that is aligned in different directions.

Step 1: Select the object in the reference image

First, the reference image is read and the image is reduced to a rectangular ROI. That is, a template image is derived that contains only the label of one specific side of a cookie box.

```
read_image (Image, 'packaging/cookie_box_01')
gen_rectangle1 (Rectangle, 224, 115, 406, 540)
reduce_domain (Image, Rectangle, ImageReduced)
```

As a calibrated matching is performed, the camera parameters and the reference pose of the label relative to the camera are needed. For a precise matching, these should be obtained, e.g., by a camera calibration as described in the Solution Guide III-C, [section 3.2](#) on page 61. In the example, the pose is obtained by corresponding points, in particular by the image coordinates and the estimated world coordinates of the corner points of the rectangle from which the ROI was created.

Step 2: Create the model

The camera parameters and the reference pose are input to `create_calib_descriptor_model`, which is applied to create the calibrated descriptor model. Within the operator, the detector is selected and parameters for the detector as well as for the descriptor are specified. In this case, the parameters for the detector are specified with `[]`, i.e., the default values are selected. To be able to reuse the created descriptor model, it is stored to file with `write_descriptor_file`.

```
create_calib_descriptor_model (ImageReduced, CamParam, Pose, \
    'harris_binomial', [], [], ['depth', \
    'number_ferns', 'patch_size', \
    'min_scale', 'max_scale'], [11, 30, 17, \
    0.4, 1.2], 42, ModelID)
write_descriptor_model (ModelID, 'cookie_box_model.dsm')
```

Step 3: Find the object again

`find_calib_descriptor_model` now locates the cookie box label in the search images. The returned 3D poses describe the relation between the world coordinates of the model and the world coordinates of the found matches. Figure 3.31 shows a search image with the visualized result.



Figure 3.31: 3D pose of cookie box label obtained by calibrated descriptor-based matching.

```
for Index := 1 to 10 by 1
  read_image (Image, 'packaging/cookie_box_' + Index$.02')
  find_calib_descriptor_model (Image, ModelID, [], [], [], [], 0.25, 1, \
                              CamParam, 'num_points', Pose, Score)
endfor
```

Step 4: Visualize the match

The result is visualized by different means. First, the interest points of the found model instance are queried with `get_descriptor_model_points` and can be immediately displayed as crosses using `gen_cross_contour_xld`.

```
get_descriptor_model_points (ModelID, 'search', 0, Row, Col)
gen_cross_contour_xld (Cross1, Row, Col, 6, 0.785398)
```

Then, the 3D coordinate system of the 3D pose is visualized by the procedure `disp_3d_coord_system`.

```
disp_3d_coord_system (WindowHandle, CamParam, Pose, 0.07)
```

Finally, the rectangle that presents the outline of the cookie box label is displayed. For that, the rectangle must pass through different transformation steps. In particular, a transformation from the reference image to the world coordinate system (WCS) is followed by a 3D affine transformation within the WCS, which is followed by a transformation from the WCS to the search image.

As regions in HALCON cannot be transformed by a 3D transformation, the corner points of the rectangle must be transformed instead. Here, the world coordinates of the rectangle's corner points are obtained with `image_points_to_world_plane`.

```
image_points_to_world_plane (CamParam, Pose, RowsRoi, ColumnsRoi, 'm', \
                             XOuterBox, YOuterBox)
```

To apply the following 3D affine transformation with `affine_trans_point_3d`, the 3D pose that was returned by the matching must be converted into a 3D homogeneous transformation matrix using `pose_to_hom_mat3d`.

```
pose_to_hom_mat3d (Pose, HomMat3D)
affine_trans_point_3d (HomMat3D, XOuterBox, YOuterBox, [0, 0, 0, 0], \
                    XTrans, YTrans, ZTrans)
```

The 3D coordinates obtained by the 3D affine transformation are then projected with `project_3d_point` into the search image, so that the rectangle can be reconstructed and displayed.

```
project_3d_point (XTrans, YTrans, ZTrans, CamParam, RowTrans, \
                ColTrans)
gen_contour_polygon_xld (Contour, RowTrans, ColTrans)
close_contours_xld (Contour, Contour)
dev_display (Contour)
```

The following sections go deeper into the details of the individual steps of a descriptor-based matching and the parameters that have to be adjusted.

3.5.2 Select the Model ROI

As a first step of the descriptor-based matching, the region of interest that specifies the template image must be selected as described, e.g., in [section 2.1.1](#) on page 17. This region can have an arbitrary shape, i.e., it can also contain holes or consist of several parts that are not connected. Thus, it is possible to create an ROI that contains only “robust” parts of the object. The ROI must be selected so that potentially important interest points are not directly at the border of the region, as the immediate surroundings (neighborhood) of the interest points are needed to obtain the model. Additionally, the part of the object that is visible in the region of interest needs to be planar. After having selected a suitable ROI, the reduced image is used as template image for the creation of the model.

3.5.3 Create a Suitable Descriptor Model

Having derived the template image from the reference image, the descriptor model can be created using

- `create_uncalib_descriptor_model` for an uncalibrated descriptor-based matching or
- `create_calib_descriptor_model` for a calibrated descriptor-based matching.

Except for the camera parameters and the reference pose, which are needed only for the calibrated case, the parameters that have to be adjusted are the same for both operators. Here, we will take a closer look at how to adjust them. In particular, we show how to

- select and adjust the detector ([section 3.5.3.1](#)),
- adjust the descriptor ([section 3.5.3.2](#)), and
- specify the camera parameters and the reference pose that are needed for the calibrated matching ([section 3.5.3.3](#) on page 100).

Note that after the creation of the model, the model can still be modified. In [section 3.5.3.4](#) on page 100 the possibilities for the inspection and modification of an already created model are shown.

3.5.3.1 Select and Adjust the Detector (DetectorType, DetectorParamName, DetectorParamValue)

The interest points that build the model are extracted from the image by the so-called detector. The type of detector is selected via the parameter `DetectorType`. Available types are 'lepetit', 'harris', and 'harris_binomial', which correspond to the HALCON point operators `points_lepetit`, `points_harris`, and

`points_harris_binomial`. 'lepetit' can be used for a very fast extraction of significant points, but the obtained points are not as robust as those obtained with 'harris'. Especially, if a template or search image is very dark or has got a low contrast, 'lepetit' is not recommended. 'harris_binomial' is a good compromise between 'lepetit' and 'harris', because it is faster than 'harris' and more robust than 'lepetit'.

For each detector type a set of generic parameters is available that can be adjusted with the parameters `DetectorParamName` and `DetectorParamValue`. The first one is used to specify the names of the generic parameters and the second one is used to specify the corresponding values. We recommend to apply tests with the selected point operator before creating the model. That is, you apply the corresponding point operator to the template image and visualize the returned points using, e.g., `gen_cross_contour_xld`. For a good result, about 50 to 450 points should be uniformly distributed within the template image. If you have found the appropriate parameter setting for the selected point operator, you can set these parameters also for the model in `create_calib_descriptor_model` or `create_uncalib_descriptor_model`, respectively. Note that in most cases the default values of the detectors (`DetectorParamName` and `DetectorParamValue` set to `[]`) are sufficient.

3.5.3.2 Adjust the Descriptor (`DescriptorParamName`, `DescriptorParamValue`)

The currently implemented descriptor uses randomized ferns to classify the extracted points, i.e., to build characteristic descriptions of the location and the local gray value neighborhood for the interest points.

The descriptor can be adjusted with the parameters `DescriptorParamName` and `DescriptorParamValue`. The first one is used to specify the names of the generic parameters that have to be adjusted and the second one is used to specify the corresponding values.

The parameters can be divided into parameters that control the size of the descriptor and thus allow to control the detection robustness, speed, and memory consumption, and in parameters that control the simulation, especially the spatial range in which the model views are trained. The **size of the descriptor** is controlled by the following parameters:

- 'depth' specifies the depth of the classification fern. Interest points can better be discriminated when selecting a higher depth. On the other hand, a higher depth leads to an increasing runtime.
- 'number_ferns' specifies the number of used fern structures. Using many fern structures leads to a better robustness but also to an increasing runtime.
- 'patch_size' specifies the side length of the quadratic neighborhood that is used to describe the individual interest point. Again, a too large value can disadvantageously influence the runtime.

The selection of values for the depth and the number of ferns depends on your specific requirements. If a **fast online matching** is required, few ferns with a large depth are recommended. If a **robust matching result** is needed, many ferns are needed and a large depth may additionally increase the robustness, although it might also significantly increase the runtime of the matching. If the **memory consumption** is critical, many ferns with a small depth are recommended. For many applications, a trade-off between the different requirements will be needed.

The **simulation, i.e., the training of the model** is controlled by the following parameters:

- 'tilt' is used to switch 'on' or 'off' the projective transformations during the simulation phase, which leads either to an enhanced robustness of the model or to a speed-up of the training.
- 'min_rot' and 'max_rot' define the range for the angle of rotation around the normal vector of the model.
- 'min_scale' and 'max_scale' define the scale range of the model.

The restriction to small ranges for the angle of rotation and the scale can be used to **speed up the training** significantly. But note that during the later applied matching the model can be found only if its angle and scale is within the trained scope. Note further that the training is faster for small images. Thus, you can speed up the training by selecting a small reference image and small template images and setting 'tilt' to 'off'. Note also that the reference image and the search image should have the same size.

An example for the restriction of the orientation and scale of the model is given in the HDevelop example program `%HALCONEXAMPLES%\hdevelop\Applications\Object-Recognition-2D\`

`detect_brochure_pages.hdev` that creates an uncalibrated descriptor model for the matching of different brochure pages. There, the orientation is restricted to an angle range of $\pm 90^\circ$ ('default' is $\pm 180^\circ$) and the scale range is changed to a scaling factor between 0.2 and 1.1 ('default' is between 0.5 and 1.4).

```
create_uncalib_descriptor_model (ImageReduced, 'harris_binomial', [], \
                                [], ['min_rot', 'max_rot', \
                                      'min_scale', 'max_scale'], [-90, 90, \
                                                                    0.2, 1.1], 42, ModelID)
```

3.5.3.3 Specify the Camera Parameters and the Reference Pose (CamParam, ReferencePose)

For a calibrated matching, additionally the camera parameters ([CamParam](#)) and a reference pose ([ReferencePose](#)) have to be specified. We recommend to obtain both using a camera calibration as described in the Solution Guide III-C, [section 3.2](#) on page 61. Other approaches for determining the pose of the object plane comprise a manual measurement of the extents of the model, which is rather intricate and often inaccurate, stereo vision (see Solution Guide III-C, [chapter 5](#) on page 117), or 3D laser triangulation, e.g., using sheet of light (see Solution Guide III-C, [chapter 6](#) on page 147).

3.5.3.4 Inspect and Modify the Descriptor Model

If you want to visually inspect an already created model, you can get the coordinates of the interest points that are contained in it using `get_descriptor_model_points` with `Set` set to 'model'.

```
get_descriptor_model_points (ModelID, 'model', 'all', Row_D, Col_D)
```

To inspect the current parameter values of a model, you query them with `get_descriptor_model_params`. This may be necessary if during the creation of the model an automatic parameter selection was used or if the model was created within another program, saved to file with `write_descriptor_model`, and read from this file in the current program with `read_descriptor_model`. Additionally, you can query the coordinates of the origin of the model using `get_descriptor_model_origin`.

After the creation of the model and before you search the object in a search image, you can further modify the model. In particular, you can apply `set_descriptor_model_origin` to change its origin. But note that this is not recommended because the accuracy of the matching result may decrease, which is shown in more detail for shape-based matching in [section 3.2.6.6](#) on page 69. For the case that you nevertheless need to modify the point of reference and you want to apply a calibrated matching, we refer to the corresponding description for perspective deformable matching in [section 3.4.3.7](#) on page 92. There, the relation between the reference pose, the model pose, and an offset that is manually applied to the point of reference is introduced.

Note that **after the matching**, you can use `get_descriptor_model_points` also to query the interest points of a specific match. Then, `Set` must be set to 'search' instead of 'model'. Here, the interest points of the first found model instance (index 0) are queried.

```
get_descriptor_model_points (ModelIDs[Index2], 'search', 0, Row, Col)
```

Additionally, after the matching, `get_descriptor_model_results` can be used to query selected numerical results that were accumulated during the search like the scores for the correspondences between the individual search points and model points (`ResultNames` set to 'point_classification').

3.5.4 Optimize the Search Process

To locate the same interest points that are stored and described in the model in unknown images of the same or a similar object, the following operators are applied:

- `find_calib_descriptor_model` searches for the best matches of a calibrated descriptor model. It returns the 3D pose of the object and the score describing the quality of the match.

- `find_uncalib_descriptor_model` searches for the best matches of an uncalibrated descriptor model. It returns a 2D projective transformation matrix (homography) and the score describing the quality of the match.

Except for the camera parameters (`CamParam`), the uncalibrated and the calibrated case need the same parameters to be adjusted. Note that the camera parameters, assuming the same setup for the creation of the model and the search, remain the same that were already specified (see [section 3.5.3](#) on page 98). If a different camera is used for the search, we recommend to apply a new camera calibration as described in the Solution Guide III-C, [section 3.2](#) on page 61. In the following, we show how to

- restrict the search space to a region of interest ([section 3.5.4.1](#)),
- adjust the detector for the search, which is recommended only in very rare cases ([section 3.5.4.2](#)),
- adjust the descriptor for the search ([section 3.5.4.3](#)),
- specify the similarity of the object by adjusting the parameter `MinScore` ([section 3.5.4.4](#)),
- search for multiple instances of the object by adjusting the parameter `NumMatches` ([section 3.5.4.5](#)), and
- select a score type by adjusting the parameter `ScoreType` ([section 3.5.4.6](#)).

3.5.4.1 Restrict the Search to a Region of Interest

The obvious way to restrict the search space and thus speed up the matching is to apply the operator `find_uncalib_descriptor_model` or `find_calib_descriptor_model` not to the whole image but only to an ROI. The corresponding procedure is explained in more detail for shape-based matching in [section 3.2.7](#) on page 70. For descriptor-based matching you simply have to use `find_uncalib_descriptor_model` or `find_calib_descriptor_model`, respectively.

3.5.4.2 Adjust the Detector for the Search (`DetectorParamName`, `DetectorParamValue`)

The parameters that control the detection, i.e., the extraction of the interest points from the image, are adjusted via `DetectorParamName` and `DetectorParamValue`. They correspond to the detector parameters that were already specified during the creation of the model (see [section 3.5.3](#) on page 98). In most cases, they should not be changed for the search. That is, you simply pass an empty tuple `()` to the parameters.

In rare cases, especially when there are significant illumination changes between the reference image and the search image, you may change the parameter values. For example, if a search image is extremely dark and 'lepetit' was selected as detector, you can set 'min_score' to a smaller value.

Generally, to test if it is necessary to change any of the parameter values, you can apply the point operator that corresponds to the detector not only to the reference image, which was proposed for the creation of the model (see also [section 3.5.3.1](#) on page 98), but also to the search image. Again, about 50 to 450 uniformly distributed points should be extracted to get a good matching result. If the point operator needs different parameters for the reference and the search image, it might be necessary to change the values of the corresponding detector parameters accordingly.

3.5.4.3 Adjust the Descriptor for the Search (`DescriptorParamName`, `DescriptorParamValue`)

The parameters that control the determination of the correspondences between the interest points of the search image and those of the model are adjusted via `DescriptorParamName` and `DescriptorParamValue`. Two generic parameters can be set:

- The parameter 'min_score_descr' can be set to a value larger than 0.0 (but preferably below 0.1) to increase the minimal classification score that determines if the individual points are considered as potential matches. Thus, the number of points for further calculations is reduced, so that the speed of the matching can be enhanced. But note that this speed-up is obtained at the cost of a reduced robustness, especially if the number of extracted points is low.

- The parameter 'guided_matching' can be switched off by setting it from 'on' to 'off'. If the guided matching is switched on, which is the default, the robustness of the estimation of the model's location is enhanced. In particular, points are extracted from the search image and classified by the descriptor. The points that were accepted by the classification are used to calculate an initial projective 2D (uncalibrated case) or homogeneous 3D (calibrated case) transformation matrix. This is used then to project all model points into the search image. If a projected model point is near to one of the originally extracted points, i.e., independently from its classification, this point is used for the final calculation of the homography that is returned by the matching as 2D projective transformation matrix or 3D pose, respectively. As typically without the classification more points can be used for the calculations, the obtained homography is more robust. On the other hand, in some cases the runtime of the matching may increase up to 10%. Thus, if robustness is less important than speed, 'guided_matching' can be switched off.

The HDevelop example program %HALCONEXAMPLES%\hdevelop\Applications\Object-Recognition-2D\detect_brochure_pages.hdev is an example for setting 'min_score_descr' to a higher value (0.003) to speed up the matching. As the number of points in the images is large enough, the matching is still sufficiently robust.

```
find_uncalib_descriptor_model (ImageGray, ModelIDs[Index2], \
                             'threshold', 600, \
                             ['min_score_descr', \
                              'guided_matching'], [0.003, 'on'], \
                             0.25, 1, 'num_points', HomMat2D, \
                             Score)
```

3.5.4.4 Specify the Similarity of the Object (MinScore)

The parameter [MinScore](#) specifies the minimum score a potential match must have to be returned as match. The score is a value for the quality of a match, i.e., for the correspondence, or “similarity”, between the model and the search image. Note that for the descriptor-based matching, different types of score are available for the output parameter [Score](#) (see [section 3.5.4.6](#)). But for the input parameter [MinScore](#), always the score of type 'inlier_ratio' is used. It calculates the ratio of the number of point correspondences to the number of model points. In most cases, [MinScore](#) should be set to a value of at least 0.1. To speed up the search, it should be chosen as large as possible, but of course still as small as necessary for the success of the search, as, e.g., a value of 1.0 is rather unlikely to be reached by a matching.

3.5.4.5 Search for Multiple Instances of the Object (NumMatches)

All you have to do to search for more than one instance of the object is to set the parameter [NumMatches](#) to the maximum number of instances you want to find. If you select the value 0, all matches are returned.

Locating more than one instance, the operators [find_uncalib_descriptor_model](#) and [find_calib_descriptor_model](#) return the results for the individual model instances concatenated in the output tuples. That is, the [Score](#), which is typically a single value for a single model instance (see [section 3.5.4.6](#) for exceptions), is now returned as a tuple for which the number of elements corresponds to the number of found model instances. The number of elements for the projective transformation matrix (HomMat2D) or the 3D pose (Pose), which are tuples already for a single model instance, is the number of elements of the single instance multiplied by the number of found instances. How to access the values for a specific match is shown in [section 3.5.5.1](#). Note that a search for multiple objects is only slightly slower than a search for a single object.

3.5.4.6 Select a Suitable Score Type (ScoreType)

The parameter [ScoreType](#) is used to select the type of score that will be returned in the parameter [Score](#). Available types are 'num_points' and 'inlier_ratio':

- For 'num_points', the number of point correspondences per instance is returned. As any four correspondences define a mathematically correct homography between two images, this number should be at least 10 to assume a reliable matching result.

- For 'inlier_ratio' the ratio of the number of point correspondences to the number of model points is returned. Although this parameter may have a value between 0.0 and 1.0, a ratio of 1.0 is rather unlikely to be reached by a matching. Yet, objects having an inlier ratio of less than 0.1 should be disregarded.

Typically, one of both score types is selected, but it is also possible to pass both types in a tuple. Then, the result for a single found model instance is returned in a tuple as well.

3.5.5 Use the Specific Results of Descriptor-Based Matching

The descriptor-based matching returns for the uncalibrated case a 2D projective transformation matrix ([HomMat2D](#)), for the calibrated case a 3D pose ([Pose](#)), and for both cases a score ([Score](#)) that evaluates the quality of the returned object location. The 2D projective transformation matrix and the 3D pose can be used, e.g., to transform a structure of the reference image into the search image as described in [section 2.5.4](#) on page 40 for the 2D projective transformation matrix and in [section 2.5.5](#) on page 43 for the 3D pose. The score can be interpreted as described in [section 3.5.4.6](#) on page 102. Similar to the shape-based matching, descriptor-based matching can return multiple instances of a model. How to use these multiple instances is shown in [section 3.5.5.1](#).

3.5.5.1 Deal with Multiple Matches

If multiple instances of the object are searched and found, the results are concatenated in tuples. In particular, [Score](#), which is typically a single value for a single model instance (see [section 3.5.4.6](#) on page 102 for exceptions) is now a tuple for which the number of elements corresponds to the number of found model instances. The access of the value for a specific instance is similar to the proceeding described in [section 3.1.4.5](#) on page 51 for correlation-based matching.

The parameters [HomMat2D](#) or [Pose](#) are already returned as tuples for a single model instance. For multiple instances, the values are simply concatenated. To access them for a specific match, you must know the number of elements returned for a single model instance, which is nine for a single projective transformation matrix and seven for a single 3D pose. Then, e.g., in the tuple returned for the 3D poses, the first seven values belong to the first instance, the next seven values belong to the second instance, etc. An example for the access of the values for each instance is described in more detail for calibrated perspective deformable matching in [section 3.4.4.5](#) on page 94.

Index

- 2D affine transformation, 31
- 2D projective transformation, 32
- 2D rigid transformation from points and angle, 31
- 2D transformation, 30
- access results of multiple matching model instances (descriptor-based), 103
- access results of multiple matching models (local deformable), 87
- access results of multiple matching models (perspective deformable), 95
- align image using shape-based matching, 38
- align regions of interest using shape-based matching, 35
- allow orientation range for matching (correlation-based), 49
- allow orientation range for matching (local deformable), 82
- allow orientation range for matching (perspective deformable), 91
- allow orientation range for matching (shape-based), 65
- correlation-based matching (NCC)
 - first example, 47
 - overview, 47
- create (train) matching model (correlation-based), 48
- create (train) matching model (descriptor-based), 98
- create (train) matching model (local deformable), 81
- create (train) matching model (perspective deformable), 89
- create (train) matching model (shape-based), 58, 64
- create matching model from DXF file, 24
- create matching model from synthetic template, 21
- create matching model from XLD contours, 22
- create template, 17
- descriptor-based matching
 - first example, 96
 - overview, 95
- determine training parameters for matching (correlation-based), 48
- determine training parameters for matching (local deformable), 81
- determine training parameters for matching (perspective deformable), 89
- determine training parameters for matching (shape-based), 58, 64
- display results of matching, 34
- find matching model (correlation-based), 50
- find matching model (descriptor-based), 100
- find matching model (local deformable), 83
- find matching model (perspective deformable), 92
- find matching model with perspective distortion, 77
- find model for matching, 17
- find multiple matching models (correlation-based), 51
- find multiple matching models (shape-based), 72
- find multiple model instances (correlation-based), 51
- find multiple model instances (descriptor-based), 102
- find multiple model instances (local deformable), 84
- find multiple model instances (perspective deformable), 94
- find multiple model instances (shape-based), 66
- get matching model contours (local deformable), 83
- get matching model origin (correlation-based), 49
- get matching model origin (descriptor-based), 100
- get matching model origin (perspective deformable), 92
- get matching model parameters (descriptor-based), 100
- get matching model parameters (local deformable), 83
- get matching model parameters (perspective deformable), 92
- get multiple model results of matching (shape-based), 76
- get multiple results of matching (shape-based), 76
- image pyramid, 25
- inspect matching model (correlation-based), 49
- inspect matching model (descriptor-based), 100
- inspect matching model (local deformable), 83
- inspect matching model (perspective deformable), 92
- local deformable matching
 - first example, 78
 - overview, 77
- localize and visualize results of matching, 33
- matching approaches, 47
- matching model contours (perspective deformable), 92
- matching model parameters (correlation-based), 49
- matching model points (descriptor-based), 100
- model creation (training), 17
- perspective deformable matching, 87
- re-use matching model (correlation-based), 49

- re-use matching model (descriptor-based), 100
- re-use matching model (local deformable), 83
- re-use matching model (perspective deformable), 92
- re-use model, 25
- read matching model (correlation-based), 49
- read matching model (descriptor-based), 100
- read matching model (local deformable), 83
- read matching model (perspective deformable), 92
- rectify image for matching (shape-based) to adapt to new camera orientation, 77
- reference point for matching, 18
- region of interest from matching model, 19
- region of interest from matching model (shape-based), 56
- restrict orientation range for matching (correlation-based), 50
- restrict orientation range for matching (local deformable), 84
- restrict orientation range for matching (perspective deformable), 93
- results of matching, 29, 30
- rotate 2D homogeneous matrix, 31
- scale 2D homogeneous matrix, 31
- score, 45
- select score type for matching (descriptor-based), 102
- select suitable matching approach, 8
- set camera parameters for matching (descriptor-based), 100
- set camera parameters for matching (perspective deformable), 91
- set clutter parameters for matching (shape-based), 67
- set generic parameters (local deformable), 82
- set generic parameters (perspective deformable), 91
- set generic parameters for speedup (local deformable), 84
- set generic parameters for speedup (perspective deformable), 94
- set matching model metric (correlation-based), 49
- set matching model metric (local deformable), 82
- set matching model metric (perspective deformable), 91
- set matching model metric (shape-based), 62
- set matching model origin (correlation-based), 49
- set matching model origin (descriptor-based), 100
- set matching model origin (local deformable), 83
- set matching model parameters (correlation-based), 53
- set number of pyramid levels for matching (correlation-based), 53
- set number of pyramid levels for matching (local deformable), 84
- set number of pyramid levels for matching (perspective deformable), 94
- set threshold to extract matching model (local deformable), 82
- set threshold to extract matching model (perspective deformable), 90
- set threshold to extract matching model (shape-based), 58
- set training model origin (perspective deformable), 92
- shape-based matching
 - first example, 55
 - overview, 53
- specify accuracy for matching (correlation-based), 53
- specify accuracy for matching (shape-based), 69
- specify iconic objects for matching (local deformable), 84
- specify object similarity for matching (correlation-based), 50
- specify object similarity for matching (descriptor-based), 102
- specify object visibility for matching, 65
- specify object visibility for matching (local deformable), 84
- specify object visibility for matching (perspective deformable), 94
- speed up matching, 27
- speed up matching (local deformable), 84
- speed up matching (perspective deformable), 94
- speed up matching (shape-based), 29, 73
- speed up matching using greediness, 29
- speed up matching with subsampling, 28
- speed up matching with subsampling (correlation-based), 49
- speed up matching with subsampling (local deformable), 82
- speed up matching with subsampling (perspective deformable), 90
- speed up matching with subsampling (shape-based), 59
- timeout for matching (correlation-based), 53
- translate 2D homogeneous matrix, 31
- use 3D pose result of template matching, 43
- use deformed contours (local deformable), 86
- use extended parameter estimation for shape-based matching, 63
- use homography results of matching, 40
- use rectified image (local deformable), 85
- use region of interest, 28
- use region of interest for matching (correlation-based), 48, 50
- use region of interest for matching (descriptor-based), 98, 101
- use region of interest for matching (local deformable), 80, 83
- use region of interest for matching (perspective deformable), 89, 93
- use region of interest for matching (shape-based), 70
- use results of matching (descriptor-based), 103
- use results of matching (local deformable), 85
- use results of matching (perspective deformable), 95
- use results of matching (shape-based), 76
- use synthetic model for matching, 20

use vector field (local deformable), [85](#)

visualize shape-based matching score contribution
contours, [75](#)

write matching model (correlation-based), [49](#)

write matching model (descriptor-based), [100](#)

write matching model (local deformable), [83](#)

write matching model (perspective deformable), [92](#)

